
Программирование на языке Паскаль
Язык был разработан в 1970 г. профессором Никлаусом Виртом из Цюриха, и относится к языкам третьего поколения. Паскаль широко использовался как международный язык для описания алгоритмов, а в последнее время - для обучения программированию. Этот язык наиболее приспособлен для составления корректных программ методами структурного программирования и реализации метода "сверху вниз".

1. СТРУКТУРА ПРОГРАММЫ НА ЯЗЫКЕ ПАСКАЛЬ

Программа состоит из двух разделов:

1) описания используемых данных;

2) последовательности операторов, которые обрабатывают данные.

Такое разделение позволяет избежать многих ошибок в программировании. Для отделения разделов используются специальные служебные слова (английские), которые для других целей не должны применяться.

Таким образом, структура Паскаль-программы имеет вид рис. 2.1.

Рис. 2.1. Структура паскаль-программы

Program, Begin и End – служебные слова.

Имя – это имя программы (последовательность латинских букв и цифр, начинающаяся с буквы).

End.– конец программы (точка обязательна, она указывает, что это последнее слово в программе). Точка с запятой – указатель окончания программной единицы (оператора, описателя). Такое разделение позволяет писать несколько операторов в строке. Но обычно для удобства чтения программы и работы с нею операторы записывают по одному в строке и даже лесенкой. Отступы позволяют видеть структуру программы.

2. ОПИСАНИЕ ДАННЫХ

В любом языке данные делятся на 2 типа:

1) постоянные (константы);

2) переменные.

Постоянные не изменяют своего значения при выполнении программы, а переменные – изменяют. Перед выполнением программы компилятор записывает в память ЭВМ значения констант и выделяет области (без записи значений) для переменных. Описание данных связывает с ними некоторые имена (идентификаторы), по которым их можно найти в памяти ЭВМ.

Имя (идентификатор) – последовательность латинских букв и цифр, начинающаяся с буквы. Длина его не ограничивается, но значимыми являются только первые 63 символа.

Пример.
Правильные имена

Неправильные имена

X



1Y

– начинается с цифры

SUM1


S2.1

– точка (не буква)

Faitstudent

Fait Student
– есть пробел
F2S13


X*Y

– неправильный знак

2.1. Константы

Константы бывают двух типов:

1) числовые;

2) строковые.

Числа представляются в обычной форме, дробная часть отделяется от целой десятичной точкой. Строки заключаются в апострофы, например: 'АБРАКАДАБРА'. В них могут использоваться и русские буквы.

Описанию констант предшествует служебное слово

Const,

после которого перечисляются имена и значения всех констант, разделяемые точкой с запятой. Общий вид описания констант приведен ниже:

Const

имя1 = значение1;

имя2 = значение2;




. . . . . . . . . 

Пример описания констант.

PROGRAM FIRST;

CONST

FAM  = 'Иванов';

FAK  = 'АИТ';

KURS  = 1;

GRUP  = 5;

2.2. Переменные

Типов переменных больше, чем констант. Это могут быть:

1) числа:

а) целые (тип Integer);

б) вещественные (Real);

2) символы (тип Char);

3) строки (только для TURBO PASCAL, тип String).

Целые числа могут принимать значения в диапазоне от -32768 до 32767. Переменной символьного типа соответствует только 1 символ, а строке – до 255 символов.

Описанию переменных предшествует служебное слово

Var,

за которым следует перечисление их имен с указанием типа, разделяемых точкой с запятой. Общий вид описания переменных приведен ниже:

Var

имя1: тип1;

имя2: тип2;

список имен: тип;

Пример.

Var

x,y : Integer;

kod : Char;

Описатели констант и переменных располагаются в строгом порядке:

сначала Const
. . . . . . . ,

а потом Var
. . . . . . .

Если констант нет, то первый раздел отсутствует.

Для устранения ошибок в вычислениях при описании переменных может быть задан диапазон их значений. Такое описание представляется с помощью границ диапазона в виде min_значение .. max_значение. Тип переменной определяется этими границами.

Пример.

Var

Kurs: 1..5;

{целые от 1 до 5}

Grup: 5..9;

KodFak: 'A'..'Z';
{символы от 'A' до 'Z'}

Если при выполнении программы переменным присваиваются значения, лежащие вне диапазона, программа завершается с выдачей сообщения об ошибке.

3. КОММЕНТАРИИ

Комментарии – это пояснения к программе, содержащие указания о методах решения и обозначениях переменных.

Они облегчают чтение программы и ее изменение, если это требуется. Мы будем считать необходимыми комментарии к сложным программам, содержащим более 15–20 операторов. Комментарии могут располагаться в любом месте программы (обычно после заголовка и операторов) и заключаются в парные фигурные скобки {} или сочетание соответствующей круглой скобки и звездочки: (*  *):

{






} 

или

(*






*)

Это начало




   Это конец

комментария




комментария

Комментарии могут содержать русские буквы и слова. Они игнорируются компилятором.

4. ОПЕРАТОРЫ ЯЗЫКА ПАСКАЛЬ

Оператор (Statement) – это основная единица программы, которая выполняет определенные действия над данными.

Операторы в паскаль-программе располагаются после описания данных и отделяются от него служебным словом BEGIN. Их можно условно разделить на 2 типа:

1) обрабатывающие, которые обрабатывают данные;

2) управляющие, которые управляют выполнением обрабатывающих операторов.

К первому типу относятся операторы:

1) ввода;

2) вывода;

3) присваивания (вычисления новых значений);

ко второму — операторы:

1) изменяющие естественный порядок выполнения других операторов в зависимости от некоторого условия или безусловно;

2) обеспечивающие повторение некоторых групп операторов.

5. ОПЕРАТОРЫ ОБРАБОТКИ ДАННЫХ

Программа для ЭВМ должна позволять обрабатывать любые наборы данных. При этом данные должны быть описаны как переменные. Во время выполнения программы переменным присваиваются конкретные значения, относящиеся к решаемой задаче.

Присваивание может быть выполнено двумя способами:

1) вводом данных (чисел, символов, строк) с устройства ввода (клавиатуры);

2) вычислением нового значения по формулам, в которых используются уже определенные ранее значения.

Присваивание сводится к записи в память ЭВМ числа, строки или символа на место, отведенное под заданную переменную во время компиляции. Для этого используются специальные операторы.

5.1. Операторы ввода

Они служат для присвоения переменным конкретных значений. В Паскале определены два таких оператора. В общем виде они записываются так:

Read (список ввода);

Readln (список ввода);
Список ввода – это одно или несколько имен переменных, разделяемых запятыми.

Пример.
Read(number);






(5.1)

Readln(X,Y);







(5.2)

Операторы в таком виде еще не вводят данные. Это произойдет только при выполнении программы! При этом на устройстве ввода (клавиатуре) оператором-человеком набираются необходимые значения.

Пример. 

В памяти ЭВМ на месте переменных появятся следующие значения:
Number = 20;  X=10; Y=4.5.

Таким образом, числа при вводе с клавиатуры представляются как последовательности цифр. Конец числа – пробел или символ, отличный от цифры. Если вводится несколько чисел, то они разделяются пробелами или символом конца строки (Enter). Количество пробелов перед первым числом и между числами может быть произвольным.

Отличие оператора Readln от Read в том, что он позволяет считать целую строку, поэтому он чаще используется для ввода строк.

Пример.

VAR

A,B,C,D : CHAR;

. . . . . . .

Readln(A,B,C,D);

Если при выполнении этого оператора пользователь наберет на клавиатуре последовательность символов 
ФАИТ и нажмет клавишу <Enter>, то в памяти ЭВМ переменные из примера получат следующие значения: А = 'Ф'; В = 'А'; С = 'И'; D = 'Т'.

5.2. Операторы вывода

Служат для передачи данных из памяти ЭВМ на устройство вывода (дисплей, печать). Их тоже два. В общем виде они записываются так:

Write(список вывода);

Writeln(список вывода);       

Список вывода – это последовательность строковых констант или имен переменных, разделяемых запятыми.

Пример.

При выполнении оператора

Writeln('ФАКУЛЬТЕТ: ',А,В,С,D);

на экране появится текст

ФАКУЛЬТЕТ: ФАИТ

При выполнении оператора

Write(number,X);

на экране появится текст


20

10

Строковая константа выводится без изменений. При выводе переменных на экране появляются не имена, а значения из памяти ЭВМ (не Х, а 10). Отличие оператора Writeln от Write заключается в том, что после оператора Writeln происходит переход на новую строку. Если оператор Writeln записать без списка вывода, т.е. в виде 






Writeln;

то будет выполнен переход на новую строку.

Данные при выводе можно разместить в заданном месте экрана или бумаги. Для этого после имени переменной или строковой константы указывается ширина поля листа (экрана), в которое ее нужно поместить. Ширина поля – это число символов, которое уместится в поле, включая и первые пробелы. Для имен, следующих за первым, она указывает отступ от предыдущего элемента на экране.

Общий вид: 

Write (имя1:n1, имя2:n2,...,стр.1:m1,...);
Пример.

Program Vivod;

Const

X = 10;

Y = 45;

Begin

Writeln('Результаты':19);

Writeln;

       { Пропуск строки }

Write ('X=’, X:5, 'Y=':14, Y:5);

End.

Форму вывода вещественных чисел можно задать с использованием двух значений в виде

Имя: w:n,

где w — общая ширина поля, отводимого под число (включая знак и десятичную точку);

n — количества цифр в дробной части.

Пример.



Write(z:8:2);
Оператор Write в диалоговых программах используется в паре с Read для подсказок при вводе, которые имеют вид

Writeln('Введите X,Y');

Read(X,Y);

5.3. Вычисление по формулам. Оператор присваивания

Формулы в языках программирования называются выражениями. Выражение – это правило для нахождения нового значения.

Вид формулы (выражения) зависит от типа входящих в него переменных. Причем, над каждым типом данных могут выполняться свои разрешенные типы операций. Таким образом, выражение должно:

1) содержать данные одного типа (числа или символы, но не числа и символы одновременно);

2) использовать операции, разрешенные для данных этого типа. Так, для чисел разрешены следующие арифметические операции

Для целых






Для вещественных
Сложение (+)







+,

Вычитание (-)







-,
Умножение (*)







*,

Деление нацело – Div






/,

Вычисление остатка от деления – Mod



нет аналога

Пример.

Операция Div – деление нацело с отбрасыванием остатка.

5 Div 2 ( 2

и
7 Div 3 ( 2

Операция MOD – дает остаток от деления одного числа на другое, которое называют модулем.

5 Mod 2 ( 1, 

7 Mod 3 ( 1
и 
18 Mod 9 ( 0.

Выражение содержит имена констант и переменных, соединенные знаками разрешенных для них операций и разделенные скобками.

Пример. Пусть переменные x, y, z описаны как целые. Тогда арифметическое выражение х2 + y/z  в программе должно быть записано так: x*x + y div z

При вычислении значения выражения учитывается старшинство (приоритет) операций:

1) *, /, Div, Mod 
– высший;

2) +, - 


– низший.

Для изменения порядка расставляются скобки. Количество скобок может быть произвольным, но каждой открывающейся скобке должна соответствовать своя закрывающаяся. 

Знаки операций в выражениях не опускаются. 

Операции одного приоритета выполняются слева направо.

Оператор присваивания позволяет вычислить и запомнить в памяти ЭВМ новое значение переменной или изменить уже существующее значение.

Оператор состоит из двух частей, разделенных знаком присваивания ":=", и имеет вид

имя := выражение;

Пример.

Y:= X*X + A Mod Z;

Здесь типы переменных Y,X,A, и Z – целые. К моменту выполнения оператора числовые значения X, A и Z должны быть заданы.

Как отмечалось выше, оператор присваивания позволяет изменять значение уже существующей переменной.

Пример.

K:= K + 1; 

{ увеличение K на 1 }.

6. ЛИНЕЙНЫЕ ПРОГРАММЫ

Программы этого типа реализуют линейные алгоритмы и содержат операторы ввода, присваивания и вывода, которые выполняются в естественном порядке (в порядке расположения).

Пример. Ввести в ЭВМ трёхзначное число и вывести на экран в столбик его цифры, начиная со старшей.

Для выделения десятичных цифр воспользуемся операциями Div и Mod.

Алгоритм (вывод со старшей цифры)

1. Ввести число

2. Выделить и вывести на экран цифры

3. Закончить.

Уточняем п. 2 и получаем алгоритм.

1. Ввести число.

2.1. Ц1 = число div 100;                 {получение старшей цифры}. 

2.2. Вывести Ц1;

2.3. Ц2 = (число mod 100) div 10;  {получение средней цифры }.

2.4. Вывести Ц2;

2.5. Ц3 = число mod 10;                  {получение младшей цифры }.

2.6. Вывести Ц3;

3. Закончить.

Программа, реализующая алгоритм, будет иметь вид

Program cifra;

Const

d = 10;

s = 100;

Var

num, c1, c2, c3 : integer;

Begin

Writeln('Введите целое трехзначное число');

Read(num);

C1:= num div S;

Writeln ('Цифра 1: ', С1);

C2:= (num mod S) div d;

Writeln('Цифра 2: ', С2);

С3:= num mod d;

Writeln('Цифра 3: ', С3);

end.

7. УПРАВЛЯЮЩИЕ ОПЕРАТОРЫ

7.1. Разветвляющиеся алгоритмы. Оператор IF (ЕСЛИ)

Разветвляющийся алгоритм предполагает выполнение вычислений по одному из возможных направлений в зависимости от исходных данных или промежуточных результатов. Ветвление в таких алгоритмах производится при проверке некоторых условий. Для этой цели в Паскале используется специальный оператор - IF (если).

Правило записи оператора IF следующее:

IF условие then

Оператор 1

Else

Оператор 2;
Схема алгоритма его работы приведена на рис. 2.2.

Пример. Приводимому ниже фрагменту программы 

if X<>0 then

Y:= A/X

else

Y:= A;

соответствует следующий ход вычислений:

если в момент выполнения программы окажется, что X ≠ 0, то будет вычислено Y = A/X. 

Если же окажется, что X = 0, то будет вычислено Y = А.

Оператор IF – это один оператор, поэтому ";" после оператора 1, т.е. перед Else, не ставится.

Часть со служебным словом Else может отсутствовать. В этом случае выполняется или не выполняется только оператор 1 (см. рис. 2.3).

Пример.

If X>0 Then

K:= K+1;  {увеличение К на 1 (подсчет числа положительных Х)}

Если с помощью оператора IF необходимо управлять группой операторов, то эта группа заключается в блок вида

Begin
операторы

End;

Пример.


if Big<Littl then

Begin

    A := Big;
{перестановка чисел местами:}

  Big := Littl;

{Big – большее}

Littl := A;


{Litll – меньшее}

End;

Операторы 1 и 2 могут быть любыми, в том числе и условными, т.е. допустима вложенность операторов IF друг в друга на любую глубину.

Пример.

if X>0 then

if Y>0 then

Z:= Sqrt(X*X+Y*Y)

else

Z:= X*X

else

Z:= Y*Y;

При этом важно не запутаться, какой Else к какому If относится. 

В общем случае служебное слово Else связывается с ближайшим словом If, которое еще не связано со служебным словом Else. Для удобства чтения и понимания логики программы записывайте Else под соответствующим ему If.

7.2. Пример разветвляющейся программы

Задача. Ввести в ЭВМ 3 целых числа. Расположить их в порядке возрастания.

Алгоритм

1. Ввести числа.

2. Расположить по порядку.

3. Вывести числа.

Уточняем п. 2.

2.1. Если первое число больше второго, то поменять их местами (выполнить их перестановку).

2.2. Если второе больше третьего, то выполнить их перестановку.

2.3. Если первое больше второго, то выполнить их перестановку.

Проверим правильность алгоритма для чисел: 125  67  3

Тогда после выполнения пунктов алгоритма должны получаться следующие результаты: 

после пункта   1: 
125
67
3

после пункта 2.1: 
  67
125
3

после пункта 2.2:
  67
3
125

После пункта 2.3:
    3
67
125

Программа будет иметь вид

Program Order;

Var

A,B,C : Integer;
{основные переменные}

X : Integer; {вспомогательная (рабочая) переменная}

Begin

Writeln('Введите 3 целых числа');

Read(А,В,С);

       {п. 2.1}

IF A>B Then

Begin
{перестановка}

X:= A;

A:= B;

B:= X;

End;

       {п. 2.2}

IF B>C Then

Begin
{перестановка}

X:= B;

B:= C;

B:= X;

End;

       {п. 2.3}

IF A>B Then

Begin
{перестановка}

X:= A;

A:= B;

B:= X;

End;

Writeln('Результат:', A:5,B:5,C:5);

End.

7.3. Оператор CASE

Оператор If позволяет выбрать один из двух возможных путей решения. В ряде случаев таких путей бывает гораздо больше. При этом также необходимо выбрать один из них. Применение оператора If для таких задач неэффективно.

В Паскале для выбора одного из нескольких путей решения используется оператор Case (вариант из) – обобщение условного оператора. В общем виде он записывается так:

Case выражение Of
Значения1 : Оператор1;

Значения2 : Оператор2;

. . . . . . . . . .

ЗначенияN : ОператорN;

End;
{ Конец списка выбора }
Выражение записывается относительно специальной переменной выбора (селектора), которая может быть любого перечисляемого типа (Integer, Char).

Значения1, . . . , Значения N – значения выражения, при которых выполняются Оператор1, . . . , Оператор N соответственно.

Если селектор принял значение, не принадлежащее ни одному из перечисленных, то оператор Case пропускается и выполняется следующий за ним оператор (как при неполном If).

Пример.
Case  Num Mod 2  Of

0 : Writeln('Четное');

1 : Writeln('Нечетное');

End;

или
Case Month Of

1,2,3 :
Writeln('Первый квартал');

4,5,6 :
Writeln('Второй квартал');

End;

Если после Значений i необходимо выполнять несколько операторов, они заключаются в блок:

Значенияi:
Begin

операторы_i

End;

Рассмотрим еще одну задачу, демонстрирующую эффективность применения оператора Case.

Пример. Перевести цифру римской системы счисления в ее десятичный эквивалент. Римская цифра вводится c клавиатуры (в виде похожей на нее латинской буквы), а затем на экране должен появиться ее десятичный эквивалент.

Программа будет иметь вид

Program Rim;

Var 

R : Char;

N : Integer;

Begin

Writeln('Введите римскую цифру');

Readln(R);

Case R Of

'M': N:= 1000;

'D': N:= 500;

'C': N:= 100;

'L': N:= 50;

'X': N:= 10;

'V': N:= 5;

‘I': N:= 1

End; { Case }

Write ('Десятичное число:', N);

End.

8. ЦИКЛИЧЕСКИЕ АЛГОРИТМЫ И ПРОГРАММЫ

8.1. Общая схема цикла

Основными задачами, решаемыми на ЭВМ являются такие, в которых необходимо выполнять одни и те же операции над различными данными. Программы таких задач содержат гораздо меньше операторов, чем в действительности выполняет ЭВМ. Это достигается за счет того, что некоторые операторы выполняются много раз.

Определение 1. Группа многократно повторяющихся операторов называется циклом, а содержащая их программа – циклической.

Количество повторений цикла определяется значением некоторой управляющей переменной.

Определение 2. Переменная, управляющая повторением цикла, называется параметром цикла.
Общая схема циклического участка программы (алгоритма) может быть представлена в следующем виде (рис. 2.4).


В блоке "Подготовка" задаются начальные значения переменных, используемых в вычислениях, и, в первую очередь, параметра цикла.

Блок "Условие выполнения" в зависимости от значения параметра либо начинает (продолжает) выполнение цикла, либо прекращает его (выход из цикла).

В блоке "Рабочая часть" содержатся многократно повторяемые операторы, которые и выполняют необходимые вычисления.

В блоке "Изменение параметра цикла" задается способ изменения параметра и выполняется переход к его новому значению.

Блоки 1, 2 и 4 являются необходимыми и типовыми для циклических алгоритмов и реализуются специальными операторами цикла. Блок 3 (Рабочая часть) - индивидуален для каждой задачи.

По типу параметра и условию окончания циклы делятся на:

1) циклы со счетчиком;

2) итерационные.

В циклах со счетчиком количество повторений заранее известно, а в итерационных – нет. В последних выход из цикла осуществляется при определенном условии (достижении заданной точности, появлении некоторой ситуации).

8.2. Циклы со счетчиком

Для их организации в Паскале используется оператор For (для), который в общем виде записывается так:

For Параметр := НачальноеЗначение  to  КонечноеЗначение  do 

 Оператор1; 






(8.1)

или

For Параметр := НачальноеЗначение  Downto  КонечноеЗначение  do 

 Оператор1; 






(8.2)

Параметром цикла в этом операторе может служить переменная так называемого перечисляемого типа:

1) целого;

2) символьного.

Оператор вида (8.1) работает в соответствии со схемой, приведенной на рис. 2.5,а. Параметр пробегает значения от начального до конечного через 1. При этом конечное значение должно быть больше или равно начальному. В противном случае оператор не выполняется.

В операторе вида (8.2) использовано слово Downto. При этом значения параметра убывают (изменяются от большего – начального значения – до меньшего с шагом -1). Работу оператора иллюстрирует схема рис. 2.5,б.

Если начальное значение равно конечному, то в обоих случаях (при То и Downto) оператор выполняется один раз.

В общем случае (если параметр пробегает несколько значений) оператор 1 выполняется несколько раз.

Пример: 
{к величине S прибавляется 10 чисел от 1 до 10}

For i := 1 To 10 Do

S:= S+i;

Если многократно нужно выполнить несколько операторов, то они помещаются в блок операторов – между служебными словами Begin и End.

Пример. Ввести в ЭВМ 10 трёхзначных чисел и распечатать их цифры в обратном порядке, т.е. начиная с младшей.

Алгоритм (укрупненный).

1. Для номера от 1 до 10 выполнить

Ввести и обработать число

2. Конец.

Детализируем этот алгоритм.

1. Для номера от 1 до 10 выполнить

1.1. Ввести число.

1.2. Ц.1 = число mod 10;    {выделяем младшую цифру}

1.3. Число = число div 10;       {выделяем две старших цифры}

1.4. Ц.2 = число mod 10;            {выделяем среднюю цифру}

1.5. Ц.3 = число div 10;              {выделяем старшую цифру}

1.6. Вывести Ц1, Ц2, Ц3.

Программа для этого алгоритма будет иметь вид

Program Dig;

Const

D=10;

Var
Num,C1,C2,C3 : Integrer;  {исходное число и цифры}

i : Integrer;


  {номер числа }

Begin

For I := 1 To D Do

Begin

Readln(Num);

C1:=Num Mod D;

Num:=Num Div D;

C2:=Num Mod D;

C3:=Num Div D;

Writeln(C1:2,C2:2,C3:2)

End;

End.

Пункты 1.2 – 1.6 можно также выполнить в цикле вида

For C2:=1 To 3 Do

Begin

C1:=Num Mod D;

Writeln (C1);

Num:=Num Div D;

End;

8.3. Итерационные циклы

Это циклы, количество повторений которых заранее неизвестно. Выход из цикла определяется условием, записанным относительно параметра.

Пример. Вывести степени двойки, не превышающие 10000. 

Количество выводимых чисел – степеней двойки – заранее неизвестно (хотя и может быть вычислено), но есть условие окончания их поиска: выводимое число должно быть меньше 10000.

Для циклов с неизвестным числом повторений в Паскале используются 2 оператора:

1) с пост-условием;

2) с пред-условием.

8.3.1. Оператор цикла с пост-условием

Это оператор Repeat (ПОВТОРЯТЬ), который в общем виде записывается так:

Repeat

Операторы

Until условие;

Схема его работы представлена на рис. 2.6.

В теле цикла записывается один или несколько операторов, которые будут выполняться многократно. Они располагаются между служебными словами Repeat и Until, поэтому обрамлять их словами Begin и End не нужно. После слова Until записывается условие выхода из цикла (по значению параметра).

При этом необходимо обязательно изменять параметр внутри цикла, иначе цикл будет выполняться бесконечно (произойдет "зацикливание"). В соответствии со схемой цикл выполняется по крайней мере 1 раз.

Пример. Изобразить степени двойки, не превышающие 10000.

Алгоритм

1. Положить число = 2.

2. Повторять:

2.1. Изобразить число.

2.2. Число: = число * 2.

Пока число НЕ превысит 10000.

3. Закончить

Программа для этого алгоритма будет иметь вид

Program Two;

Var

Num: Integer;

Begin

Num:=2;

Repeat

Writeln (Num);

Num:=Num*2;

Until Num>1000;

End.

8.3.2. Оператор цикла с пред-условием

Он отличается от оператора Repeat тем, что условие выполнения цикла проверяется до начала. Цикл выполняется, если оно истинно. Оператор с пред-условием начинается служебным словом While. Общая форма его записи:

While условие do
Оператор1;

Схема работы оператора приведена на рис. 2.7.
В этом операторе условие также записывается относительно параметра. Если в теле цикла необходимо выполнить несколько операторов, то они заключаются в блок следующего вида:

Begin

операторы

End;
В отличие от Repeat цикл While может не выполняться ни разу.

Пример. Вычисление множителей последовательности целых чисел (см. пример словесного описания алгоритма из раздела "ПОНЯТИЕ АЛГОРИТМА И ЕГО СВОЙСТВА").

Алгоритм

1. Ввести первое число.

2. Пока нет признака конца последовательности выполнять
2.1. Положить число простое

2.2. Для делителя от 2 до ]число/2[ выполнять
если число делится на делитель, то 

а) вывести делитель и

б) изменить предположение – «не простое».

2.3. Если число простое, то вывести: "простое число"

2.4. Прочитать следующее число.

3. Закончить.

Программа для этого алгоритма будет иметь вид

Program Dividers;

Const

d=2;

Var

Num, Del,P : Integer;

Begin

Writeln ('Введите число');
{ п.1 }

Readln(Num);

{ п.2 }

While Num <>0 Do

Begin

P := 0;


{ п.2.1 – число простое }

For  Del := d To Num Div d Do

Begin

If Num Mod Del=0 Then

Begin

Write(Del);

P := 1;

{ число не простое }

End;

End;

If P=0 Then{ п.2.3 }

Writeln('Число ',Num,' - простое');

Writeln('Введите число');   { п.2.4 }

Read(Num);

End;

End.

В программе использован оператор While, так как уже первое число может быть нулем и цикл выполнять не нужно.

Сведем основные операторы языка Паскаль и описатели данных в табл. 2 и 3 соответственно.

Основные операторы языка Паскаль







Таблица 2
Название
Общий вид
Действие
Пример записи

1. Присваивания
Переменная = выражение;
Значение выражения записывается на место переменной 
X:= 0;

Y:=A/B+C*D;

2. Ввода
Read(список);

Readln(список);
Перепись данных с устройства ввода в ОП. Readln – переход на новую строку
Read (X,Y,A);

3. Вывода
Write(список);

Writeln(список);
Перепись данных из ОП на устройство вывода. Writeln - переход на новую строку
Write ('X=',X:8:5.);

4. Условный
If  Условие  Then
Оператор1

Else

Оператор2;
Ветвление (выполнение оператора1 или оператора2)
If X>0 Then
         Y := Sqrt(X)

Else

         Y := Sqr(X);

5. Выбора
Case Выражение Of
Значение1 : Оператор1;

    . . . . . . 

ЗначениеN : ОператорN;

End;
Ветвление по N направленям
Case Month  Of
3..5 : Writeln(' Весна');

6..8 : Writeln(' Лето');

End;

6. Цикла со 
счечиком (For)
For Пар.:= Н.З. To К.З. Do
Оператор 1;

For Пар.:= Н.З.Downto К.З. Do
Оператор 1;
Выполнение оператора1  n  раз: 

   n = кон.знач. – нач.знач.+1 

Выполнение оператора1  n  раз: 

   n = нач.знач. – кон.знач.+1
S := 0;

For I := 1 To 20 Do
   S := S+I;

7. Цикла с 
постусловием
Repeat

        Операторы

Until Условие;
Выполнение операторов пока условие "ложно". Выполняется не менее 1 раза.
Num := 1;

Repeat

    Num := Num*2;

Until Num>1000;

8. Цикла с 
предусловием 
While  Условие  Do
Оператор1;
Выполнение оператора 1 пока условие "истинно".
Num := 1;

While Num<1000 Do
    Num := Num*2;

9. Заголовок
 программы
Program  Имя;
Начало программы
Program Pr1;

10. Начало блока операторов
Begin
Обозначение начала блока 
Begin

11. Конец  блока, программы
End; 

End.
Обозначение конца блока
или программы
End;

End.

Таблица 3.

 Описание данных

Название
Общий вид
Действие
Пример

1. Описание констант
Const

Имя1 = Зн1;

Имя2 = Зн2;
Записывается в ОП значение на место имени
Const

X = 0;

A = ’СТРОКА’;

2. Описание переменных
Var
Имя : Тип;

Имя : Min..Max;
Отводит в ОП место для переменных, но не задаёт их значения.
Var

X, Y : Integer;

A, B, C:Char;

9.4. Функции для работы с символами

Символы в ЭВМ представляются двоичными кодами. Для латинских букв коды упорядочены по возрастанию в соответствии их расположением в алфавите от 'А' до 'Z', от 'a' до 'z'.

В Паскале имеются 4 функции, позволяющие работать с символами:

Ord (символ) – вычисляет порядковый номер (код) символа;

Chr (код) – определяет символ с заданным кодом;

Pred (символ) – определяет символ, имеющий код, предшествующий данному;

Succ (символ) – определяет символ, имеющий код, следующий за данным.

Пример.
Ord('A')  =>  65;



Pred('C')  =>  'B'

Сhr (66)  =>  'B';



Succ('C')  =>  'D'

Пример. Вывести на экран дисплея номера (коды) букв латинского алфавита.

Program Bukva;

Var

Sim : Char;

Begin

For Sim:= 'A' To 'Z' Do

Writeln(Ord(Sim),'     ',Sim);

End.

Основные функции языка Паскаль сведены в табл. 4.

Таблица 4.

Функции языка Паскаль

Форма записи 
Название
Типы

       аргумента                        результата

1. Abs(X)
Абсолютная величина
целый, вещественный
соответствует типу аргумента

2. Sqr(X)
Квадрат X  
целый, вещественный
соответствует типу аргумента

3. Sqrt(X)
Корень квадратный
вещественный
вещественный

4. Exp(X)
Экспонента   eX
вещественный
вещественный

5. Ln(X)
Натуральный логарифм
вещественный
вещественный

6. Sin(X)
Синус
вещественный (в рад.)
вещественный

7. Cos(X)
Косинус
вещественный (в рад.)
вещественный

8. Atan(X)
Арктангенс
вещественный
вещественный (рад.)

9. Trunc(X)
Округление с 
отбрасыванием
вещественный
целый (длинный)

10. Round(X)
Округление до 
ближайшего целого
вещественный
целый (длинный)

11. Succ(X)
Следующий за X элемент
перечислимый (целый, символьный)
совпадает с типом аргумента

12. Pred(X)
Предшествующий X 
элемент
перечислимый (целый, символьный)
совпадает с типом аргумента

13. Ord(X)
Порядковый номер X
перечислимый (целый, символьный)
совпадает с типом аргумента

14. Chr(X)
Символ для ASCII кода X
целое (байт)
символьный

15. Odd(X)
Нечётность
целое
булевский

16. Eoln(X)
Конец строки
нет
булевский

9.5. Массивы

Массив – это совокупность элементов, имеющих одинаковое имя и отличающихся номером (индексом) их расположения в последовательности.

Типы данных, которые мы до сих пор рассматривали, называют скалярными. Их использование не всегда эффективно. Так, если в задаче данные являются родственными, например оценки группы студентов, то скалярные типы применять неудобно. Для определения количества студентов (KolStud), имеющих балл (BallStud) выше среднего (Sr), пришлось бы записать 25 операторов вида

KolStud := 0;

If BallStud1 > Sr  Then 

KolStud := KolStud + 1;

If BallStud2 > Sr Then

KolStud := KolStud + 1;

и т. д.

Поскольку обрабатываются данные одного типа, удобнее в этом случае использовать массивы.

Примеры массивов:
1) последовательность чисел (оценок)

х1, x2, x3, ..., xn – 1 индекс



2) матрица чисел


Примером представления данных в виде матрицы может служить запись коэффициентов системы линейных уравнений.

Для описания массива используется служебное слово Array (Массив). Кроме того, необходимо задать:

1) имя массива;

2) диапазон изменения каждого индекса..

Количество индексов называется размерностью массива. По количеству индексов различают:

– одномерные массивы (последовательность чисел, символов);

– двумерные массивы (матрицы);

– трехмерные и т.д. массивы.

Как и скалярные переменные, массивы описываются в разделе описания переменных (т.е. после служебного слова Var), например:

Var

Имя_Массива : Array[Диапазоны_индексов] Of Тип_Элементов_Массива;

где Диапазоны_индексов – список границ изменения каждого индекса.

Пример.

Const

N=100;

M=5;

Var

X:Array[1..N] Of Real;   {массив вещественных чисел}

A:Array[1..M,1..M] Of Integer; {мaтрица целых чисел}

Следует различать описание массива и значение его элемента при обработке (внутри программы, после первого Begin). Элемент массива в программе записывают в виде

Имя_Массива [индекс] или

Имя_Массива [список индексов]

Пример.

X[2] => x2
A[i,j] => Aij
Ваll [k,i+2] => Ballk,i+2
Индекс – это число, переменная или выражение перечисляемого целого или символьного типа.

Рассмотрим пример программы обработки массива.

Задача. Ввести в ЭВМ оценки студентов одной группы по одному предмету. Подсчитать средний балл и количество оценок выше средней.

Алгоритм

1. Ввести оценки.

2.1. Положить сумму баллов = 0.

2.2. Для номера оценки (i) от 1 до Количества_студентов_в_группе выполнить

Подсчитать сумму баллов.

3. Вычислить средний балл.

4. Положить количество оценок выше средней,  равной 0.

5. Для номера (i) от 1 до  Количества_студентов_в_группе выполнить

Подсчитать количество оценок выше средней.

6. Закончить.

Уточняем алгоритм.

1.1. Ввести количество оценок n.

1.2. Для номера оценки (i) от 1 до n выполнить.

1.2.1. Ввести баллi
2.1. Положить сумму баллов  = 0.

2.2. Для номера оценки (i) от 1 до n выполнить

2.2.1. Сумма_баллов = сумма_баллов + баллi
3. Вычислить средний балл = сумма_баллов / n.

4. Положить количество оценок выше средней  = 0.

5. Для номера от 1 до n выполнить

5.1. Если баллi > среднего, то 

количество = количество +1.

Программа для этого алгоритма будет иметь вид

Program Ball;

Const

M=25;

Var

B:Array[1..M] OF Integer;

S:Real;

{сумма и средний балл}

N,I,K:Integer;
{K - счетчик}

Begin

Writeln('Введите количество оценок');
{ п.1.1 }

Read (N);

Writeln('Введите оценки');

{ п.1.2 }

For I:= 1 To N Do

Read(B[I]);

S:=0;                 { п.п. 2.1 - 2.2 }

For I:= 1 To N Do

S := S+B[I];

S := S/N; { п.3 - средний балл }

Writeln('Средний балл - ',S);

K: = 0;

     { п.4 }

For I:= 1 To N Do    { п.5 }

If B[I]>S Then

K:=K+1;

Writeln('Количество оценок выше средней = ',K);

End.

Итак, программа обработки массива – это цикл, в котором параметром является индекс (номер элемента). Если индексов несколько, то для обработки массивов используются циклы, вложенные друг в друга. Количество таких циклов зависит от числа индексов.

10. Примеры программ обработки массивов

Пример 1. Ввести в ЭВМ массив, содержащий не более 100 чисел вещественных чисел. Вывести числа, стоящие на четных местах в массиве.

Рrogram Wtr;

Const

Nmax=100;

Var

X : Array[1..Nmax] Of Real;

i, n : Integer;

Begin
Writeln('Введите количество чисел');

Readln(n);

Writeln('Введите массив');

For i:= 1 To n Do

Read (X[i]);

For I:= 1 To n div 2 Do

Write (X[2*i]);

End.

Пример 2. Ввести в ЭВМ массив (последовательность) из n чисел (n<=100). Подсчитать количество положительных и отрицательных чисел, их суммы и средние значения.

Алгоритм (общий подход).

1. Ввести числа.

2. Положить количества положительных, отрицательных чисел и суммы = 0

3. Для номера числа от 1 до n выполнить

вычислить соответствующие суммы и количества

4. Вывести суммы, количества и средние.

5. Закончить.

Уточним этот алгоритм.

1.1. Ввести количество чисел  n
1.2. Для номера числа (i) от 1 до n выполнить

ввести числоi
2. Положить количества положительных, отрицательных чисел и их сумм =0

3. Для номера числа от 1 до n выполнить

3.1. Если числоi >0, то

а) Количество_Пол_Чисел = Количество_Пол_Чисел +1 и

б) Сумма_Пол_Чисел = Сумма_Пол_Чисел + числоi
иначе

если числоi >0, то

а) Количество_Отр_Чисел = Количество_Отр_Чисел+1 и

б) Сумма_Отр_Чисел = Сумма_Отр_Чисел + числоi
4.1 Если Количество_Пол_Чисел. > 0, то

Среднее_Пол_Чисел =  Сумма_Пол_Чисел / Количество_Пол_Чисел

4.2 Если Количество_Отр_Чисел > 0, то

Среднее_Отр_Чисел = Сумма_Отр_Чисел / Количество_Отр_Чисел

Программа для этого алгоритма будет иметь вид

Program Count_sred;

Const

Nmax = 100;

Var

X : Array[1..Nmax] Of Real;

N, I, Ko, Kp : Integer;

So, Sp, Sro, Srp : Real;

Begin

Writeln('Введите количество чисел');

Readln(n);

Writeln('Вводите элементы массива');

For i:=1 To n Do

Read(X[i]);

Ko:=0;

Kp:=0;

So:=0;

Sp:=0;

For i:=1 To N Do

If X[i]>0 Then

Begin

Kp:=Kp+1;

Sp:=Sp+X[i]

End

Else

If X[i]>0 Then

Begin

Ko:=Ko+1;

So:=So+ X[i]

End;

If Ko<>0 Then

Begin

Sro:= So/Ko;

Writeln('Количество отрицательных чисел = ', Ko);

Writeln('Сумма отрицательных чисел = ', So:8:2);

Writeln('Среднее отрицательных чисел = ',Sro:8:2);

End
Else
  Writeln('В массиве нет отрицательных чисел.');

If Kp<>0 Then

  Begin

Srp:= Sp/Kp;

Writeln('Количество положительных чисел = ', Kp); 

Writeln('Сумма положительных чисел = ', Sp:8:2); 

Writeln('Среднее положительных чисел = ', Srp:8:2);

End
Else
  Writeln('В массиве нет положительных чисел.');

End.

11. Особенности алгоритмов и программ 
с накапливанием

Вычисление суммы и произведения нескольких чисел, а также подсчет количества в большинстве языков программирования выполняется постепенно, путем накапливания.

1) Сумма вычисляется по формуле

сумма := cумма + очередное слагаемое;

Перед вычислениями начальному значению суммы нужно присвоить ноль.

2) Произведение вычисляется по формуле

произведение := произведение * очередной сомножитель

Перед вычислениями начальному значению произведения нужно присвоить 1.

3) Подсчет количества выполняется по формуле, подобной формуле для вычисления суммы:

количество := количество + 1

Начальное значение количества должно быть = 0.

12. Алгоритм нахождения минимума и максимума

Задача 1. Ввести в ЭВМ последовательность из n чисел x1, х2,..., хn, n<200. Найти минимальное и максимальное из них.

При поиске минимума или максимума используется дополнительная переменная min (или max), которой:

1) вначале присваивается значение первого числа из последовательности (x1), т.е. принимается, что первое число является текущим минимумом (максимумом);

2) начиная со второго числа, производится сравнение этого числа со значением переменной min (или max) и если число из массива меньше min (больше max), то на место min (max) записывается это число. Теперь это число будет текущим минимумом (максимумом).

Ясно, что после просмотра всех чисел последовательности в переменной min (или max) будет находиться окончательное значение минимума (или максимума). Программа для этого алгоритма будет иметь вид

Program Minmax;

Const

Nmax=200;

Var

X : Array [1..Nmax] Of Real;

Min, Max : Real;

    N, i : Integer;

Begin

Writeln('Введите количество чисел');

Readln(n);

Writeln('Вводите элементы массива');

For i := 1 To n Do

Read(X[i]);

Min := X[1];

Max := X[1];

For i : =2 To n Do

If Max < X[i] Then

Max := X[i]

Else

If Min > X[i] Then

Min :=X [i];

Writeln ('Min= ', Min:8;2,'Max= ', Max:8:2);

End.

В некоторых случаях бывает необходимо не только найти минимальное или максимальное число в последовательности, но и запомнить его номер (индекс).

Задача 2. Найти минимум в последовательности из n чисел (n>=100) и номер минимального числа. В этой задаче необходимо использовать ещё одну переменную: не только Min, но и номер, – например Imin. Каждый раз, когда Min присваивается очередное значение, нужно запомнить и индекс (номер) числа. Соответствующая программа может быть представлена в виде

Program Minimum;

Const

Nmax = 100;

Var

X : Array[1..Nmax] Of Real;

Min : Real;

n, i, Imin : Integer;

Begin

Writeln('Введите количество чисел');

Readln(n);

Writeln('Вводите элементы массива');

For i:=1 To n Do

Read(X[i]);

Min := X[1];

Imin := 1;

For i := 2 To n Do

If Min > X[i] Then

Begin

Min := X[i];

Imin := i;

End;

Writeln('Минимум = ', Min:8:2, ' Номер = ', Imin)

End.

13. Задача сортировки

Сортировка – это расположение чисел в порядке возрастания или убывания.

Наиболее распространенный и простой метод сортировки – метод "пузырька". Он требует минимального объема памяти для данных, но затраты времени на реализацию этого метода велики. Суть метода "пузырька" в следующем.

Пусть дано n чисел, которые необходимо расположить (для определенности) в порядке возрастания. При упорядочении выполняются следующие операции:

1) числа сравниваются попарно: первое со вторым; второе с третьим; i-тое – с (i+1) - тым;

2) если меньшее стоит в паре на втором месте (числа в паре не упорядочены по возрастанию), то сравниваемые числа меняются местами.

За один такой просмотр массива минимальное число "вытолкнется", по крайней мере, на одно место вверх (вперед), а максимальное – переместится в самый конец (вниз), т.е. минимальное число как легкий пузырек воздуха в жидкости постепенно "всплывает" в начало последовательности. Отсюда – название метода. За n-1 просмотр произойдет полное упорядочение массива при любом исходном расположении чисел в нем. Рассмотрим работу метода на примере, приведенном на рис. 2.8.

В пятом просмотре перестановок нет. Сортировка окончена.

Рис. 2.8. Иллюстрация метода сортировки "пузырьком"

Алгоритм сортировки
1. Ввести n чисел.

2. Для номера_просмотра от 1 до n-1 выполнить

2.1. Для номера_числа (i) от 1 до n-1 выполнить

Если число[i] > число[i+1], то поменять их местами.

3. Вывести отсортированную последовательность.

4. Закончить.

Программа для этого алгоритма будет иметь вид

Program Sort;

Const

Nmax = 100;

Var

X : Array [1..Nmax] Of Real;

A : Real;

n, k, i : Integer;

Begin

Writeln('Введите количество чисел');

Readln(n);

Writeln('Введите массив чисел');

For i := 1 To n Do

Read (X[i]);

{ Сортировка }

For k := 1 To n-1 Do

For i := 1 To n-1 Do

If X[i] > X[i+1] Then

Begin

A:=X[i];

X[i]:=X[i+1];

X[i+1]:=A

End;

Writeln('Отсортированный массив чисел:');

For i:=1 To n Do

Write (X[i]);

End.

Глубину просмотра можно уменьшать, основываясь на том, что большие числа "опускаются" вниз (в конец последовательности) и затем не переставляются:

For k := 1 To n-1 Do

For i := 1 To n-k Do

If X[i] > X[i+1] Then

 . . . . . . . .

Сокращение количества просмотров улучшает временные характеристики метода. Из алгоритма можно понять, что если на одном из просмотров не было перестановок, то их не будет и потом – данные уже отсортированы, процесс сортировки следует закончить. Такой подход дает значительную экономию времени при работе с большими "почти отсортироваными" массивами. Примером такого массива может быть упорядоченный по алфавиту список сотрудников фирмы, на которую время от времени принимают новых работников.

Приведем алгоритм для этого метода. 

В этом алгоритме используется переменная "Перестановка_есть", которой сначала присваивается значение "истина", а как только в одном из просмотров не будет перестановок – ей присвоится значение "ложь". Это позволит прервать выполнение цикла "Пока".

1. Ввести массив.

2. Перестановка_есть = истина.

3. Номер_просмотра = 1.

4. Пока перестановка_есть = истина выполнить

сортировку массива.

5. Вывести результат.

6. Закончить.

Уточним отдельные пункты этого алгоритма. 

1. Ввести массив.

2. Перестановка_есть = истина 
3. Номер_просмотра (к)=1

4. Пока перестановка_есть = истина выполнить

4.1. Количество_перестановок = 0

4.2. Для i от 1 до n-k выполнить

Если x[i] > x[i+1], то

4.2.1. поменять x[i] и x[i+1]

4.2.2. количество_перестановок = количество_перестановок + 1

4.3. Если количество_перестановок = 0, то

перестановка_есть =ложь.

4.4. к=к+1

5. Для i от 1 до n выполнить

Вывести x[i].

6. Закончить.

Программа для этого алгоритма будет иметь вид

Program SortUsk;

Const

Nmax = 100;

Var

X : Array [1..Nmax] Of Real;

A : Real;

P : Boolean;

L, K, I, N : Integer;

Begin

Writeln('Введите количество чисел');

Readln(n);

Writeln('Введите массив чисел');

For i := 1 To n Do

Read(X[i]);

P := True;
{Перестановка есть}

K := 1;

{Номер просмотра}

While P Do

Begin

L:=0;

{Кол. перестановок}

For i := 1 To n-k Do

If X[i] > X[i+1] Then

Begin

A := X[i];

X[i]:=X[i+1];

X[i+1]:=A;

L:=L+1

End;

If L=0 Then

P:=False;

k:=k+1;

End;

Writeln('Отсортированный массив чисел');

For i := 1 To n Do

Write(X[i]);

End.

14. Обработка многомерных массивов

Многомерные массивы имеют два и более индексов. Для их обработки организуют несколько циклов, причем один вкладывается в другой. Такие циклы называются вложенными или циклами в цикле. При этом параметром в каждом цикле будет соответствующий индекс элемента массива. Для составления алгоритма и программы необходимо определить, как они меняются.

Пример 1. Ввести матрицу D размерностью m*n (n<=8, m<=5). Подсчитать количество отрицательных чисел в ней и заменить эти числа нулями. Вывести исходную и новую матрицу.

Алгоритм решения этой задачи может быть таким.

1. Ввести матрицу.

2. Вывести матрицу по строкам

3. Подсчитать количество нулей и заменить.

4. Вывести матрицу.

5. Закончить.

Для уточнения распишем матрицу

Нужно ввести и проанализировать все элементы матрицы, т.е. оба индекса должны принять все свои допустимые значения:

номер строки (обозначим i),

изменяется от 1 до m;

номер столбца (обозначим j),

изменяется от 1 до n
Уточненный алгоритм может быть таким.

1.1. Ввести количество строк (m) и столбцов (n)

1.2. Для i от 1 до m выполнить

Для j от 1 до n выполнить

ввести Dij
2. Для i от 1 до m выполнить

2.1. Для j от 1 до n выполнить

вывести Dij
2.2. Перейти на новую строку

3.1.Положить количество отрицательных чисел = 0

3.2. Для i от 1 до m выполнить

Для j от 1 до n выполнить

Если Dij<0, то

а)кол. отр.=кол. отр.+1

б)Dij=0

Программа для этого алгоритма может быть такой.

Program Matr;

Const

Mmax = 8;

Nmax = 5;

Var

D : Array [1..Mmax, 1..Nmax] Of Real;

m, n, k, i, j : Integer;

Begin

Writeln('Введите количество строк и столбцов');

Readln(m,n);

Writeln('Вводите элементы матрицы по строкам');

For i := 1 To m Do

For j :=1 To n Do

Read (D[i,j]);

Writeln('Исходная матрица');

For i := 1 To m Do

Begin 

For j:=1 To n Do

Write(D[i,j]);

Writeln;       {Переход на новую строку}

End;

K := 0;



For i := 1 To m Do

For j := 1 To n Do

If D[I,J] <0 Then

Begin

K := K+1;

D[I,J] := 0

End;

Writeln('Результирующая матрица');

For i := 1 To m Do

Begin 

For j:=1 To n Do

Write(D[i,j]);

Writeln; {Переход на новую строку}

End;

Writeln('Количество отрицательных чисел в матрице - ',K);

End.

Пример 2. Ввести матрицу A размерностью m*n (n<=10, m<=8). Подсчитать суммы чисел по строкам и вывести исходную матрицу, а в конце каждой строки – сумму.

Для разработки алгоритма распишем матрицу и выясним, как меняются индексы обрабатываемых элементов и сколько будет сумм. Количество сумм равно числу строк (m). Введем обозначения: i-номер строки, j - номер столбца.

Алгоритм решения задачи может быть таким.

1.1. Ввести количество строк (m) и столбцов (n)

1.2. Для i от 1 до m выполнить

для j от 1 до n выполнить 

ввести aij
2. Для i от 1 до m выполнить                                  { Для каждой строки }

2.1. Положить суммаi = 0                                   { Сумма в i-той строке }

2.2. Для j от 1 до n выполнить                             { Для каждого столбца }

а) вычислить суммаi = суммаi+aij  {Накопление сумма в i-ой строке }

б) вывести aij
2.3. Вывести суммаi
3. Закончить

Программа, реализующая этот алгоритм, будет иметь вид

Program SumMatr;

Const

Mmax = 8;

Nmax = 10;

Var

A : Array [1..Mmax, 1..Nmax] Of Real;

m, n, i, j : Integer;

S : Array[1..Mmax] Of Real;

Begin

Writeln('Введите количество строк и столбцов');

Readln(m,n
Writeln('Введите матрицу');

For i := 1 To m Do

For j := 1 To n Do

Read(A[i,j]);

For i := 1 To m Do 

Begin

S[i] := 0; 

For j := 1 To n Do

Begin

S[i]:=S[i]+A[i,j];

Write(A[i,j]:8:3);

End;

Writeln(S[I]:10:3);

End;

End.

В некоторых случаях при обработке матриц достаточно организовать только один цикл. Это, как правило, задачи, в которых обрабатываются диагонали матриц.

Пример 3. Ввести матрицу В размерностью n*n (n<10). Вывести элементы ее главной и побочной диагоналей.

Запишем матрицу и обратим внимание на изменение индексов элементов, расположенных на главной и побочной диагоналях.

1. Элементы главной диагонали имеют одинаковые индексы. Поэтому обращение к элементу, стоящему на этой диагонали можно записать так: Bii.
2. Как видно из выписанных ниже элементов побочной диагонали, сумма индексов для любого из них равна n+1. Поэтому при известном первом индексе i второй может быть вычислен как  n+1-i. Другими словами, обращение к элементу побочной диагонали можно записать так: Bi,n+1-i.

Учитывая сказанное, можно записать программу для примера 3 в таком виде.

Program Diag;

Const

Nmax = 10;

Var

B : Array [1..Nmax, 1..Nmax] Of Real;

n, i, j : Integer;

Begin

Writeln('Введите количество строк и столбцов');

Readln(N);

Writeln ('Введите матрицу');

For i := 1 To n Do

For j := 1 To N Do

Read(B[I,J]);

Writeln('Главная диагональ');

For i := 1 To n Do

Write(B[i,i]:8:4);

Writeln;

Writeln('Побочная диагональ');

For i := 1 To n Do

Write(B[i,n+1-i]:8:4);

Writeln;

End.

15. Программы обработки строк символов (текстов)

Тексты (символьная информация) – это анкеты, справочники, библиотеки и т.д. К задачам их обработки относятся задачи редактирования, разработки трансляторов с языков программирования и др. Они сводятся к поиску некоторых символов или строк в соответствующих текстах, подсчету количества их вхождений и корректировке некоторых мест текста.

Для работы с символьной информацией в стандартном Паскале используются уже известные нам данные символьного типа: Char. Причем одна переменная такого типа представляет собой всего один символ. В Турбо Паскале, как отмечалось ранее, имеются данные строкового типа (String).

Символы в ЭВМ, как известно, представляются двоичными кодами длиной в 1 байт. Поэтому отношение вида Sim1 < Sim2 имеет смысл. Причем оно соответствует  отношению Ord(Sim1) < Ord(Sim2).
15.1. Простейшие алгоритмы и программы обработки строк

В Турбо Паскале, как отмечалось выше, определены данные строкового типа - String. 

Пример определения переменных строкового типа:

Var

str : String;

line : String[80];

После слова String может указываться длина строки (максимальная). Если длина не указана, то считается, что она переменная и определяется числом символов, записанных на место переменной типа String при вводе или присваивании. Это число хранится в нулевом байте соответствующей переменной. 

Важно, что в переменной строкового типа доступен каждый отдельный символ. Обращение к нему производится путем указания его номера (индекса). Конечно, индекс символа не должен быть больше длины строки.

Для работы со строками в Паскале имеется несколько процедур и функций. Наиболее распространенные из них приведены ниже.

1. Операция «Объединение строк» изображается символом + .

Пример. Пусть переменные s1, s2 и s3 имеют тип String. Тогда в результате выполнения операции

s3 := s1 + s2;

получим строку s3, являющуюся объединением строк s1 и s2.

Если, например, s1 имела значение 'СамГТУ,', а s2 – значение ' ФАИТ', то s3 получит значение 'СамГТУ, ФАИТ'.

2. Функция Copy (исходная_строка, начальный_номер, длина) – выделение подстроки заданной "длины" из "исходной строки", начиная с символа, определенного "начальным_номером".

Если параметр "длина" больше остатка строки, то выделяется остаток;, если параметр "начальный_номер" превышает длину строки, то результатом будет пустая строка.

Пример.

. . . . . .

s1 := 'СамГТУ';

s2 := Copy(s1,4,3);{Результат: s2 получит значение 'ГТУ'}

3. Процедура Delete(исх_строка, номер, длина) – удаляет из "исходной строки" подстроку заданной "длины", начиная с заданного "номера":

Пример
. . . . .

S := 'СамГТУ, ФАИТ';

Delete(S,4,5);{Результат: S получит новое значение 'СамФАИТ'}

. . . . .

4. Функция Length(строка) – возвращает длину строки (количество символов в ней).

5. Функция POS(подстрока, строка) – возвращает номер символа, начиная с которого "подстрока" встречается в "строке" (если не встречается, то функция возвращает ноль). Прописные и строчные буквы считаются различными.

Пример
. . . .

N := Pos('ра', 'Абракадабра');  {N получит значение 3}

M := Pos('Ра', 'Абракадабра');  {M получит значение 0}

Рассмотрим примеры простейших программ обработки строк символов на примере одной строки. Тексты большей длины могут быть представлены в виде массива строк, каждая из которых обрабатывается аналогично.

Пример 1. Ввести в ЭВМ предложение длиной не более 80 символов и подсчитать количество слов в нем и количество букв "а". Будем считать, что одно слово от другого отделяется одним пробелом.

Алгоритм

1. Количество_слов =0; Количество_а=0.

2.1. Ввести предложение.

2.2. n=Длина (предложения).

2.3. Для номера символа (i) от 1 до n выполнить

2.3.1. Если символ =' ' (пробел), то

Количество_слов = Количество_слов +1

Иначе

Если символ ='а', то

Количество_а = Количество_а +1

3. Вывести Количество_слов и Количество_а.

4. Закончить.

Программа для этого алгоритма будет иметь вид

Program Predl;

Var

Pr : String;

i, n, Ks, Ka : Integer;

Begin

Ks := 0;
{п.1}

Ka := 0;

Write('Введите предложение'); { п.2.1 }

Readln(Pr);

n := Length(Pr); { п.2.2 } {Длина предложения}

{ п.2.3 }

For i := 1 to  n do

If Pr[i] = ' ' Then

Ks:= Ks + 1

Else

If Pr[i] = 'a' Then

Ka := Ka + 1;

If Pr[n]<> ' ' Then

Ks:=Ks+1;
{если в конце предложения нет пробела}

{ п.3 }

Writeln('Количество слов в предложении равно ',Ks, 

'количество букв а - ',Ka);

End.

15.2. Анализ символов в строке

Пример 1. Ввести в ЭВМ строку символов. Количество символов в строке (n <40). Вывести встречающиеся в ней символы без повторов и количество этих символов.

Решение задачи сводится к следующему. Каждый символ строки сравнивается со всеми остальными, находящимися после него. Если данный элемент имеет номер i, то следующий – i+1 и т.д., до конца. При этом i изменяется от 1 до n-1. Если в результате сравнения окажется, что анализируемый символ встречается в правой части строки, то он не уникальный. Пока его можно не выводить, он будет выведен потом. Если i-тый элемент не равен ни одному из последующих, то он уникальный и его нужно вывести.

Например для строки 'АВВСА' уникальные символы будут выведены в следующем порядке: ВСА

Этот алгоритм можно представить в следующем виде.

1. Ввести строку символов.

2. n = Длина(строки).

3. Положить Количество_символов =0

4. Для i от 1 до n-1 выполнить

4.1. Положить количество совпадений (k)=0.

4.2. Для j от i+1 до n выполнить

Если символi=символj, то

k=k+1.

4.3. Если k=0, то

а) Количество_символов = Количество_символов +1 и

б) Вывести символi.

5. Вывести Количество_символов.

6. Закончить.

Программа, записанная в соответствии с этим алгоритмом, приведена ниже.

Program Unicum;

var

s : string;

n, i, j, k, kols : integer;

begin

writeln('Введите строку'); {п.1}

readln(s);

n:=length(s);              {п.2}

kols:=0;                   {п.3}

writeln('Символы строки без повторов ');

for i:=1 to n-1 do          {начало п.4}

begin

k:=0; 



   {п.4.1}

for j:=i+1 to n  do    {п.4.2}

if s[i]=s[j] then

k:=k+1;

if k=0 then              {п.4.3}

      begin

   kols:=kols+1;

   write(s[i]:2);

   end;

end;

writeln('Количество различных символов в строке - ',kols);

writeln('Работа окончена. Нажмите клавишу ENTER');

readln;

end.

Пример 2. Сформировать массив из символов, встречающихся в тексте (вариант задачи примера 1).

Алгоритм

1. Ввести исходную строку.

2. Колич_нов_символов {символов в новой строке} =0.

3. Для номера символов в исходной строке (i) от 1 до n выполнить

Проверить, есть ли этот символ в новой строке и

если его нет, то записать в нее.

4. Вывести новую строку.

5. Закончить.

Уточнение п.3

3.1. Признак {Факт совпад. символа из новой строки с симв. исх. строки}=0

3.2. Для номера нового символа(j) от 1 до Колич._нов_символов. выполнить

Если нов.сим.j=ст.сим.i, то признак =j

3.3.Eсли признак = 0, то

а) Колич_нов_символов = Колич_нов_символов +1;

б) нов.cим.[Колич_нов_символов] = ст.сим.[i]

Обозначим:

Старый сим.{символ из исходной строки}
=>
s
Колич_нов_символов
=>
n
Признак

=>
jn
Новый сим.
=>
ns
Колич_нов_символов
=>
ks
Программа для этого алгоритма будет иметь вид

Program Psk;

Var

 s : String;

ns : String;

i, j, ks, jn : Integer;

Begin
writeln('Введите строку');

readln(s);                    { п.1 }

n:=length(s);

ks:=0;                        { п.2 }

For i := 1 to n do            { п.3 }

begin

jn := 0;                 { п.3.1 }

for  j := 1 to ks  do    {п.3.2}

If ns[j]=s[i]  then

jn:=J;

If  jn=0 then            {п.3.3}

begin

ks := ks+1;

ks[ks] := s[i];

end;

end;

Writeln('Символы, входящие в исходную строку'); 

For i := 1 to ks do

Write(ns[i]);

End.

Пример 3. (продолжение примера 2). Подсчитать, сколько раз каждый символ встречается в исходном тексте.

В алгоритм примера 2 вместо пункта 5 (Закончить) вставим новый:

5. Подсчет количества вхождений каждого символа в исходную строку.

Чтобы сделать это, будем каждый символ новой строки сравнивать с символами исходной строки и подсчитывать количество совпадений.

Уточняем алгоритм

5.1. Для j от 1 до Колич_нов_символов выполнить

5.2. Кол.=0

5.3. Для i от 1 до n выполнить

Если нов.сим.j=ст.сим.i, то

кол.=кол+1

5.4. Вывести кол.

Для рассматриваемого случая полный текст программы приводить не будем. Фрагмент, помешенный ниже, должен быть вставлен в программу для примера 3 перед последним оператором (End.)

{Подсчет количества вхождений символов}

For j := 1 to ks do             {п. 5.1 }

begin

kol := 0;                  {п. 5.2}

For i := 1 to n do       {п. 5.3}

If ns[j] = s[i] then

kol := kol + 1;

                        {п. 5.4}

writeln('Количество букв : ',ns[j],' равно ',kol);

end;

Пример 4. Ввести в ЭВМ строку из n символов латинского алфавита (n<30). Расположить их по алфавиту.

Воспользуемся тем, что буквы латинского алфавита 'A' – 'Z', 'a' –'z' имеют коды, значения которых возрастают. При этом для решения задачи можно применять алгоритм сортировки "пузырьком".

Program Sorts;

var

s : string;

sr : char;

n, i, k : integer;

Begin
writeln('Введите строку');

readln(s);

n := length(s);

for k := 1 to n-1 do

for i := 1 to n-k do

if s[i]>s[i+1]then

begin

sr :=s[i];

s[i] := s[i+1];

s[i+1] := sr;

end;

writeln('Строка с упорядоченными символами:');

writeln(s);

End.

Пример 5. Ввести в ЭВМ предложение. Подсчитать количество гласных в нем. В более общей постановке задачи необходимо определить, сколько раз встречаются в предложении символы из некоторого эталонного набора (согласные, звонкие согласные, цифры и т.д.).

При разработке алгоритма решения этой задачи учтем, что гласные (вообще символы из произвольного набора) располагаются в разных местах кодировочной таблицы. Для подсчета их количества в программе необходимо задать строку гласных (эталонную строку), а затем определить, сколько раз символы эталонной строки встречаются в предложении.

Можно предложить следующий алгоритм решения такой задачи.

1. Задать эталонную строку.

2. Ввести предложение.

3. Подсчитать количество символов эталонной строки в предложении
               и вывести его.

4. Закончить.

Убедившись в правильности шагов этого алгоритма, уточним его и получим.

1.1 Задать Эталонную_строку.

1.2. m = Длина(Эталонной_строки).

2.1. Ввести Предложение.

2.2. n = Длина(Предложения).

3.1. Количество_символов = 0.

3.2. Для номера символа Предложения (i) от 1 до n выполнить

3.2.1. Для номера символа Эталонной_строки (j) от 1 до m выполнить

Если Предложениеi = Эталонная_Строкаj то

Количество_символов = Количество_символов + 1.

3.3. Вывести Количество_символов.

4. Закончить.

Программа для этого алгоритма приведена ниже.

Program Count_Gl;

Var

Predl,Gl:String; { Предложение и эталонная строка }

n,m,i,j,kol:Integer; { Вспомогательные переменные }

Begin
Gl:=’аеёиоуыэюяАЕЁИОУЫЭЮЯ’; {Строка гласных (эталонов)}

m:=Length(Gl);

WriteLn('Введите предложение');

ReadLn(Predl);

n := Length(Predl);

kol := 0;

For i:=1 to n do

For j:=1 to m do

If Predl[i] = Gl[j] then

kol:=kol + 1;

Writeln('Количество гласных в предложении = ',kol);

Writeln('Работа окончена. Нажмите клавишу ENTER');

Readln;

End.

Пример 6 (вариант примера 5). Ввести в ЭВМ предложение. Найти и вывести на экран наиболее редко встречающуюся в нем гласную.

В этой задаче необходимо использовать несколько (массив) счетчиков: столько, сколько символов в строке эталонов. В начале подсчитывается, сколько раз каждый эталонный символ встречается в предложении, а затем применяется известный алгоритм нахождения минимума. При этом необходимо учесть, что тривиальное значение минимума, равное нулю, не может быть решением, так как такой символ вообще отсутствует в предложении.

Составим алгоритм.

1. Задать эталонную строку.

2. Ввести предложение.

3. Подсчитать, сколько раз каждый символ эталонной строки встречается в
             предложении.

4. Найти наиболее редкий символ и вывести его

5. Закончить.

Убедившись в правильности шагов этого алгоритма, уточним его.

Уточненный алгоритм

1.1 Задать Эталонную_строку.

1.2 m = Длина(Эталонной_строки).

2.1. Ввести Предложение.

2.2. n = Длина(Предложения).

3.1. Для номера символа Эталонной_строки (j) от 1 до m выполнить 

3.1.1. Количество_символовj = 0.

3.1.2. Для номера символа Предложения (i) от 1 до n выполнить

Если Предложениеi = Эталонная_Строкаj то

Количество_символовj = Количество_символовj + 1.

4.1. Min = MaxInt (Максимальное целое).

4.2. Номер_Min = 0. ( Минимум не найден).

4.3. Для номера символа Эталонной_строки (j) от 1 до m выполнить 

4.3.1. Если Количество_символовj > 0 ( символ встречается в предложении), то

а) Min = Количество_символовj;

б) Номер_Min = j.

4.4. Вывести Эталонная_Строка Номер_Min и Min.

5. Закончить.

Поиск минимума в данном случае выполняется в условиях некоторой неопределенности, так как неизвестно, какие из массива найденных количеств будут иметь ненулевые значения. Чтобы избежать присваивания переменной Min нуля, на ее место перед началом поиска минимума записывается абсурдно большое число - MaxInt (Максимальное целое), которое затем будет заменено первой же ненулевой величиной.

Программа для приведенного алгоритма будет иметь вид

Program Min_Gl;

Const
Nmax= 20; { Максимальное количество счетчиков гласных }

Var

Predl,Gl:String; { Предложение и эталонная строка }

Kol:Array[1..Nmax] Of Integer; { Счетчики гласных }

n,m,i,j,Min,Nom:Integer; { Вспомогательные переменные }

Begin
Gl:=’аеёиоуыэюяАЕЁИОУЫЭЮЯ’; {Строка гласных (эталонов)}

m:=Length(Gl);

WriteLn('Введите предложение');

ReadLn(Predl);

n := Length(Predl);

For j:=1 to m do

Begin

Kol[j] := 0;

For i:=1 to n do

If Predl[i] = Gl[j] then

Kol[j]:=kol[j] + 1;

End;

Min:=MaxInt;

Nom:=0;

For j:=1 to m do

If kol[j]>0 then

If kol[j]< Min then

Begin

Min:= kol[j];

Nom:=j;

End;

Writeln('Наиболее редкая гласная в предложении -', Gl[Nom]);

Writeln('Она встречается ',Min,' раз');

Writeln('Работа окончена. Нажмите клавишу ENTER');

Readln;

End.

16. Типовые программы обработки строк

В Турбо Паскале, как отмечалось выше, используются данные строкового типа – string. Операции, процедуры и функции, применяемые для этих типов, уже рассмотрены.

Пример 1. Ввести в ЭВМ массив из n (n<=50) слов на английском языке. Расположить их по алфавиту.

Латинские буквы любой ЭВМ, как известно, представляются кодами, которые упорядочены по алфавиту (от "a" до "z"). Слово – это строка, которую можно считать длинным двоичным кодом. Для упорядочения таких кодов можно использовать методы упорядочения чисел, например, метод "пузырька", который был изучен ранее. 

Программа будет иметь вид

Program SortW;

  Const

      Nmax = 20;
{Количество слов}

      m = 10;

{Максимальная длина слова}

  Var

    W: Array[1..Nmax] of string[m];{Массив для хранения слов}

    Wr: String[m]; {Промежуточная переменная для перестановок}

    n, i, k : integer;

Begin
  WriteLn('Введите количество слов');

  Readln(n);

  WriteLn('Введите слова');

  for i := 1 to n do

    ReadLn(w[i]);

    for k := 1 to n-1 do

      for i := 1 to n-k do

        if w[i] > w[i+1] then

          begin

            Wr := w[i];

            w[i] := w[i+1];

            w[i+1] := Wr;

          end;

    WriteLn(Слова, упорядоченные по алфавиту:');

    for i:= 1 to n do

      WriteLn(w[i]);

End.

В таблицах кодировки символов, содержащих буквы кириллицы, коды букв кириллицы также упорядочены по алфавиту. Студентам предоставляется возможность проверить самостоятельно работу алгоритма на русских словах.

Наиболее типичными задачами обработки текстов являются задачи синтаксического анализа, которые сводятся к выделению слов из предложений и анализу этих слов. В частности, такие задачи решаются при трансляции и компиляции программ с языков высокого уровня.

Основными особенностями структур данных при этом являются следующие.

1. Строка текста или одно слово – одномерный массив символов, а несколько слов – это массив строк (двумерный массив символов).

2. Длина слов в тексте, как правило, неодинакова. Обычно для их разделения используются пробелы. Найти пробел в предложении можно, просматривая последовательно каждый его символ (так как пробел –  это тоже символ). Если очередной пробел найден, то предыдущее слово закончилось.

Формирование отдельных слов может быть осуществлено посимвольно, путем объединения слова (строки) с очередным символом, отличным от пробела. Ясно, что вначале формируемое слово должно быть пустым.

Второй метод – запоминание положения первой и последней букв очередного слова в предложении и использование функции Copy для его выделения.

Возможны 2 способа представления текстов:

1) слова разделены всегда одним пробелом;

2) слова разделены одним или несколькими пробелами (общий случай).

Методы выделения слов при этом несколько отличаются.

16.1. Выделение слов из текста
(слова разделены одним пробелом)
Условие. Ввести в ЭВМ предложение, в котором слова разделены одним пробелом. Выделить слова, переписать их в массив слов и вывести на экран.

В самом общем виде алгоритм решения задачи может быть таким.

1. Ввести предложение

2. Выделить слова

3. Вывести слова

4. Закончить.

Уточняем пункты алгоритма.

1.1. Ввести предложение

1.2. n = Длина (предложения)

2.1. k = 1; {Номер слова}

2.2. Слово[k] = '';  { пустое слово }

2.3. Для номера буквы в предложении (i) от 1 до n выполнить

2.3.1. Если предл[i] <> ' ', то  {очередной символ – не пробел}

Слово[k] = Слово[k] + предл[i]

Иначе {т.е. предл[i]= пробелу; значит, конец предыдущего  слова}

а) k = k+1

б) Слово[k] =''  { пустое слово }

3.1. Для номера слова (i) от 1 до k  выполнить

3.1.1. Вывести слово[i] 

4. Закончить.

Приводимая ниже программа реализует рассмотренный алгоритм.

Program GetSl1;

CONST

  m = 50;
{Максимальное количество слов в предложении}

VAR

   Predl : string;                  {Предложение}

   Slovo : array[1..m] of string;   {Массив для  слов}

   n, k, i: integer;       {Вспомогательные переменные}

BEGIN

WriteLn('Вводите предложение:');

ReadLn(Predl);

n := Length(Predl); {п. 1.2.}

k := 1;

Slovo[k] := ''; {Пустое слово – 2 апострофа подряд}

{п. 2.3.}

For i := 1 to n do

If predl[i] <> ' ' then

Slovo[k] := Slovo[k] + Predl[i]

Else

begin

k := k + 1;

Slovo[k] := '';

end;

{п. 3.1.}

WriteLn('Слова из предложения: ');

For i := 1 to k do

WriteLn(Slovo[i]);

END.

16.2. Выделение слов из текста 
(слова разделены несколькими пробелами)

Условие. Ввести в ЭВМ предложение, в котором слова могут быть разделены одним или несколькими пробелами. Выделить слова и вывести их на экран.

В этом случае задача осложняется тем, что номер слова необходимо менять только при встрече с первым пробелом, а все последующие пробелы – просто пропускать. Алгоритм будет отличаться от предыдущего в основном в части, связанной с обработкой пробелов.

Алгоритм для этой задачи

1.1. Ввести предложение.

1.2. n := Длина (предложения).

2.1. Номер слова (k) := 1.

2.2. Слово[k] := ''.

2.3. Номер символа (i) := 1.

2.4. Пока i <= n  повторять

2.4.1.
Если предл[i] <> ' ', ТО

а) Слово[k] := Слово[k] + предл[i];

б) i := i + 1

Иначе

а) k := k + 1;

б) Слово[k] := ''

в) Пока (предл[i]=' ') И (i<=n) выполнить

i = i + 1;

3.1. Для i от 1 до k выполнить

вывести слово[i];

4. Закончить.

Приводимая ниже программа реализует рассмотренный алгоритм.

Program GetSl2;

const

m = 50;

Var

Predl : string;

Slovo : array [1..m] of string;

n,k,i : integer;

Begin

n := Length(Predl);

k := 1;

Slovo[k] := '';

i := 1;

While i<=n do

If predl[i] <> ' ' then   {символ – не пробел}

begin

Slovo[k] := slovo[k] + predl[i];

i := i + 1;

end

Еlse                       {символ – пробел}

begin

k := k + 1;

slovo[k] := '';

While (predl[i]=' ') and (i <= n) do

                 i := i + 1;

end;

WriteLn('Слова из предложения:');

For i := 1 to k do

WriteLn(slovo[i]);

END.

16.3. Некоторые типовые алгоритмы и программы
обработки массивов строк (слов из текстов)

Рассмотрим примеры некоторых наиболее часто встречающихся задач обработки слов из текстов. В таких задачах вначале выделяются слова из текста с использованием алгоритмов, описанных выше. Затем полученные массивы слов (строк) обрабатываются в соответствии с условиями задачи.

Задачи типа 1 связаны с определением длины слов.

Пример 1.1. Ввести в ЭВМ предложение, в котором слова могут быть разделены несколькими пробелами. Выделить слова и вывести их на экран. Затем подсчитать количество слов с четным числом букв и вывести эти слова и их количество.

Для решения задачи можно предложить следующий алгоритм.

1. Ввести предложение

2. Выделить слова

3. Вывести слова

4. Подсчитать и вывести количество слов с четным числом букв

5. Закончить.

Уточняем пункты алгоритма.

1.1. Ввести предложение

1.2. n := Длина (предложения)

2.1. Номер слова (k) := 1

2.2. Слово[k] := ''

2.3. Номер символа (i) := 1

2.4. Пока i <= n  повторять

2.4.1.
Если предл[i] <> ' ', то

а) Слово[k] := Слово[k] + предл[i];

б) i := i + 1

Иначе

а) k := k + 1;

б) Слово[k] := ''

в) Пока (предл[i]=' ') И (i<=n) выполнить

i = i + 1;

3.1. Для i от 1 до k выполнить

вывести слово[i]

4.1.Количество_слов = 0.

4.2. Для i от 1 до k выполнять

Если Длина(слово[i]) mod 2 = 0 то

а) вывести слово[i];

б) Количество_слов = Количество_слов +1.

4.3. Вывести Количество_слов.

5. Закончить.

Приводимая ниже программа реализует рассмотренный алгоритм.

Program Not_Odd_Sl;

const

m = 50;

Var

Predl : string;

Slovo : array [1..m] of string;

n,k,i,kol : integer;

BEGIN

{ Выделение слов из предложения – п.п. 1 -3 }

n := Length(Predl);

k := 1;

Slovo[k] := '';

i := 1;

While i<=n do

If predl[i] <> ' ' then   {символ – не пробел}

begin

Slovo[k] := slovo[k] + predl[i];

i := i + 1;

end

Еlse                       {символ – пробел}

begin

k := k + 1;

slovo[k] := '';

While (predl[i]=' ') and (i <= n) do

                 i := i + 1;

end;

WriteLn('Слова из предложения');

For i := 1 to k do

WriteLn(slovo[i]);

Kol:=0;

WriteLn('Слова с четным количеством букв');

For i := 1 to k do

If Length(slovo[i]) mod 2 =0 then

Begin

WriteLn(slovo[i]);

Kol:= kol+1;

End;

WriteLn('Количество слов с четным числом букв=', kol);

Writeln('Работа окончена. Нажмите клавишу ENTER');

Readln;

END.

Пример 1.2. Ввести в ЭВМ предложение. Упорядочить его слова по возрастанию длины и вывести их в столбик.

Для решения этой задачи воспользуемся методом сортировки «пузырьком», который применим к длинам слов.

Алгоритм будет следующий.

1. Ввести предложение.

2. Выделить слова.

3. Вывести слова.

4. Упорядочить слова по возрастанию длины.

5. Вывести упорядоченные слова.

6. Закончить.

Уточним пункты 4 и 5 и получим алгоритм.

4. Для j от 1 до k-1 выполнить

4.1. Для i от 1 до k-j выполнять

Если слово[i] > слово[i+1] то

Поменять их местами.

5. Для i от 1 до k выполнять

вывести слово[i].

Приведем фрагмент программы, реализующий эти пункты.

{Упорядочение слов}

For j := 1 to k-1 do

For i := 1 to k-j do

If Length(slovo[i])> Length(slovo[i+1]) then

Begin

RSl:= slovo[i];

slovo[i]:= slovo[i+1];

slovo[i+1]:=Rsl;

End;

WriteLn('Упорядоченные слова из предложения');

For i := 1 to k do

WriteLn(slovo[i]);

Задачи типа 2 связаны с определением окончаний слов.

Пример 2.1. Ввести в ЭВМ строку, содержащую названия единиц физических величин. Вывести названия единиц длины и подсчитать их количество. При разработке алгоритма учтем, что искомые названия имеют окончание «метр»: сантимерт, километр, дециметр и т.д.

Алгоритм может быть следующий.
1. Ввести строку.

2. Выделить из нее слова. Общее количество слов = k.
3.1. Количество_длин = 0.

3.2. Для i от 1 до k выполнить
3.2.1. n = Длина(слово[i]).

3.2.2. Если Подстрока(из слово[i] от номера n –3 длиной 4) = «метр» то

а) вывести слово[i];

б) Количество_длин = Количество_длин +1.

3.3. Вывести Количество_длин.

4. Закончить.

Приведем фрагмент программы, реализующий пункт 3.

kol:=0;

WriteLn('Названия единиц длины');

For i := 1 to k do

Begin

n:=Length(slovo[i]);

If Copy(Slovo[i], n-3,4) = ‘метр’ then

Begin

WriteLn(slovo[i]);

Kol:= kol+1;

End;

End;

WriteLn('Количество названий единиц длины=', kol);

Задачи типа 3 связаны с анализом первых букв слов.

Пример 3.1. Ввести в ЭВМ строку, содержащую слова и числа в символьном виде. Вывести входящие в нее числа и их количество. При разработке алгоритма будем считать, что слово из строки, начинающееся с цифры, является числом. Кроме того, воспользуемся тем, что коды цифровых символов упорядочены от ‘0’ до ‘9’.

Алгоритм для решения этой задачи может быть следующий.

1. Ввести строку

2. Выделить из нее слова. Общее количество слов = k
3.1. Количество_чисел = 0.

3.2. Для i от 1 до k выполнять

Если (слово[i,1] >=’0’) И (слово[i,1] <=’9’) то

а) вывести слово[i];

б) Количество_чисел = Количество_ чисел +1.

3.3. Вывести Количество_чисел.

4. Закончить.

Приведем фрагмент программы, реализующий пункт 3.

{Нахождение чисел}

kol:=0;

WriteLn('Числа из строки');

For i := 1 to k do

If (Slovo[i,1]>=’0’) And (Slovo[i,1]<=’9’) then

Begin

WriteLn(slovo[i]);

Kol:= kol+1;

End;

WriteLn('Количество чисел=', kol);

17. Алгоритмы поиска

Процедуры поиска широко используются в информационно-справочных системах (базах и банках данных), при разработке синтаксических анализаторов и компиляторов (поиск служебных слов, команд и т.п. в таблицах). В процессе поиска всегда используется таблица неповторяющихся эталонов. Искомый элемент сравнивается с каждым из этих эталонов. Результат поиска может быть булевский (элемент есть в таблице или нет) или числовой (номер эталона в таблице).

Рассмотрим алгоритм поиска на примере программы – словаря. Пусть в ЭВМ введена таблица-словарь из n русских и английских слов вида табл. 5.

Словарь 


Таблица 5
Русское слово
Английское слово

программа

оператор

метка
. . . 
program

statement

lable

. . .

Размер таблицы фиксирован и известен.

Таблица должна храниться в виде двух массивов слов, причем русскому слову с номером 3 ("метка") соответствует английское с таким же номером ("lable") и т.д.

Далее при работе программы должны вводиться английские слова, а программа должна выводить на экран их русские эквиваленты. Если слова в словаре нет, то выводится "не знаю". Перевод введенного слова на русский язык осуществляется путем его сравнения с каждым английским словом из таблицы. Если слово найдено, то перевод выполняется.

Простейший алгоритм поиска - линейный. При этом эталонный массив просматривается последовательно от первого до последнего элемента. Наиболее сложный случай, когда слова нет в словаре (не найдено). Суждение об этом можно сделать только по окончании просмотра всего массива.

17.1. Алгоритм линейного поиска

В соответствии с этим алгоритмом, как отмечалось выше, эталонный массив просматривается последовательно от первого до последнего элемента. Наиболее сложным является случай, когда слова нет в словаре (не найдено).

Общий алгоритм поиска можно описать так.

1. Ввести два исходных массива слов (таблицу-словарь).

2. Повторять

ввести новое слово и вывести русское

пока не надоест.

3. Закончить.

Пусть для окончания работы с программой (когда переводить надоело) необходимо ввести слово "End".

Уточненный алгоритм будет таким.

1. Ввести два исходных массива слов (таблицу-словарь).

2. Повторять

2.2.1. Ввести Слово (для перевода);

2.2.2. Номер Русского_Слова (Num):= 0;

2.2.3. Для номера Английского_Слова (i) от 1 до n выполнить

Если Слово = Английское_Слово[i] то

номер Русского_Слова (Num):= i;

2.2.4. Если номер Русского_Слова (Num) = 0 то

вывести: 'Введенного слова нет в словаре',

Иначе

вывести Русское_Слово[Num].

пока не будет Слово = 'End'.

3. Закончить.

Обозначим:

Rsl - массив русских слов в словаре

Asl - массив английских слов в словаре

Slovo - слово для перевода.

Программа линейного поиска будет иметь вид

 Program Poisk;

Const

n = 100;

dl = 15; { длина слова }

Var

       Rsl, Asl : Array[1..n] of string[dl]; 

            Slovo : String[dl]; 

           i, Num : Integer;

  BEGIN
Writeln('Введите словарь');

For i := 1 to n do

       Begin

     ReadLn(Rsl[i]);

     ReadLn(Asl[i]);

       End;

{   перевод  }

Repeat
WriteLn('Введите слово для перевода');

    ReadLn(Slovo);

    Num:=0;  { номер слова в словаре  }

    For i := 1 to n do

      If Slovo = Asl[i] then

            Num := i;

      If Num = 0 then

        Writeln('Нет такого слова в словаре')

           Else

         Writeln(Rsl[Num]);

Until Slovo='End';

Writeln('Работа окончена. Нажмите клавишу ENTER');

Readln;

  END.

Основной недостаток алгоритма линейного поиска – большое время. При использовании оператора For для поиска ВСЕГДА выполняется ровно n операций сравнения, не зависимо от того, найдено слово или нет. Поэтому программа, обнаружив слово в начале словаря, продолжает его просмотр до конца, т.е. выполняет бесполезную работу. Время поиска может быть существенно сокращено, если обеспечить его прекращение, когда слово найдено. При равномерном распределении элементов в таблице эталаонов (слов в словаре) среднее время поиска может стать пропорциональным величине n / 2.

Этого можно достичь, изменив пункт 2.2 в рассмотренном выше алгоритме следующим образом.

Ускорение линейного поиска

2.2. Повторять

2.2.1. Ввести Слово
2.2.2. Номер Русского_Слова := 0

2.2.3. Номер в таблице i := 1

2.2.4. Пока (Num=0) и (i<=n) выполнить

2.2.4.1. Если Слово = Английское_слово[i], То Num := i

2.2.4.2. i := i+1

2.2.5.Если Num = 0,  ТО вывести "не знаю",

 Иначе вывести русское слово[Num];

Соответствующий фрагмент программы приведен ниже.

{ Ускорение линейного поиска }

Repeat
WriteLn('Введите слово для перевода');

  ReadLn(Slovo);

Num := 0;

    i := 1;

    While (Num=0) And (i<=n) do

      Begin

        If Slovo = Asl[i] then

          Num := i;

          i:=i+1;

      End;

    If Num = 0 then

      Writeln('Нет такого слова в словаре')

    Else

      Writeln(Rsl[Num]);

  End;

Если при программировании используется Турбо Паскаль 7.0, то прервать выполнение цикла For после совпадения введенного слова со словом в словаре можно более изящно, чем в приведенном выше примере. Для этого можно использовать процедуру Break, которая специально предназначена для досрочного выхода из циклов For, While и Repeat.

Соответствующий фрагмент программы приведен ниже, где добавленные операторы выделены.

. . . . . . . . . .

For i := 1 to n do

      If Slovo = Asl[i] then

        Begin

          Num := i;

          Break;

        End; 

      If Num = 0 then

        Writeln('Нет такого слова в словаре')

           Else

         Writeln(Rsl[Num]);

End;

17.2. Алгоритм дихотомического поиска

Этот алгоритм является более быстрым, чем линейный, но применяется только к упорядоченным массивам. Среднее время дихотомического поиска пропорционально величине log2n, где n — количество элементов таблицы эталонов.

Метод основан на последовательном делении на 2 диапазона поиска. При этом на каждом шаге либо находится элемент, либо происходит переход в одну из половин диапазона. В процессе поиска выполняется не только сравнение на равенство, но и на больше-меньше. Последняя операция позволяет выбрать очередную половину диапазона таблицы. Если массив эталонов не упорядочен, то выбор будет сделан неверно, и результат можно не получить никогда (говорят, что алгоритм расходится).

Общий алгоритм поиска будет таким же, как в п. 18.1, т.е.

1. Ввести два исходных массива слов (таблицу-словарь).

2. Повторять

ввести новое слово и вывести русское

пока не надоест.

3. Закончить.

Для корректности работы алгоритма необходимо упорядочить словарь (массивы слов) по алфавиту. Эта операция выполняется применительно к основному массиву, используемому при поиске — к английским словам.

Уточненный алгоритм можно представить в следующем виде.

1.1. Ввести количество слов в словаре (n).

1.2. Для номера слова (i) от 1 до n выполнить

1.2.1. Ввести Английское_Слово[i].

1.2.2. Ввести Русское_Слово[i].

2. Упорядочить слова по алфивату:

Для номера просмотра (k) от 1 до n - 1 выполнить

Для номера слова (i) от 1 до n - k выполнить

Если Английское_Слово[i] > Английское_Слово[i+1] то

а) поменять местами Английское_Слово[i] и Английское_Слово[i+1];

б) поменять местами Русское _Слово[i] и Русское _Слово[i+1].

3. Повторять

3.2.1. Ввести Слово (для перевода);

3.2.2. Граница_левая (диапазона поиска):= 1;

3.2.3. Граница_правая (диапазона поиска):= n;

3.2.4. Признак:= "Не найдено".

3.2.5. Повторять

а) Номер Русского_Слова (k):= (Граница_левая + Граница_правая)/2;

б) Если Слово = Английское_Слово[i] ТО

Признак:= "Найдено"

Иначе

Если Слово > Английское_Слово[i] ТО

k:= Граница_левая

Иначе

k:= Граница_правая.

Пока не будет "Найдено" Или (Граница_левая = Граница_правая - 1).

3.2.6. Если "Найдено" 0 ТО

вывести Русское_Слово[k]

Иначе

вывести: 'Введенного слова нет в словаре'.

пока не будет Слово = 'End'.

4. Закончить.

Программа дихотомического поиска, соответствующая этому алгоритму, будет иметь вид.

 Program DixPoisk;

Const

Nmax = 100;

dl = 15; { длина слова }

Var

       Rsl, Asl : Array[1..Nmax] of string[dl]; 

          Slovo : String[dl]; 

        i, k, n: Integer;

Lev, Prav: Integer; { границы }

Naideno: Boolean; { признак }

  Begin

Writeln('Введите количество слов');

ReadLn(n);

Writeln('Введите словарь');

For i := 1 to n do

       Begin

     ReadLn(Rsl[i]);

     ReadLn(Asl[i]);

       End;

{  Упорядочение словаря по английскому алфавиту }

For k:=1 to n-1 do

   For i:=1 to n-k do

If Asl[i] > Asl[i+1] then

Begin

Slovo:= Asl[i];

Asl[i]:= Asl[i+1];

Asl[i]:= Slovo;

Slovo:= Rsl[i];

Rsl[i]:= Rsl[i+1];

Rsl[i]:= Slovo;

End;

{   Перевод  }

Repeat
WriteLn('Введите слово для перевода');

ReadLn(Slovo);

Lev:= 1;

Prav:= n;

Naideno:= False; { не найдено }

Repeat

k:=Round(Lev + Prav) / 2);

If Slovo = Asl[k] then

Naideno:= True { не найдено }

Else

If Slovo > Asl[k] then

k:=Lev

Else

k:= Prav;

Until Naideno Or (lev=Prav-1);

If Naideno then

 Writeln(Rsl[k])

Else

Writeln('Нет такого слова в словаре');

Until Slovo='End';

Writeln('Работа окончена. Нажмите клавишу ENTER');

Readln;

  END.

18.  ПРОЦЕДУРЫ, ФУНКЦИИ И МОДУЛИ В ПАСКАЛЕ

При программировании «сверху вниз» алгоритм записывается в виде последовательности шагов. В каждый момент времени человек может рассматривать небольшое число деталей, поэтому на верхнем уровне алгоритма число шагов обычно небольшое. При написании программы не следует скрывать такую структуру. В большинстве случаев полезно иметь в виду следующую рекомендацию: из чтения текста программы должно быть ясно, что делает программа, а не как она это делает. Лица, которых интересует ответ на второй вопрос, должны найти его в другом месте, просматривая другие объекты, но не основной блок программы.

Другими словами, чтобы программы было проще понимать и записывать, следует делать их более компактными. Для этого в языке программирования должны быть средства, позволяющие разбивать программу на логически связанные блоки, каждый из которых решает некоторую подзадачу. Кроме того, нередко бывает, что такого рода подзадачи должны выполняться в программе неоднократно.

В Турбо-Паскале такими средствами являются процедуры, функции и модули.

Процедуры и функции — это законченные программные единицы, которые решают некоторую подзадачу и, возможно, выполняются многократно.

Модули являются основой модульного программирования в Турбо-Паскале. Они используются для создания библиотек, которые могут включаться в различные программы (при этом совсем необязательно иметь в наличии исходный текст), а большие программы могут разделяться на логически связанные модули.

Достоинства модульного программирования:

1) упрощение процесса написания и отладки;

2) сокращение объема программ.

Недостаток — некоторое увеличение затрат времени и памяти на оформление модулей и обращение к ним.

18.1. Процедуры

Процедура — это независимый программный блок (программная единица), выполняющий решение некоторой самостоятельной подзадачи.

Процедуры (и функции) размещаются в программе сразу же после описания переменных (в разделе Var) и перед словом Begin, указывающим начало ее основной части, т.е. следующим образом:

Program

Const

Var

. . . . .

Процедура_1

. . . . .

Процедура_n

Begin  { Начало основной программы }

. . . . . .

End.

Сама процедура оформляется следующим образом:

Procedure Имя(Формальные_параметры); {заголовок процедуры}
Описание локальных переменных

Begin
Операторы
<= Тело процедуры

End;

Из этого текста видно, что процедура описывается так же, как программа. Отличие заключается в том, что текст программы заканчивается служебным словом End с точкой, а процедура заканчивается служебным словом End с точкой с запятой после него. Поэтому саму Паскаль-программу можно считать главной процедурой.

Формальные параметры – это список, который содержит имена переменных, массивов и типов, являющихся исходными данными и результатами процедуры. Он может отсутствовать. Элементы списка параметров описываются полностью, с указанием типов.

Локальные переменные — это переменные, которые используются только внутри данной процедуры.

На вход процедуры (в качестве исходных данных) поступает совокупность параметров – аргументов, а на выходе формируются параметры – результаты. Причем процедура позволяет получить один или несколько результатов, а функция — только один.

Пример. Процедура, выводящая горизонтальную линию из n одинаковых символов, может быть такой.

Procedure GorLine(n : Integer; Sim : char);

{Изображение горизонтальной линии символом Sim}

     Var

       I : Integer;

     Begin

       For i := 1 To n Do

         Write(Sim);

     End;

Здесь 
n и Sim – формальные параметры, являющиеся входными величинами;

i – локальная переменная.

Процедуры в программе выполняются не в том порядке, в каком они записаны. К ним обращаются (их вызывают) из главной программы по имени с указанием фактических параметров.

Например:

Имя; 
{ если процедура не имеет параметров } или

Имя (Фактические_параметры); {если процедура с параметрами}

Обращения к процедуре GorLine возможны в следующих формах:

GorLine(25);

GorLine(WhereX + 10);

После выполнения процедуры происходит возврат в основную программу к оператору, стоящему после вызывающего (см. рис. 2.9).

Пример. Пусть в массиве Dan имеются 20 значений целого типа (1≤ Dan[i] ≤75), которые нужно представить в виде гистограммы (графика, представляемого столбцами символов). Значения элементов массива можно получить с помощью генератора случайных чисел. Для простоты примем, что ось Y располагается вдоль экрана. 

При составлении программы воспользуемся ранее написанной процедурой GorLine. Столбцы графика будем изображать символом "*" (звездочка).

Program Graf1;

Const

  N = 20;

Var

Dan : Array[1..N] of Integer;

K,I : Integer;

Procedure GorLine(n : Integer; Sim : char);

Var

I : Integer;

Begin

For i := 1 To n Do

Write(Sim);

End;  { GorLine }

Begin        { Основная программа }

   Randomize;{Инициализация генератора случайных чисел}

For I :=1  To N Do        { Заполнение массива }

     Dan[I] := Random(75)+1; { случайными числами }

For I :=1  To N Do

Begin

K := Dan[i];

GorLine(K, ’*’);

WriteLn;

End;

WriteLn(’Нажмите ENTER’);

ReadLn;

End.

Здесь 
N — формальный параметр (в процедуре),

K — фактический (в основной программе).

Различие между формальными и фактическими параметрами такое же, как между символами в общей формуле для нахождения корней квадратного уравнения и значениями, которые подставляются в эту формулу при решении конкретного уравнения. Другими словами, формальные параметры – это просто объекты, которые позволяют описать, что будет делать процедура. Сами действия будут выполняться над теми значениями объектов, которые передаются процедуре при ее вызове. Поэтому при вызове процедуры должны соблюдаться следующие требования:

– количество передаваемых фактических параметров должно равняться количеству описанных в процедуре формальных параметров.

– порядок следования фактических параметров и их типы должны соответствовать порядку следования и типам формальных параметров. Это, в частности означает, что массивы должны иметь одинаковую размерность и содержать данные одного типа. 

Пример. Пусть в программе описана процедура P1 так, как это показано ниже.

Procedure P1(X,Y,Z:Real;K:Integer);

. . . . . . .

Begin

 { Тело процедуры }

End;

. . . . . . .

Begin  { Основная программа }

. . . . .

P1(A,B,C,N);    { Вызов процедуры P1}

. . . . .

End.

При вызове процедуры P1 будет установлено следующее соответствие между фактическими и формальными параметрами:

Для обеспечения соответствия массивов и некоторых других типов данных (записей, файлов) стандарт языка Паскаль требует их описания в разделе Type. Типы, определяемые таким образом, имеют тот же статус, что и стандартные: Real, Integer и др. Описатель типов в программе располагается после описания констант и перед описанием переменных. В общем виде типы пользователя описываются так.

Type

Имя_типа_1=описание этого типа;

Имя_типа_2=описание этого типа;

. . . . . .

Пример.
Const

Nmax = 10;

Mmax = 100;

Type

MasInt =Array[1.. Nmax,1.. Nmax] of Integer;

MasStr =Array[1.. Mmax] of String;

Параметры процедур делятся на 2 класса:

1) параметры–значения;

2) параметры–переменные.

Параметр–значение используется в процедуре, но после окончания ее работы не изменяется. Как правило, это аргумент (исходное данное) процедуры.

Так, в процедуре GorLine параметры n, Sim, а в Р1 – все 4 параметра являются параметрами–значениями. При обращении к процедуре Р1 значения фактических параметров A, B, C, N присваиваются соответствующим формальным параметрам X, Y, Z, K. Поэтому, если в ней был, например, оператор Z:=Z+1, то изменится именно значение Z, а переменная A - нет. Другими словами, после завершения работы процедуры все переданные ей параметры–значения в главной программе не меняются.

Если формальный параметр является параметром–значением, то фактический параметр может быть выражением соответствующего типа, например:

GorLine (20,’+’);

GorLine (2*K-1, Pred(Simvol));

Параметр–переменная не только используется в процедуре, но и позволяет изменять значение соответствующего фактического параметра. Это – результат работы процедуры. При вызове процедура получает доступ к области памяти, в которой хранится значение фактического параметра, соответствующего параметру–переменной. Таким образом, обеспечивается возможность изменения этого фактического параметра.

В заголовке параметру-переменной должен предшествовать описатель (слово) Var.

Пример.
Procedure SumXY(X,Y:Real;Var Z,Q:Real);

Begin

Z:=X+Y;

If Z<0 Then

Q := -Z

Else

Q := Z;

End;
{ SumXY }

Описатель Var записывается столько раз, сколько разных типов параметров-переменных используется в процедуре. Параметр-переменная при вызове не может представляться выражением.

18.1.1. Пример программы с процедурой

Задача. Выполнить упорядочение по возрастанию нескольких массивов вещественных чисел. В программе используется процедура сортировки методом «пузырька», алгоритм которой описан ранее.

Исходными данными для процедуры являются массив вещественных чисел и его размерность. Стандарт языка Паскаль, как отмечалось выше, требует описания массива в разделе Type (в программе это новый тип - MasReal). При использовании процедуры сортировки для согласования параметров необходимо в основной программе и процедуре ссылаться только на этот тип.

Program Sort;

Const

M = 200; { Максимальный размер массива }

Type

MasReal = Array [1..M] Of Real;

Var

A,B : MasReal;

I,N : Integer;

Procedure SortY (L:Integer; Var Y:MasReal);

{ Сортировка вещественных чисел методом "пузырька" }

{ L – размер массива; Y – массив вещественных чисел}

Var

i,k : Integer;

  Z : Real;

Begin

For k:=1 To L-1 Do

For i:=1 To L-k Do

If Y[i]>Y[i+1]Then

Begin

Z := Y[i];

Y[i] := Y[i+1];

Y[i+1] := Z

End;

End; { SortY }

Begin
{Основная программа}

Writeln(’ Введите размер массивов’);

ReadLn(n);

Writeln(’Введите первый массив’);

For i : =1 To n Do

ReadLn(A[i]);

Writeln (’Введите второй массив’);

For i := 1 To n Do

ReadLn(B[i]);

SortY (n,A); { Сортировка массива А }

SortY (n,B); { Сортировка массива В }

Writeln (’Первый  массив’);

For i := 1 To n Do

Write (A[i]:8:3);

Writeln;

Writeln (’Второй массив’);

For i := 1 To n Do

Write(B[i]:8:3);

Readln;   { Ожидание нажатия клавиши ENTER }

End.

18.1.2. Расположение процедур в программе


Процедуры могут быть вложенными друг в друга (см. рис. 2.10,а).

P1 и P3 - внешние процедуры,                      Р1, P2 и P3 - внешние 

Р2 – внутренняя






  процедуры

Рис. 2.10. Расположение процедур в программе

При таком расположении обращение к внутренней процедуре P2 возможно только из охватывающей ее процедуры P1, а из внешних (на рис. – Р3) - невозможно. Более предпочтительным является расположение процедур в виде рис. 2.10,б, на котором все процедуры являются внешними по отношению друг к другу.

При таком расположении:

1) любая процедура имеет доступ к любой другой (в том числе и к самой себе);

2) облегчается отладка процедур, т.к. они могут быть написаны и отлажены независимо друг от друга.

18.2. Функции

Функция — это независимый программный блок (программная единица), который позволяет получить только один результат.

Примером являются стандартные функции Паскаля. Имеется возможность писать свои функции. Располагаются они в программе так же, как процедуры: после описания данных и перед операторами.

Пример. 

Program . . .;

Const

. . .

Var
. . .

Функция_1

. . . . .

Функция_n

Begin  { основная программа }

. . . . .

End.

Функция оформляется так же, как процедура. Отличие имеется в заголовке и способе передачи результата. В общем виде функция оформляется так:

Function Имя(формальные_параметры):Тип_Результата;

Описание локальных переменных

Begin

. . . .

Имя := Выражение;

. . . .

End;

Формальные параметры – это аргументы функции. Они могут отсутствовать. Результат передается через имя функции, поэтому в заголовке указывается тип результата. По этой же причине среди операторов в теле функции должен быть хотя бы один, в котором имени функции присваивается значение. Именно это значение и будет возвращено функцией в качестве результата.

Пример. Оформить в виде функции операцию возведения вещественного числа в целую степень.

Function Stepen(x:Real;n:Integer):Real;

Var

i:Integer;

P:Real;

Begin

  If X = 0 Then

    If n = 0 Then 



Stepen := 1

    Else

      Stepen := 0

Else

  If n = 0 Then

     Stepen := 1

  Else

If n > 0 then

Begin




 P := 1;




 For i := 1 to n do





P := P*X;

 Stepen := P;

End 

else

Begin

  P:=1;

             For i := 1 to Abs(n) do

    P := P/X;

    Stepen := P;

End;

End;  { Stepen }

Для вычисления значения функции с конкретным параметром ее, как и процедуру, вызывают из основной программы с указанием фактических параметров–аргумен​тов. В отличие от процедуры имя функции может встречаться в выражениях в качестве операнда. Когда выражение вычисляется, функция выполняется и значением операнда становится величина, возвращаемая функцией.

Формальные параметры функции могут быть только параметрами–значениями. При ее вычислении они не изменяются. Аргументы функции могут быть любых типов:

1) скалярные — числа, символы, булевские;

2) массивы;

3) другие типы.

Сама функция (результат) имеет только скалярный тип, (т.е. число - целое или вещественное, а также символ или булевская переменная).

При обращении к функции (вызове ее) на место формальных параметров подставляются фактические, как в процедуре. Типы, количество и порядок следования фактических и формальных параметров должны совпадать.

В теле функции могут использоваться (вызываться) и другие функции, стандартные или написанные пользователем.

Пример. Вычислить гиперболический синус по формуле:

Sinh(x) = (ex – e-x)/2

Функция будет иметь вид

Function Sinh(X:Real):Real;

Begin

Sinh := (Exp(X)-Exp(-X))/2;

End;

К функции можно обращаться из основной программы; из другой функции и из самой себя.

Функции, которые вызывают сами себя, называются рекурсивными. Такие функций часто встречаются в математике, например, вычисление факториала, возведение числа в целую степень, вообще, нахождение любых значений методом математической индукции. Так, факториал вычисляется по формуле:

n! = n*(n -1)!
-  при n>0 и

n! = 1


- при n=0.

Для рекурсивных функций необходимо вырабатывать условие завершения (иначе произойдет зацикливание). Это условие для n! записано во второй строчке (при n=0) – так как рекурсия производится на основе второго сомножителя - (n-1)!=(n-1)*(n-2)! и т.д. - до нуля.

Запишем рекурсивную функцию вычисления факториала.

Function Fact (N:Integer):Integer;

Begin

If N=0 Then 

  Fact := 1

      Else

        Fact := N*Fact(N-1);

End;

18.2.1. Пример программы с функцией

Задача. Найти сумму элементов в нескольких массивах.

Program Count_Sum;

Const

Nmax = 200;

Type

Mas = Array[1..Nmax] Of Real;

Var

A,B : Mas;

Suma,Sumb : Real;

i,n : Integer;

Function SumMas (k:Integer;X:Mas):Real;

{Нахождение суммы элементов массива вещественных чисел}

Var 

I : Integer;

S : Real; { Вспомогательная переменная - сумма}

Begin

S := 0;

For i := 1 to k Do

S:=S + X[i];

SumMas:=S;

End; { SumMas } 

Begin  

{ Основная программа }

Writeln(’Введите размер массивов’);

ReadLn(n);

Writeln(’Введите первый массив’);

For i := 1 to n Do

ReadLn (A[i]);

Writeln(’Введите второй массив’);

For i := 1 to n Do

ReadLn (B[i]);

Suma := SumMas(n,A);

Sumb := SumMas(n,B);

Writeln(’Сумма чисел 1-го массива =’,Suma:8:2);

Writeln(’Сумма чисел 2-го массива =’,Sumb:8:2);

WriteLn(’Работа окончена. Нажмите клавишу ENTER’);

Readln;

End.

18.3. Внешние процедуры и функции

Если тексты процедур и функций размещены непосредственно в программе (как это было описано в предыдущем разделе), то их можно назвать внутренними. Когда возникает необходимость применить некоторые из этих процедур и/или функций в другой программе, придется переписать их туда. Хотя сделать это совсем не сложно, но при большом их количестве увеличение текста программы может оказаться чрезмерным. В Турбо Паскале используется более изящное решение этой проблемы – процедуры и функции можно помещать в модули, которые потом допустимо присоединять к любым программам (при этом совсем необязательно иметь в наличии исходный текст).

Процедуры и функции, описанные в модуле, являются внешними по отношению к программе, которая их использует.

В Турбо Паскале существует 2 вида модулей:

1) модули пользователей;

2) стандартные модули.

18.3.1. Модули пользователей

Структура модуля (стандартного или написанного пользователем) приведена на рис. 2.11.

Модуль должен начинаться с заголовка, включающего служебное слово Unit и имя_модуля. 

Пример.

Unit MyModul;  или 

Unit Sort;

Имя файла, содержащего модуль, и имя модуля должны быть одинаковыми! 

Применительно к приведенному примеру это означает, что модуль MyModul должен храниться в файле MyModul.pas, а модуль Sort – в файле Sort.pas.

Интерфейсная секция — общедоступная часть, "видимая" из любой программы (модуля). В ней описываются так называемые глобальные данные (константы, типы, переменные), используемые в модуле и доступные программе (или другим модулям, к которым присоединен данный). Здесь же содержатся заголовки (но не тексты!) процедур и функций, помещенных в модуль. Любая программа имеет доступ к этим объектам, как если бы они были описаны в ней.

Секция реализации содержит тела процедур и функций, описанных в интерфейсной секции. Здесь же описываются константы, переменные, процедуры и функции, являющиеся локальными, т.е. недоступными вне этого модуля.

После нее может быть секция инициализации. Она не имеет заголовка и может содержать (или не содержать) некоторую последовательность операторов, выполняющуюся при использовании модуля. 

Итак, последняя часть модуля всегда имеет следующий вид:

Begin
Операторы, выполняющиеся при инициализации модуля, которые могут отсутствовать.

End.

Модуль компилируется отдельно. Получается файл с расширением tpu, например, MyModul.tpu или Sort.tpu. Откомпилированный модуль компилятором должен быть помещен на диск. Если это не делается по умолчанию, то в меню Compile должно быть устанавлено Destination - disk (а не Memory).

Чтобы программа могла работать с какими-либо модулями, сразу после ее заголовка должен быть записан оператор использования, определяемый служебным словом Uses. После этого слова через запятую перечисляются имена всех модулей, непосредственно используемых программой, а также модулей, используемых первыми.

Пример. 

Program Test;

    Uses 

MyModul, Sort, Dos;

Const
Пример 1. Напишем модуль, содержащий процедуру сортировки массива вещественных чисел, и функцию, определяющую сумму его элементов.

Unit Sort;

Interface { интерфейсная секция  }

Const
{  Глобальная константа и тип }

Nmax=200;

Type

mas = array[1..Nmax] of real;

Procedure SortY(Var Y : mas; m : integer);

Function SumMas (Y : mas; m : integer): real;

Implementation
{ секция  реализации }

Procedure SortY;
{ можно только имя }

Var

i,k : integer;

a : real;

Begin

for k := 1 to m-1 do

for i := 1 to m-k do

if Y[i]>Y[i+1]then

Begin

a := Y[i];

Y[i] := Y[i+1];

Y[i+1] := a;

end;

end; { SortY }

Function SumMas;  { можно только имя }

{Нахождение суммы элементов массива вещественных чисел}

Var 

I : Integer;

S : Real; { Вспомогательная переменная - сумма}

Begin

S := 0;

For i := 1 to k Do

S:=S + X[i];

SumMas:=S;

End; { SumMas } 

Begin

{ секция инициализации не содержит операторов }

End.

Основная программа, которая использует этот модуль, может быть такой.

Program Main;

Uses

  Sort;      {пользовательский модуль}

Var

X:mas; { mas - глобальный тип, описанный в модуле sort}

n,i,j:integer;

z:real;

Begin

Writeln(’Введите размер массива’);

ReadLn (n); { размер массива }

Writeln(’Введите массив’);

For i := 1 to n do

ReadLn (x[i]);

SortY (x,n);

Z := SumMas (x,n);

Writeln(’Сумма элементов массива = ’,Z:8:2);

Writeln(’Упорядоченный массив’);

For i := 1 to n do

Write (x[i]:8:2);

Writeln; 

Writeln ('Конец работы. Нажмите клавишу ENTER');

ReadLn;

End.

19. Итерационные циклы

В этих циклах количество повторений заранее неизвестно. Выход из цикла происходит, если достигнута заданная точность результата или появилась стандартная ситуация (конец данных). Такие циклы используются при решении задач численными методами: приближенное вычисление функций, определенных интегралов, решение уравнений и т.д. 

Основной особенностью итерационных циклов является то, что новое значение искомой величины (Yслед) вычисляется с использованием старого (Yпред) по формуле:

Yслед=f(Yпред)

Обычно для вычисляемой величины Y задается некоторое начальное значение (приближение) Yo и погрешность вычисления результата (Y) в виде: E=(0,5; 1)*10-m, где m – целое число. Как правило, E=(0,1; 0,5, 0,01; 0,05, 0,001 и т.д.). Считается, что заданная точность достигнута, если найденные значения Yпред и Yслед отличаются на величину, меньшую E, т.е. условие окончания вычислений можно записать так:

|Yслед – Yпред|<E

Для программирования итерационных циклов в Паскале используются операторы While и Repeat.

19.1. Приближенное вычисление функций

Пример 1. Вычислить значение функции SinX по приближенной формуле

F = SinX = X - X3/3! + X5/5! - ...+(–1)2n-1 X2n-1/(2n-1)!,   n=1, 2, 3 ...

с погрешностью E=10-5.

Функция F здесь представляет собой сумму членов степенного ряда:

F = U1 + U2 + ... + Un
Условие окончания вычисления функции:

|F - Fслед|< E

Заметим, что |F-Fслед| = |Uслед| – очередное слагаемое. Можно сказать, что вычисления прекращают, как только получают член ряда, меньший, чем заданная абсолютная погрешность.

Обратите внимание, что вычислять слагаемые ряда непосрественно по формуле (находить степень и факториал, делить их) крайне нерационально. В этих случаях целесообразно выражать последующий член через предыдущий. Покажем, как это делается. 

Пусть речь идет о слагаемом с номером n. Тогда на основании общей формулы для члена ряда это слагаемое (предыдущее) без учета знака запишется так:

Номер следующего слагаемого будет n+1, тогда само слагаемое запишется так:

Разделив последнее выражение на предыдущее, получим выражение для вычисления следующего члена ряда по известному предыдущему

Обратите внимание, что в этой формуле n означает номер предыдущего члена. Если считать, что n имеет начальное значение 1 и изменяется с шагом 2, то знаменатель Uслед может быть вычислен как произведение вида:

(n + 1) (n+2).

Заметим еще, что при программировании подобных задач часто вводят одну переменную для представления как предыдущего слагаемого, так и последующего.

Вначале: n=1; U=X; F=0 {U – играет роль предыдущего и следующего члена}

Алгоритм

1. Ввести X и Е. 

2. Положить n=1, U = X, F=0.

3. Пока |U|>E
Вычислить F=F+U;

Вычислить очередной член U = –U X2/((n + 1) (n+2));

Получить его номер n=n+2.

4. Вывести X и F.

5. Закончить.

Программа для рассмотренного алгоритма будет иметь следующий вид

Program NovSin;

Var

F, X, E, U : Real;

n : Integer;

Begin

Writeln('Введите аргумент Х и погрешность Е');

Readln(X,E);

n := 1;

U :=X ;

F:=0;

While Abs(U)>E Do

  Begin

 F:=F+U;

 U:=-U*Sqr(X)/(2*N*(2*N+1));

 N:=N+1;

  End;

Writeln(X:8:6,' SinX=',F:10:6,

' Машинный SinX=',Sin(X):10:6);

End.

Количество повторений цикла заранее неизвестно. Оно зависит от требуемой точности и от значения аргумента. Если X=0, то цикл не выполняется ни разу. Количество повторений цикла в данном случае можно узнать по номеру N последнего вычисленного члена. В других случаях это можно сделать с помощью обычного счетчика:

k := 0;

while abs(U)>E do

begin

k := k+1;

F :=.......;

. . . . . . 

end;

writeln('k= ',k);

19.2. Решение уравнений приближенными методами

В общем виде уравнение может быть записано так:

f(x)=0,

где f(x) – произвольная функция. При этом невозможно записать формулу для нахождения его корней, за исключением квадратного и линейного уравнения. Для таких случаев корни определяются приближенными методами. Наиболее распространенными из них являются:

· метод деления отрезка пополам;

· метод Ньютона и

· метод прохождения отрезка с переменным шагом.

19.2.1. Метод деления отрезка пополам
Это — наиболее простой метод, позволяющий найти корень для функции любого вида, если только правильно выбран интервал, на котором он существует. Метод использует известное из математики свойство, которое заключается в следующем. Если на некотором отрезке функция меняет знак, то на этом отрезке она пересекает ось Х, т.е. имеет корень.

Поиск корня осуществляется следующим образом.

1. Выбирается интервал [a,b] значений аргумента Х, на котором ищется корень. (На этом интервале, как отмечалось выше, функция должна менять знак).

2. Начальное значение корня X0  принимается равным левой (a) или правой (b) границе интервала.

3. Вычисляется очередное приближение по формуле

Х = (Правая_граница - Левая_граница)/2.

4. Определяются значения функции f на одной из границ отрезка (например, левой) и в точке очередного приближения Х.

5. Если эти значения имеют разные знаки, то одну из границ (правую — см. п. 4) переносят в точку Х.

Пункты 3 — 5 повторяют до тех пор, пока разность между двумя соседними значениями Х не станет меньше или равно заданной погрешности Е. Последнее приближение Х считается корнем.

Составим алгоритм и программу нахождения корня описанным методом, считая что в программе будет использована функция f(x).

Алгоритм

1. Ввести a, b и Е.

2. Найти корень функции f(x) на отрезке [a,b].

3. Закончить.

Уточненный алгоритм, в котором предыдущее и следующее приближения корня обозначены Хпред и Хслед, можно представить в виде.

1. Ввести a, b и Е.

2.1. Левая_граница:= a;

2.2. Правая_граница:= b;

2.3. Хпред:= Левая_граница;

2.4. Повторять

2.4.1. Хслед:= (Правая_граница - Левая_граница)/2.

2.4.2. у1:= f(Xпред).

2.4.3. у2:= f(Хслед).

2.4.4. Если у1 и у2 имеют разные знаки, то

Правая_граница:= Хслед

Иначе

Левая_граница:= Хслед
2.4.5. у:= |Хслед  - Xпред|

2.4.6. Xпред := Хслед
Пока не будет у ( Е.

2.5. Вывести Хслед и у2.

3. Закончить.

Программа для этого алгоритма будет иметь вид

Program  Div2;

Var

Xsl, Xpr, a, b, E, y1, y2, Lev, Prav, y: Real;

Function f(x: Real): Real;

Begin

f:= { здесь должна быть формула для вычисления функции}

End;

Begin
Writeln('Введите границы отрезка и погрешность');

Readln(a, b, E);

Lev:= a;

Prav:= b;

Xpr:=Lev;

{ вычисление корня }

Repeat

Xsl:= (Prav - Lev)/ 2;

Y1:= f(Xpr);

Y2:= f(Xsl);

If ((y1>=0) And (y2<0))Or((y1<0) And (y2>=0)) then

Prav:= Xsl

Else

Lev:= Xsl;

Y:= Abs(Xsl - Xpr);

Xpr:= Xsl;

Until y <= E;

Writeln('Корень = ', Xsl:8:4);

Writeln('Функция = ', y2:10:7);

Writeln('Работа окончена');

Readln;

End.
19.2.2. Метод Ньютона
Этот метод еще называют методом касательных. Он позволяет быстрее найти корень (сходится за меньшее количество приближений), чем предыдущий. Метод заключается в следующем.

1. Выбирается интервал [a,b] значений аргумента Х, на котором ищется корень. На этом интервале функция должна менять знак.

2. Начальное значение корня X0 принимается равным левой (a) или правой (b) границе интервала.

3. Вычисляется очередное приближение по формуле

Хслед = Хпред – f(Хпред)/f '(Хпред)

4. Если |Хслед - Хпред| <= E,  то Хслед – корень, в противном случае

присвоить значение Хслед переменной Хпред и перейти к п. 3.

В отличие от метода деления отрезка пополам рассмотренный алгоритм позволяет определить корень не для любой функции, а если:

1) f(x) дифференцируема;

2) f '(x) ≠ 0 в Е-окрестности корня.

Неудачный выбор Х0 может привести к тому, что приближения "расходятся" от точки корня.

Рассмотрим пример программы нахождения корня уравнения методом Ньютона. Будем, как и в п. 19.2.1 считать, что в программе будет использована функция f(x) и ее производная Prf(x).

Обозначим:

Хo
=>
x0;
E
=>
E;     Хпред
=>
x;     Хслед
=>
xn

Программа будет иметь вид

Program  Newton;

Var

y, x0, X, xn, E : Real;

Function f(x: Real): Real;

Begin

f:= { здесь должна быть формула для вычисления функции}

End;

Function Prf(x: Real): Real;

Begin

Prf:= { здесь будет формула для вычисления производной}

End;

Begin

Writeln('Введите начальное приближение корня и погрешность');

Readln(a, E);

x := a;

{ вычисление корня }

Repeat

xn := x-f(x)/Prf(x);

y := Abs(xn-x);

x := xn;

Until y <= E;

Writeln('Корень = ',xn:8:4,'   разность = ',y:8:6);

Y:=f(x);

Writeln('Функция = ', y:10:7);

Writeln('Работа окончена');

Readln;

End.

Можно определить количество приближений (итераций), оно равно числу повторений цикла.

19.2.3. Метод прохождения отрезка с переменным шагом

В отличие от двух предыдущих этот метод позволяет найти не один, а несколько корней, если таковые имеются на выбранном интервале. Он предоставляет возможность локализовать каждый из искомых корней (определить все отрезки [a,b]). Сами корни находят одним из первых двух методов.

Суть метода заключается в следующем.

1. Выбирается интервал [А, В] значений аргумента Х, на котором ищутся корни.

2. Определяется начальное значение шага Н = (В - А)/ n, где n — начальное количество точек на интервале.

3. Проходят интервал [А, В] с шагом Н, вычисляя значения функции f(x) и f(x+H).
4. Если вычисленная пара значений функции имеет разные знаки, корень локализован. Его можно определить, например методом Ньютона. После нахождения корня отступают от него на величину шага Н.

5. Уменьшают шаг, например: Н:= Н / 2

6. Пункты 3 — 5 выполняют до тех пор, пока не будут найдены все корни.

Алгоритм нахождения корня описанным методом, который предполагает, что в программе будет использована функция f(x) и ее производная Prf(x), а также процедура нахождения корня уравнения методом Ньютона, приведен ниже.

1.1. Ввести А, В и Е.

1.2. Ввести Предполагаемое количество корней.

1.3. Вычислить начальный шаг Н = (В - А)/ n.

2.    Повторять

2.1. Найдено_корней:= 0.

2.2. Х:= А.

2.3. Пока Х < B выполнить.

2.3.1. Вычислить у1:= f(X);

2.3.2. Вычислить y2:= f(x+H);
2.3.3. Если у1 и у2 имеют разные знаки, то

а) положить Левая_граница:= Х;

б) положить Правая_граница:= Х + Н;

в) найти корень методом Ньютона и вывести его;

г) Найдено_корней:= Найдено_корней + 1;

2.3.4. Х:= Х + Н;

2.4. Н:= Н / 2;

Пока не будет (Предполагаемое = Найдено_корней) Или (Н ( Е).

3. Закончить.

Программа для этого алгоритма будет иметь вид

Program  Roots;

Const
n= 20; { начальное количество точек на интервале }

Var

A, B, X, y1, y2, lev, Prav, E : Real;

Predpol, Naideno: Integer; { количества корней }

Function f(x: Real): Real;

Begin

f:= { здесь должна быть формула для вычисления функции}

End;

Function Prf(x: Real): Real;

Begin

Prf:= { здесь будет формула для вычисления производной}

End;

Procedure Newton(Lev,E: Real; Var x:Real);

  { нахождение корня методом Ньютона }

{ Lev - левая граница, x - корень }

Var

Y, xn: Real;

Begin

x := Lev;

{ вычисление корня }

Repeat

xn := x-f(x)/Prf(x);

y := Abs(xn-x);

x := xn;

Until y <= E;

X:= xn;

End;

Begin

Writeln('Введите интервал нахождения корней и погрешность');
Readln(A, B, E);

Writeln('Введите предполагаемое количество корней');

Readln(Predpol);

{ Начальный шаг }

H:=(B-A)/2;

{ Поиск корней  }

Repeat

Naideno:=0; { найдено корней }

X:=A;

While x < B do

Begin

Y1:= f(x);

Y2:= f(x+H);

If ((y1>=0) And (y2<0))Or((y1<0) And (y2>=0)) then

Begin
{  корень локализован }

Lev:= x;

Newton(Lev, E, x);

Y1:= f(x);

Writeln('Корень = ', Xsl:8:4);

Writeln('Функция = ', y2:10:7);

End;

X:= x + H;

End;

H:= H / 2;

Until (Predpol = Naideno) Or (H <= E);

If Predpol <> Naideno Then

Writeln('Количество корней задано неверно');

Writeln('Работа окончена');

Readln;

End.

19.3. Вычисление определенных интегралов

Известно, что определенный интеграл некоторой функции на f(x) интервале [a,b] равен площади фигуры, ограниченной кривой f(x), осью х и вертикальными линиями, проходящими через границы интервала. Вычисление такого интеграла приближенными методами предполагает замену указанной фигуры более простыми, площади которых можно найти по формулам, применяемым в элементарной геометрии. Очевидно, что чем меньше размеры элементарных фигур — тем точнее результат. Если разность между двумя последовательными приближениями окажется меньше заданной погрешности Е, то последнее значение можно считать результатом.

Общий алгоритм вычисления определенного интеграла можно представить следующим образом.

Алгоритм

1. Ввести a, b и Е.

2. Вычислить начальное значение площади (приближение).

3. Повторять

Вычислить очередное приближение

Пока абсолютная величина разности между соседними приближениями не будет меньше или равна Е.

4. Закончить.

В качестве элементарных фигур, которыми заменяется исходная, наиболее часто используют прямоугольники или трапеции. Соответственно методы приближенного вычисления определенного интеграла называют:

1. Метод прямоугольников и

2. Метод трапеций.

19.3.1. Метод прямоугольников

Метод основан на замене площади сложной фигуры суммой площадей прямоугольников, высота которых равна значению функции f(x), а ширина — hx, шагу по оси х. Очевидно, что чем меньше шаг, тем точнее аппроксимация. Площадь элементарного прямоугольника равна f(x)* hx.

Алгоритм вычисления определенного интеграла рассматриваемым методом может быт таким.

1. Ввести a, b и Е.

2.1. Начальное значение шага hx = (b-a)/n, где n — начальное количество точек;

2.2. Вычислить площадь S1 = ( Sпрямоуг (hx)

2.3. Повторять

2.3.1. hx := hx / 2;

2.3.2. Вычислить площадь S2 = ( Sпрямоуг (hx);

2.3.3. Y:= | S2 - S1|;

2.3.4. S1:= S2
Пока не будет y ( Е.

2.4. Вывести S2.

3. Закончить.

В приведенном алгоритме дважды вычисляется сумма площадей элементарных прямоугольников. Эту операцию можно оформить в виде процедуры или функции. Программа, которая реализует описанный алгоритм с использованием функции f(x) и процедуры вычисления площади методом прямоугольников приведена ниже.

Program Integral;

Const
n= 20; { начальное количество точек на интервале }

Var

a, b, hx, s1, s2, E, y : Real;

Function f(x: Real): Real;

Begin

f:= { здесь должна быть формула для вычисления функции}

End;

Procedure S_Rect(a,b,h:Real; Var S:Real);

{  вычисление интеграла методом прямоугольников  }

{ a и b — границы интервала, h — шаг по оси х }

{ S — результат }

Var

X: Real;

Begin

S:=0;

X:=a;

While x<b do

Begin

S:= S + f(x)*h;

x:= x+h;

End;

End;

Begin
   {  основная программа }

Writeln('Введите интервал и погрешность');
Readln(a, b, E);

{ Начальный шаг }

hx:=(b-a)/n;

{ Начальное значение интеграла (площади) }

S_Rect(a,b,hx,s1);

{  Вычисление очередных приближений  }
Repeat

hx:=(b-a)/2;

S_Rect(a,b,hx,s2);

y:=Abs(s2-s1);

s1:=s2;

Until y <= E;

Writeln('Интеграл = ', s2 :10:6);
Writeln('Работа окончена');

Readln;

End.
19.3.2. Метод трапеций
Метод основан на замене площади сложной фигуры суммой площадей трапеций, высота которых равна шагу по оси х (hx), а основания — значениям функции f(x) и f(x + hx). Очевидно, что он позволяет точнее аппроксимировать функцию и быстрее получить результат. В этом можно убедиться, подсчитав количество приближений обоими методами. Из геометрии известно, что площадь элементарной трапеции равна (f(x) + f(x + hx)) / (2* hx).

Алгоритм вычисления определенного интеграла рассматриваемым методом отличается от предыдущего только процедурой нахождения площадей s1 и s2. Эта процедура может быть предствалена в следующем виде.

Procedure S_Trap(a,b,h:Real; Var S:Real);

{  вычисление интеграла методом трапеций  }

{ a и b — границы интервала, h — шаг по оси х }

{ S — результат }

Var

X: Real;

Begin

S:=0;

X:=a;

While x<b do

Begin

S:= S + (f(x) + f(x + h)) / (2* h).;

x:= x+h;

End;

End;

Вызов в основной программе с именем Integral процедуры S_Trap вместо S_Rect позволит реализовать в ней метод трапеций.

20. ДОПОЛНИТЕЛЬНЫЕ СВЕДЕНИЯ О ТипАХ данных,
применяемЫХ в Паскале

Помимо рассмотренных ранее скалярных данных и массивов, в Паскале используются следующие типы данных:

1) перечисляемые;

2) интервальные;

3) множества;

4) записи;

5) файлы.

Ознакомимся с каждым из них поочередно.

20.1. Перечисляемый тип

Этот тип позволяет описывать алгоритмы в понятиях, близких к общепринятым в математике и в повседневной жизни. Например, дни недели, стороны света, направления движения и другие данные, значения которых можно перечислить, в программах можно представить числами или строками (словами). Но применение перечисляемого типа в такой ситуации является предпочтительным.

Перечисляемый тип задает упорядоченное множество значений путем перечисления имен констант, обозначающих эти значения. В разделе Type в общем виде он описывается так

Type
Имя=(Значение1, Значение2,…,ЗначениеN);

Порядковый номер первой из перечисленных констант равен 0, следующей – 1 и т.д. Максимальное количество констант равно 256.

Пример.
Type

Week_Days=(Sunday, Monday, Tuesday, Wednesday, Thursday, Friday, Saturday);

Side=(North, East, South, West);

Объявленные типы могут использоваться для описания переменных в разделе Var.
Пример 

Var

Days,Week_End : Week_Days;

Curs: Side;
Имена констант должны удовлетворять всем требованиям образования имен идентификаторов (в частности, кириллица недопустима). 

Перечисляемые типы невозможно вводить с клавиатуры или выводить на экран (к ним нельзя применять операторы Read и Write). Для них разрешен ограниченный набор операций: сравнение (равно, не равно, больше, меньше и т.д.), а также функции:

Pred(аргумент) – дает элемент, предшествующий аргументу; результат определен для всех элементов списка, кроме первого;

Succ(аргумент) – дает элемент, сдедующий за аргументом; результат определен для всех элементов списка, кроме последнего.

Пример
Pred(Tuesday) = Monday,

Succ(Tuesday) = Wednesday.

Рассмотрим пример программы, использующей перечисляемый тип.

Задача. Составить алгоритм и программу, определяющую новый курс корабля по предшествующему курсу и команде. Для простоты будем считать, что курс соответствует сторонам света:

Север, Восток, Юг, Запад.

Команда вводится с клавиатуры и может принимать следующие значения:

Вперед, Вправо, Назад, Влево, Стоп.

Последняя команда приводит к окончанию работы программы.

При разработке алгоритма будем ориентироваться на использование перечисляемого типа для задания курса корабля. Команду же будем задавать в виде обычной строки (слова).

Алгоритм

1. Задать начальное значение курса.

2. Повторять

Ввести команду, определить новый курс и

Вывести его на экран

Пока не будет дана команда “Стоп”.

3. Закончить.

Уточним алгоритм

1. Старый_курс = Север.

2. Повторять

2.1. Ввести Команду.

2.2. Если Команда = “Вперед”, то

Новый_курс= Старый_курс

Иначе

Если Команда = “Вправо”, то

Если Старый_курс = “Запад”, то

Новый_курс=”Север”

Иначе

Новый_курс= Следующий(Старый_курс)

Иначе

Если Команда = “Назад”, то

Выбрать Старый_курс из

Север, Восток: Новый_курс = Следующий(Следующий (Старый_курс));

Юг, Запад: Новый_курс = Предшествующий(Предшествующий (Старый_курс))

Иначе

Если Команда = “Влево”, то

Если Старый_курс = “Север”, то

Новый_курс=” Запад ”

Иначе

Новый_курс= Предшествующий(Старый_курс)

Иначе

Если Команда не = “Стоп”, то

Вывести “неправильная команда”.

2.3. Выбрать Новый_курс из

  Север : Вывести “ Север”

Восток : Вывести “Восток”

      Юг : Вывести “ Юг ”

  Запад : Вывести “ Запад "”

2.4. Старый_курс = Новый_курс.

Пока не будет Команда = “Стоп”.

3. Закончить.

Программа для этого алгоритма приведена ниже.

Program Ship;

Type

Side=( North, East, South, West);

Var

St_Curs,N_Curs : Side; { Старый и новый курс }

Com : String; { Команда }

Begin

St_Curs:= North;

WriteLn(’Идем на Север...’);

{ Определение нового курса  }

Repeat
Write(’Введите команду: ’);

Readln(Com);

If Com = ’вперед’ then

N_Curs:= St_Curs

Else

If Com = ’вправо’ then

Begin
{ Движение вправо }

If St_Curs = West then

N_Curs := North 

Else

N_Curs := Succ(St_Curs)

End

Else

If Com = ’назад’ then

Case St_Curs of

North,East : N_Curs := Succ(Succ (St_Curs));

South,West : N_Curs := Pred(Pred(St_Curs));

End

Else

If Com = ’влево’ then

If St_Curs = North then

N_Curs := West 

Else

N_Curs := Pred(St_Curs)

Else

If Com <> ’стоп’ then

Writeln(’Неправильная команда’:30);

Write(’     Новый курс: ’);

Case N_Curs of

North : Writeln(’Север’);

 East : Writeln(’Восток’);

South : Writeln(’Юг’);

 West : Writeln(’Запад’);

End;

St_Curs := N_Curs;

Until Com = ’стоп’;

Writeln(’Приехали! Нажмите Enter’);

Readln;

End.

Для самопроверки напишите программу, в которой курс корабля может быть задан более точно, с промежуточными направлениями: Север, Северо-восток, Восток и т.д.

20.2. Интервальный тип

Этот тип можно определить, указав некоторый диапазон значений из любого другого предварительно определенного порядкового типа, который в этом случае называют базовым. Диапазон определяется указанием наибольшего и наименьшего постоянных значений, входящих в диапазон, причем нижняя граница не должна превышать верхнюю.

Вновь задаваемый тип можно предварительно описать в разделе Type, как это показано ниже.

Type

Index = 0..63;

Letter = ’A’..’Z’;

или указывать его при описании переменных в разделе Var:

Const

Nmax=100;

Var
Curs : 1..5;

Bukva : ’a’..’z’;

Mas : Array[1..Nmax] Of Real;

В последней строке интервальный тип использован для описания диапазона значений индекса массива.

Если при выполнении программы значение переменной, описанной с помощью интервального типа, выходит за границы диапазона, то программа прерывается с выдачей сообщения об ошибке вида: Range Check Error.

20.3. Множества

Множество - это совокупность элементов одного и того же перечисляемого типа.

Оно задается перечислением элементов. Например, множество четных чисел от 0 до 10 в математике записывается так:

{ 0,2,4,6,8,10 }.

В Паскале оно представляется аналогично, только заключается в квадратные скобки:

[0,2,4,6,8,10]

В общем случае в разделе Var множество описывается следующим образом:

Var

Имя_множества : Set Of Тип_Компонент;

где
Тип_Компонент — обязательно порядковый.

Пример.
Type

Days = (Sunday, Monday, Tuesday, Wednesday, Thursday,

        Friday, Saturday);

Bukva = Set of ’a’..’z’;

Var

Sim : Bukva;

Chislo : Set of 1..20;

WeekEnd, Rabdny, ObDni : Set of Days;

При использовании последнего описания можно выполнять присваивания вида:

Weekend := [Saturday, Sunday];

Мощность множества — это максимальное количество элементов в нём. В Паскале оно ограничивается числом 256.

Множество может не содержать ни одного элемента. Оно называется пустым и обозначается следующим образом: [ ]. По определению пустое множество является компонентой любого множества.

Для множеств допустимы следующие операции:

1) сравнение:

а) на равенство двух множеств (совпадают множества или нет) ;

б) на вхождение одного множества в другое;

в) на принадлежность элемента множеству;

2) объединение;

3) пересечение;

4) вычитание.

1) Результат операций сравнения — булевский.

1, а) Операции сравнения на равенство выполняются, как и для других типов, например:

If ObDny = WeekEnd Then

Множества равны, если они содержат одни и те же константы. 

1, б) Проверка на вхождение выполняется с помощью знаков:

<=
или
>=

Выражение S1 <= S2 проверяет, все ли элементы множества S1 входят в S2, а
 S1 >= S2, содержит ли S1 все элементы из S2.

Пример. 
[1,2,3] = [2,3,1]

результат
True

[’A’, ’C’] <= [’A’, ’B’, ’C’, ’D’]
результат
True

[ ] <= [’A’] результат True - пустое множество входит в любое другое.

1, в) Для определения принадлежности некоторого элемента множеству в Паскале используется операция In вида:

Элемент In Множество,

которая дает значение True, если элемент принадлежит множеству.

Пример.
’А’ In [’A’,’B’,’Г’,’Д’]
результат
True

’A’ In [’X’,’Y’,’Z’]
результат
False

S In [’A’..’Z’]

результат True, если S - латинская буква.

Операция In может эффективно применяться для исключения сложных проверок. Например, вместо проверки
If (Ch=’A’) Or (Ch=’E’) Or (Ch=’I’) Or (Ch=’S’) Or (Ch=’X’)  then 

Оператор1;

можно записать более простую

If Ch In [’A’,’E’,’I’,’S’,’X’] then 

Оператор1;

2) Объединением двух множеств является множество, составленное из элементов, входящих в оба исходных. Операция объединения обозначается знаком +.

Пример.
[’A’,’B’,’C’] + [’D’,’E’] = [’A’,’B’,’C’,’D’,’E’]

[’A’,’B’] + [’A’,’E’] = [’A’,’B’,’E’]

3) Пересечением двух множеств является множество, состоящее из элементов, которые входят одновременно в оба исходных. Знак операции пересечения *.

Пример.
[’A’,’B’,’C’] * [’A’,’C’] = [’A’,’C’]

[’A’,’B’] * [’X’,’Y’] = [ ]

[ ]*[ ] = [ ]

4) Разностью двух множеств является множество, состоящее из элементов первого, которые не входят во второе. Знак операции вычитания: –.

Пример.
[’A’,’B’,’C’] – [’A’,’C’] = [’B’]

[’A’,’B’] – [ ] =  [’A’,’B’]

[’A’,’B’] – [’A’,’B’] = [ ].

20.3.1. Примеры программ с использованием множеств

При разработке алгоритмов и программ с множествами необходимо учесть, что Паскаль не обеспечивает ввод или вывод множеств непосредственно с помощью операторов Read или Write. Формирование множества, элементы которого вводятся с клавиатуры, можно выполнить, используя операцию присоединения, следующим образом.

Один из алгоритмов ввода множества

1. Множество :=[ ].

2. Повторять

     2.1. Ввести элемент;

     2.2. Множество:= Множество + [элемент].

         Пока не надоест.

Вывод содержимого множества осуществляется путем тестирования. При этом в цикле перечисляются все возможные значения его элементов и те из них, которые входят в множество, выводятся на экран.

Один из алгоритмов вывода множества

1. Для Элемента от Первого_значения до Последнего_значения выполнить

Если Элемент принадлежит (In) Множеству, то

Вывести Элемент.

Рассмотрим примеры алгоритмов и программ, использующих множества.

Пример 1. Сформировать и вывести на экран множество из n символов латинского алфавита (n<20). Если вводимая буква уже есть в множестве, выдать соответствующее сообщение.

Program  Mnog_Sim;

Const

N=20;

Var

S : Set Of Char;

B : Char;

I : Integer;

Begin

S := [];

For I:=1 To N Do

Begin

Writeln (’Введите букву’);

Readln(B);

If B In S Then

Writeln(’Буква ’,B,’ уже есть’)

Else

S:=S+[B];

End;

Writeln(’Буквы, входящие в множество’);

For B:=’A’ To ’z’ Do

If B In S Then

Write(B:2);

End.

Обратите внимание, что вводимые буквы могут быть и прописными, и строчными, поэтому при выводе в цикле For параметр цикла принимает все значения для букв латинского алфавита от A до z.

Пример 2. Используя множество гласных букв, подсчитать количество гласных в предложении. Предложение и множество гласных ввести с клавиатуры.

Program Glasn;

Var

P : String;        {предложение}

Gl : Set Of Char;  {множество гласных}

B : Char;

K,I,N : Integer;

Begin

Gl := [];

B := ’#’;

While B <> ’!’ do

Begin

Writeln (’Введите гласную, конец ввода – !’);

Readln(B);

If B <> ’!’ then

Gl := Gl+[B];

End;

Writeln (’Введите предложение’);

Readln(P);

K := 0;

N := Length(P);

For I := 1 To N  Do

If P[I] In Gl Then

K := K+1;

Writeln (’Количество гласных в предложении - ’,K);

WriteLn(’Нажмите клавишу ENTER...’);

ReadLn;

End.

Пример 3. Составить программу продажи билетов на пароход. Пусть места на пароходе нумеруются (для простоты) от 1 до n (n < 250). Покупатель заказывает номера мест. Если они свободны, то ему выдаются билеты, и номера мест исключаются из числа свободных. В противном случае предлагаются номера свободных мест. Если все билеты проданы, то начинается путешествие.

Алгоритм

1. Задать множество свободных мест: от 1 до n.

2. Повторять 

2.1. Вывести список свободных мест;

2.2. Запросить заказ на билеты;

2.3. Если места свободны, то

продать билеты (занять места, заказ принят),

иначе

сообщить, что некоторые места уже проданы (заказ не принят).

Пока не будут заняты все места.

3. Вывести сообщение, что все билеты проданы.

Уточняем условие. В задаче будем использовать 2 множества:

1) Свободные места (Sm);

2) Заказ (Zak).

Запишем уточненный Алгоритм.

1. Задать множество Свободных мест: от 1 до n.

2. Повторять

2.1. Вывести свободные места

2.2. Запросить заказ на билеты:

2.2.1. Заказ = [ ] – пустое множество.

2.2.2. Ввести номер_места

2.2.3. Пока номер ≠0 выполнить:

а) объединить номер_места с заказом;

б) ввести номер места.

2.3. Если заказ входит в Свободные места, то

2.3.1. Выдать сообщение: "заказ принят" 

2.3.2. Исключить проданные места 

(Свободные_места := Свободные_места – заказ)

Иначе

Выдать сообщение "заказ не принят".

Пока не будут заняты все места.

3. Вывести сообщение, что все билеты проданы.

Программа, реализующая этот алгоритм, приведена ниже.

Program Teploxod;

Const

N=250; { На время отладки назначьте небольшое N! }

Type

Mesta = Set Of 1..N;

Var

Sm,Zak : Mesta;{Свободные места и заказ}

I,Nom : Integer;

Begin
Sm := [1..N];{п. 1, свободные места}

Repeat              {п. 2}

WriteLn;

Writeln(’       Cвободные места: ’); {п. 2.1.}

For I:=1 To N Do

If I In Sm Then

Write (I:4);

WriteLn;

Zak := [];               {п. 2.2.1.}

    {Формирование Заказа}

Writeln(’Введите номера мест, которые Вы покупаете. ’,

        ’Конец заказа - ноль’);

ReadLn(Nom);                {п. 2.2.2.}

While Nom <> 0 Do           {п. 2.2.3.}

Begin

Zak := Zak + [Nom];

ReadLn(Nom);

End;

If Zak <= Sm Then              {п. 2.3.}

Begin

Writeln(’Заказ принят’);   {п. 2.3.1.}

Sm := Sm - Zak;            {п. 2.3.2.}

End
Else
WriteLn(’Заказ не принят. ’,

’Некоторые из заказываемых мест уже проданы.’);

Until Sm=[];

Writeln (’Все билеты проданы, нажмите Enter’); 

Readln;

End.

20.4. Записи

Записи — это совокупности элементов разных типов, имеющих общее имя и назначение. Примеры записей — адрес человека, анкетные данные, сводка об успеваемости и т.д. Массив – тоже совокупность данных, но одного типа. Компонентами (полями) записи могут быть переменные скалярного типа, строки, а также массивы и другие записи. Сами записи могут образовать массив.

В общем случае в разделе Type запись описывается так:

Type


Имя_записи = Record

Описания полей

end; 
{конец описания записи}

Имена полей в пределах одной записи не должны повторяться, но могут использоваться вне записи как имя другой переменной. Запись можно передавать в качестве параметра процедуре или функции, но функция не может возвращать значение типа запись.

Пример.

Запись, характеризующая адрес человека, очевидно, должна содержать сведения о городе, улице, № дома, № квартиры. Структура соответствующей записи в разделе Type может быть описана следующим образом.

Type

Adres = Record

Gorod : String [20];

   Ul : String [20];

  Dom : Integer;

   Kv : Integer

End;

Рассмотрим еще один пример, в котором полем записи будет массив.

Пример. Запись, описывающая студента, может иметь следующую структуру:

ФИО, номер зачетки, оценки по 5 предметам.

Type

Stud = Record

Fam : String[15];

 Im : String[10]

 Ot : String[15];

Numz : LongInt;

  O :Array[1..5] Of Integer

End;

Используя этот тип, сведения о 25 студентах группы можно поместить в массив, описав его так:
     Var
Grup : Array[1..25] Of Stud;

RabAnk : Stud;   { Рабочая переменная }

Значение переменной–записи можно присвоить другой переменной–записи того же типа. Ни в каких других операциях переменная–запись целиком участвовать не может. В необходимых случаях каждое поле записи должно обрабатываться отдельно.

При этом имя поля будет составным вида:

Имя_записи.имя_поля.

Пример.

RabAnk.NumZ := 2100756; {Поле номера зачетки в переменной RabAnk}
RabAnk.Fam := ’Иванов’; {Поле фамилии в переменной RabAnk}
Для массивов записей:

Grup[1].Fam := ’Абрамова’; {Фамилия первого студента в группе}

Writeln(Grup[i].Numz); {Номер зачетки i–го студента в группе}

Write(Grup[10].O[1]);
{оценка по первому предмету 10-го студента }

20.4.1. Примеры программ обработки записей

Пример 1. Ввести в ЭВМ ведомость студентов одной группы по пяти предметам. Вывести на экран фамилии и номера зачеток неуспевающих студентов.

Алгоритм

1. Ввести количество студентов (n).

2. Для номера студента (i) от 1 до n выполнить

ввести ведомость [i]

3. Количество_двоечников  = 0.

4. Для i от 1 до n выполнить

4.1. колич_2 = 0

4.2. Для j от 1 до 5 выполнить

Если у ведомости [i].оценка[j] < 3, то

колич_2 = колич_2 + 1

4.3. Если колич_2 > 0, то

а) вывести ведомость [i],

б) Количество_двоечников  = Количество_двоечников  +1

5. Вывести Количество_двоечников.

6. Закончить.

Программа для этого алгоритма будет иметь вид

Program Count_2;

Const

Nmax=40;  { Max количество студентов в группе }

Dl=15;    { длина фамилии и имени }

M=5;      { количество оценок }

Type

Stud = Record

Fam,Im,Ot : String [dl];

Numz : LongInt;

Otm : Array [1..m] Of Integer

End;

Var

Vedom : Array [1..Nmax] Of Stud;

i,j : Integer;

kdv,k2,n : Integer;

Begin

Writeln(’Введите количество студентов’);

Readln(n);                {п. 1.}

Writeln(’Вводите информацию о каждом студенте’);

For i:=1 to n Do          {п. 2.}

  Begin

    Write(’Фамилия: ’);

    ReadLn(Vedom[i].Fam);

    Write(’Имя: ’);

    ReadLn(Vedom[i].Im);

    Write(’Отчество: ’);

    ReadLn(Vedom[i].Ot);

    Write(’Номер зачетки: ’);

    ReadLn(Vedom[i].Numz);

    WriteLn(’Вводите оценки по одной:’);

    For j := 1 to m do

      Begin

        WriteLn(’Введите ’,j,’ –ю оценку’);

        ReadLn(Vedom[i].Otm[j]);

      End;

  End;

kdv := 0;                 {п. 3.}

Writeln(’Фамилии и номера зачеток двоечников’);

For i:=1 to n Do

Begin

k2 := 0;               {п. 4.1.}

For j := 1 to m Do     {п. 4.2.}

If Vedom[i].Otm[j] < 3 Then

k2 := k2+1;

If k2 > 0 Then           {п. 4.3.}

Begin

kdv := kdv + 1;

Writeln(Vedom[i].Fam:16,Vedom[i].Numz:8)

End;

End;

Writeln(’Количество двоечников в группе - ’,kdv);

WriteLn(’Нажмите Enter...’);

ReadLn;

End.

Для сокращения записи программ с данными типа Record используется оператор With, который в общем случае записывается так:

With  Имя_Записи Do

Оператор_1;

В Операторе_1 к полям записи автоматически добавляется Имя_Записи.

С использованием этого оператора цикл ввода данных в приведенной выше программе был бы записан так

For i:=1 to n Do

  With Vedom[i] do

    Begin

      Write(’Фамилия: ’);

      ReadLn(Fam);

      Write(’Имя: ’);

      ReadLn(Im);

      Write(’Отчество: ’);

      ReadLn(Ot);

      Write(’Номер зачетки: ’);

      ReadLn(Numz);

      WriteLn(’Вводите оценки по одной’);

      For j := 1 to m do

        Begin

          WriteLn(’Введите ’,j,’ –ю оценку’);

          ReadLn(Otm[j]);

        End;

    End;   { With }

Пример 2. Ввести в ЭВМ список характеристик N стран мира, содержащий информацию вида (N<=40):

Название, площадь

Составить программу, которая упорядочивает список по убыванию площади и выводит полученный результат на экран.

Общий алгоритм для этой задачи будет такой.

1. Ввести характеристики стран.

2. Упорядочить характеристики по убыванию площади.
3. Вывести полученный список.
4. Закончить.
Применим для упорядочения характеристик стран известный метод пузырька и получим уточненный алгоритм.
1.1. Ввести количество стран (n).

1.2. Для номера страны (i) от 1 до n выполнить

ввести страну [i].

2. Для номера просмотра (k) от 1 до n - 1 выполнить

Для номера страны (i) от 1 до n - k выполнить

Если у Страны[i]. площадь > Страны[i + 1]. площадь, то

Поменять местами Страну[i] и Страну[i + 1].

3. Для номера страны (i) от 1 до n выполнить

ввести у страны [i]. название и площадь.

4. Закончить.

Программа для этого алгоритма имеет вид.

Program Lends;

Const

Nmax=40;  { Max количество стран в списке }

Dl=15; { длина названия }

Type

Country = Record

Name : String [15];

Square : Integer;

End;

Var

Count : Array [1..Nmax] Of Country;

Count1: Country;

n, i, k : Integer;

Begin

Writeln(’Введите количество стран’);

Readln(n);

Writeln(’Вводите информацию о каждой стране’);

For i:=1 to n Do

  With Count[i] do

  Begin

    Write(’Название: ’);

    ReadLn(Name);

    Write(’Площадь: ’);

    ReadLn(Square);

  End;

{ Сортировка }

For k:=1 to n - 1 Do
For i:=1 to n - k Do

If Count[i].Square > Count[i+1].Square then

Begin

{ Перестановка }

Count1:= Count[i];

Count[i]:= Count[i+1];

Count[i]:= Count1;

End;

Writeln('Характеристики стран');

For i:=1 to n Do

  With Count[i] do

Writeln(Count[i].Name:20, Count[i].Square:8);

Writeln('Работа окончена, нажмите Enter');

Readln;

End.

21. Файлы и наборы данных

Для ЭВМ любая информация: программа, исходные данные и результаты является набором данных (НД).

Набор данных (физический файл) – это совокупность данных, имеющих общее имя и назначение, и существующая на некотором носителе.

Наборы данных различаются по следующим параметрам:

1) типу носителя (бумага, НМД);

2) типу составляющих их элементов (записей);

3) способу организации.

Сведения об этих характеристиках необходимы операционной системе ЭВМ для того, чтобы обеспечить работу с набором.

В программе на Паскале наборам данных соответствуют файлы.

Файл — это логическое понятие, связанное с физически существующим набором данных. Связь обеспечивается операционной системой при выполнении программы пользователя.

Любой файл и соответствующий ему набор данных представляют собой линейную последовательность элементов (называемых, как правило, записью). Число элементов не устанавливается описанием файлового типа. Каждый элемент файла имеет номер. Первый элемент файла имеет номер 0.

Таким образом, запись – основная единица набора. В программе одной записи могут соответствовать:

а) данное скалярного типа (обычно число);

б) массив;

в) данное типа Record.

В Паскале определены три типа файлов:

1) текстовые;

2) типизированные;

3) нетипизированные.

Различие между ними заключается в следующем.

1) Текстовый файл состоит из строк символов. Конец записи определяется концом строки. Чтение и запись в файл осуществляется посимвольно.

2) Типизированный файл может состоять из записей любого типа.
3) Нетипизированные файлы фактически представляют собой каналы ввода/вывода нижнего уровня, используемые в основном для прямого доступа к любому файлу на диске, независимо от его типа и структуры.

Обычно доступ к элементам файла осуществляется последовательно, т.е. когда элемент считывается или записывается, то текущая позиция файла перемещается к следующему по порядку элементу файла. Однако для типизированных и нетипизированных файлов можно организовать прямой доступ к любому элементу с помощью стандартной процедуры Seek (см. далее), которая перемещает текущую позицию файла к заданному элементу.

Таким образом, обработка файла может производиться двумя способами:

1) с использованием последовательного доступа;

2) с использованием прямого (произвольного) доступа.

Когда программа завершает обработку файла, его нужно закрыть. Только после этого связанный с ним внешний набор данных будет обновлен. Затем файловая переменная может быть связана с другим набором данных, или обновленный набор данных может быть связан с другой файловой переменной.

Основными операциями над файлами являются:

1) связь с набором данных;

2) открытие;

3) чтение;

4) запись;

5) закрытие;

6) усечение.

1. Связь файла с набором данных осуществляется с помощью специальной процедуры Assign, которая в общем виде записывается так:

Assign(имя_файла, имя_НД);

Эта процедура присваивает имя внешнего набора данных имя_НД переменной файлового типа имя_файла.

Пример.
Assign(dan,’Isx.txt’);

Assign(datain,’Anketa.dat’);

В операционной системе внешняя аппаратура, такая как клавиатура, принтер, дисплей, рассматриваются как устройства. С точки зрения программиста устройство можно представлять себе как набор данных и с ним можно работать, используя те же процедуры и функции, которые применяются для работы с файлом.

В операционной системе определены следующие устройства:

Con – консоль, посредством которой выводимая информация пересылается на экран дисплея, а вводимая информация воспринимается с клавиатуры;

Lpt1 или Prn – построчное устройство печати.

Кроме того, когда начинается выполнение программы, всегда автоматически открываются два текстовых файла Input и Output. Input (ввод) – это доступный только для чтения файл, связанный с клавиатурой, а Output (вывод) – это доступный только для записи файл, связанный с дисплеем.

2. Процедура Rewrite (имя_файла) создает и открывает новый файл.
Параметр имя_файла является файловой переменной, соответствующей любому типу файла. Процедура Rewrite создает новый набор данных (внешний файл), имя которого присвоено параметру имя_файла процедурой Assign. Если внешний файл с указанным именем уже существует, то он удаляется и на его месте создается новый пустой файл. Текущая позиция в файле устанавливается на начало файла.

Если открывается текстовый файл, то он становится доступным только для записи. 

3. Процедура Reset(имя_файла) открывает существующий файл. Параметр имя_файла является файловой переменной, соответствующей любому типу файла. Если файл уже открыт, то он сначала закрывается, а затем открывается вновь. Текущая позиция файла устанавливается на начало файла.

Если открывается текстовый файл, то он становится доступным только для чтения.

4. Процедура Append(имя_файла) открывает существующий текстовый файл для добавления записей в конец файла. Если файл уже открыт, то он сначала закрывается, а затем открывается вновь. Текущая позиция устанавливается на конец файла. 

5. Чтение из файла осуществляется с помощью известного оператора Read, который в общем виде записывается так:

Read(имя_файла, список);

или

Readln(имя_файла, список);
только для текстового файла!

Пример.

Read (dan,str);

Readln (datain,a,b,c);
6. Запись в файл осуществляется с помощью известного оператора Write, который в общем виде записывается так:

Write(имя_файла, список);

Writeln(имя_файла, список);
- только для текстового файла

Пример.


Write (fl,x,y);
7. Закрытие выполняется процедурой Сlose(имя_файла) для открытого файла.

Параметр имя_файла может соответствовать файлу любого типа, который был предварительно открыт с помощью процедур Reset, ReWrite или Append. В момент закрытия осуществляется полное обновление внешнего набора данных, связанного с файловой переменной имя_файла, после чего связь с набором данных разрывается. Потом файл можно открыть снова.

Пример.


Close(fl);
8. Усечение файла выполняется процедурой Truncate(имя_файла). Параметр имя_файла может соответствовать файлу любого типа. Все записи после текущей позиции в файле имя_файла удаляются и текущая позиция становится концом файла (т.е. функция EoF(имя_файла) принимает значение True).

21.1. Текстовые файлы

Это файлы, состоящие из строк символов, которые заканчиваются символом "конец строки". Текстовый файл определяется описателем Text. В разделе Var это записывается так:

Var

Имя_текстового_файла : Text;

Пример. 

Var

Dan, Datain : Text;
Примером текстовых файлов являются стандартные файлы Input и Output. Причем из файла Input можно только читать, а в файл Output – только писать.

Для чтения и записи в текстовые файлы используются стандартные операторы:

Read,

и

Write,

Readln

и

Writeln.
Общий вид операторов ввода-вывода для текстовых файлов следующий:

Read(файл, запись);

Readln(файл, запись);

- - - -

Writeln(файл, запись);

где файл    — имя файла,

запись — имя данного, которое читается (записывается) из файла. Это может быть символ, строка или число.

Каждый оператор Readln читает очередную запись файла, а Writeln – пишет запись в него. После этого указатель считывания-записи автоматически перемещается на следующую запись.

Основными задачами при работе с файлами являются создание нового и обработка уже существующего файла.

21.1.1. Создание текстового файла

Пример. Ввести в ЭВМ с клавиатуры n строк текста и записать их в текстовый файл. Имя набора данных необходимо также задать с клавиатуры.

Алгоритм

1.1. Ввести Имя_Набора_Данных

1.2. Связать Файл с Набором_Данных

1.3. Открыть Файл для записи

2.1 Ввести количество строк (n)

2.2. Для номера строки (i) от 1 до n выполнить

а) ввести строку;

б) переписать в файл строку

3. Закрыть файл

4. Закончить

Программа для этого алгоритма будет иметь вид

Program SozdText;

Var

Tx : Text; {Имя текстового файла}

FilName : String;{Имя набора данных}

Stroka : String;{Строка текста}

i,n : Integer; {Вспомогательные переменные}

Begin

Writeln(’Введите имя набора данных’);

Readln (FilName);               {п. 1.1}

Assign(Tx,FilName);             {п. 1.2}

Rewrite (Tx);                   {п. 1.3}

Writeln (’Введите количество строк текста’);

Readln(n);                      {п. 2.1}

Writeln (’Введите строки’);

For i:=1 to n do                {п. 2.2}

Begin

Readln (Stroka);    {Ввод строки с клавиатуры}

Writeln(Tx,Stroka);         {Запись в файл}

End;

Close(Tx);                       {п. 3}

End.

21.1.2. Работа с существующим текстовым файлом

Работа с существующим файлом обычно включает в себя операции открытия его для чтения, и затем считывания из него записей, пока не появится признак "Конец файла" – Eof(Файл). Поэтому соответствующие Паскаль-программы содержат циклы вида

While Not Eof (Файл) do

Begin

Readln(Файл, Запись);

- - - - - - - -

End;

Простейшая задача – вывести на экран содержимое файла. Рассмотрим ее на примере набора, созданного в предыдущем пункте.

Алгоритм

1.1. Ввести Имя_Набора_Данных.

1.2. Связать Файл с Набором_Данных.

1.3. Открыть Файл для чтения.

2.1. Количество_строк = 0.

2.2.Пока не конец файла выполнять

2.2.1. Считать из Файла строку;

2.2.2. Количество_строк = Количество_строк +1;

2.2.3. Вывести строку на экран;

2.3. Вывести Количество_строк.

2.4. Закрыть файл.

3. Закончить.

Программа будет иметь вид

Program VivodFile;

Var

Tx : Text;

FilName,Str : String;

KolStr : Integer;

Begin

Writeln(’Введите имя НД ’);

Readln(FilName);                {п. 1.1}

Assign(Tx,FilName);             {п. 1.2}

Reset(Tx,FilName);              {п. 1.3}

KolStr := 0;                    {п. 2.1}

Writeln(’Содержимое набора’);

While Not Eof(Tx) do            {п. 2.2}

Begin

Readln (Tx,Str);

KolStr := KolStr+1;

Writeln(Str); {вывод на экран}

End;

Writeln(’Количество строк в наборе данных - ’, kolstr);

Close(Tx);                      {п. 2.4}

End.

21.2. Типизированные файлы

Характерные особенности типизированных файлов:

1) файл может состоять из записей любого типа;
2) данные передаются из него или в него целыми записями (а не символами или строками)

3) при создании файла записи в нем автоматически нумеруются, причем номер первой из них равен 0.

В разделе Var файл описывается так:

Var

Имя : File Of Тип_Записей;
где Тип_Записей – скалярный (число, символ) или сложный (массив, тип Record и др.).

Пример. 

Type

Anketa = Record

Fam,Im,Ot : String[10];

Numz : Integer;

O : Array[1..5] Of Integer;

End;

Var

FilAnk : file of Anketa;

Dan : File Of Real;

Fl : File Of Char;

Rank : Anketa;

X : Real;

Операции чтения и записи для типизированных файлов записываются в следующем виде:

Read(Файл, Список_ввода);

Write(Файл, Список_вывода);

Пример.
Write(FilAnk,Rank);

21.2.1. Последовательная обработка типизированных файлов

Пример 1. Программа обработки файла, содержащего данные простого типа. При исследовании некоторого технического объекта замерены его параметры Xi и записаны в набор данных StatDan в виде вещественных чисел. Вычислить их среднее значение Mx и стандартное отклонение ( по формулам:

Алгоритм для этой задачи будет следующий.

1. Открыть файл

2. Подсчитать среднее значение Mx и стандартное отклонение (.

3. Закончить.

Уточняем алгоритм.

1.1. Связать файл из программы с набором StatDan.

1.2. Открыть файл для чтения.

2.1. Mx  := 0.

2.2. ( := 0.

2.3. N := 0.

2.4. Пока не конец файла выполнить

2.4.1. Считать из файла число.

2.4.2. Mx  := Mx + число.

2.4.3. ( := ( + х2

2.4.4. N := N + 1.

2.5. Вычислить Mx  и (.

2.6. Вывести Mx  и (.

2.7. Закрыть файл.

3. Закончить.

Программа, реализующая этот алгоритм, может быть такой:

Program Stat;

Var

F : File Of Real;

X,M,S : Real;

N : Integer;

Begin

Assign(F,’StatDan’);{ Здесь имя набора задано в виде}

Reset(F);

      { строковой константы StatDan}

S := 0;

M := 0;

N := 0;

While not Eof(F) do

Begin

Read(F,X);

N := N+1;

M := M+x;

S := S+Sqr(X);

end;

M := M/N;

S := Sqrt(S-Sqr(M))/ N;

Writeln(’Количество измерений - ’,N);

Writeln(’Среднее: ’,M:8:3,’  Отклонение: ’,S:8:3);

end.

Пример 2. Последовательное создание типизированного файла, содержащего данные сложного типа
Как отмечалось выше, к таким данным можно отнести массивы и записи Паскаля. Наиболее типичными являются файлы, содержащие данные типа Record.

Задача. Сформировать и вывести на экран файл, представляющий собой ведомость успеваемости студентов 1 курса ФАИТ. Данные об одном студенте имеют структуру вида:

Фамилия, Имя, Отчество, номер зачетки, оценки по 5 предметам.

Количество студентов (n) и сведения о них вводятся с клавиатуры. Программа для решения поставленной задачи может быть такой.

Program Sozd_Fil;
Const
  M=5;  { Количество оценок }

Type

  Stud=Record

    Fam, Im, Ot : String[10];

           NumZ : LongInt;

            Otm : Array[1..M] Of Integer;

         End;

  Var

     Vedom : File Of Stud;

         A : Stud;

   FilName : String; { имя набора данных }

     j,i,n : Integer;

  Begin
    Writeln(’Введите имя набора данных’);

    Readln(FilName);

    Assign(Vedom,FilName);

    Rewrite(Vedom);    {Открытие файла для записи}

    Writeln(’Введите количество студентов’);

    Readln(n);

    Writeln (’Вводите сведения о каждом студенте’);

    For i:=1 to n do

      Begin

        With A do

          Begin

            Write(’Фамилия: ’);

            ReadLn(Fam);

            Write(’Имя: ’);

            ReadLn(Im);

            Write(’Отчество: ’);

            ReadLn(Ot);

            Write(’Номер зачетки: ’);

            ReadLn(Numz);

            WriteLn(’Введите оценки по одной:’);

            For J := 1 to 5 do

              Begin

                WriteLn(’Введите ’,J,’ -ю оценку’);

                ReadLn(Otm[j]);

              End;

          End;

        Write(Vedom,A); {Запись в файл переписывается целиком}
      End;

    Close(Vedom);

    Writeln (’Конец работы. Нажмите клавишу ENTER’);

    ReadLn;

End.

Пример 3. Последовательная обработка файла Vedom

Задача. Вывести на экран сведения о неуспевающих студентах из файла Vedom предыдущего примера. Соответствующая программа будет иметь следующий вид.

Program Count_2;

Const

M=5;

Dl=10;

Type

Stud = Record

Fam, Im, Ot : String[Dl];

Numz : LongInt;

Otm : Array[1..M] Of Integer;

End;

Var

Vedom : File Of Stud;

FilName : String;{имя набора данных}

A : Stud;

i,j,kdv,K2 : Integer;

Begin

Writeln(’Введите имя набора данных’);

Readln(FilName);

Assign(Vedom,FilName);

Reset (Vedom); {открытие файла, 1-я запись - в буфере}

kdv := 0;

I := 0;

Writeln(’Список фамилий и номеров зачеток неуспевающих студентов’);

While Not Eof(Vedom) Do

Begin

K2 := 0;

Read(Vedom,A);

I := i+1;

With A Do

Begin

For j := 1 To M Do

If Otm[j] < 3 Then

K2 := K2+1;

If K2 > 0 Then

Begin

kdv := kdv+1;

Writeln(Fam:15,Numz:8);

End;

End;{Whith}

End;{While}

Writeln(’Проверено ’,i,’ студентов.’);

Writeln(’Из них неуспевающих ’,kdv,’ человек’);

WriteLn(’Нажмите Enter’);

Readln;

End.

21.2.2. Использование прямого доступа
к записям типизированного файла

Прямой (произвольный) доступ – это способ обработки файла, при котором можно в любой момент выполнять обращение к любой записи файла. Как уже отмечалось, такой способ возможен для файлов всех типов, кроме текстового (т.е. для типизированных и нетипизированных). Напомним также, что в таких файлах каждой записи ставится в соответствие ее порядковый номер. Номера имеют значения 0, 1, 2, … 

При использовании прямого доступа необходимо перед каждой операцией чтения (записи) задать номер записи, которая должна быть обработана. После выполнения операции указатель файла устанавливается перед следующей записью (т.е. перед записью с номером, увеличенным на 1).
Для установки указателя на нужную запись используется специальная процедура Seek, которая в общем виде записывается так:

Seek (Файл, Номер_записи);

Пример. 
Seek(Filank,10);
 – поиск 10–той записи

Seek(f,i);

– поиск i–той записи.

Обычно применяют пары:

Seek



Seek

Read

или

Write
Пример 1. Создание файла с использованием прямого доступа
Иногда встречаются случаи, когда запись в файле может определяться своим порядковым номером. Например, табельный номер служащего или телефонный номер может равняться порядковому номеру записи в файле. В таком случае использование прямого доступа является совершенно естественным. 

Как правило, такой файл рекомендуется предварительно заполнить начальной информацией (нулями, пустыми строками). Затем в него записывается основная информация по номерам записей: по порядку или в разнобой, как это требуется в задаче.

Условие. Составить программу, формирующую телефонный справочник для учрежденческой АТС, имеющей двухзначные номера. Структура записей файла следующая:

№ телефона, Фамилия, Адрес (улица, номер дома и квартиры)

Справочник имеет номера от 00 до 99.

Алгоритм создания такого файла может быть таким

1. Открыть файл для записи.

2. Сформировать "пустую запись".

3. Для номера (k) от 0 до 99 выполнить

переписать "пустую" запись в файл справочника

4. Повторять

4.1. Ввести номер,

4.2. Установить указатель файла на номер

4.3. Ввести Фамилию и Адрес,

4.4. Переписать в файл

Пока не надоест.

5. Закончить.

Уточняем алгоритм.

1. Открыть файл для записи

2. Сформировать "пустую запись":

2.1. Номер (N) = 0

2.2. Фамилия = ’’
{ Пустая строка }

2.3. Адрес = ’’

{ Пустая строка }

3. Для номера (k) от 0 до 99 выполнить

3.1. Установить указатель на номер k

3.2. Переписать запись в файл.

4. Повторять

4.1. Ввести номер,

4.2. Установить указатель файла на номер

4.3. Если не конец файла, то

а) Ввести Фамилию и Адрес,

б) Переписать запись в файл

Иначе

Вывести сообщение "Неправильный номер"

Пока не надоест.

5. Закончить.

Программа для этого алгоритма будет иметь следующий вид.

Program Sprav;

Const

N = 99; 

Type
Tel=Record
{запись справочника}

Nom : Integer;

Fam : String[15];

Adr : String[15]

End;

Var

fTel : file of Tel;
{Телефонный справочник}

Rtl : Tel;


{Одна запись файла}

K : integer;

S : char;

Begin

Assign(ftel,’tsprav’);

Rewrite (ftel);       {п. 1.}

With Rtl do          {п. 2. Создание "пустой" записи}

Begin

Nom := 0;

Fam := ’’;

Adr := ’’;

End;

  {п. 3. Заполнение НД пустыми записями}
For k:=0 to n do  

Begin

Seek(ftel,k);

Write(ftel,Rtl);

end;

     {п. 4. Заполнение Набора Данных информацией}

Repeat

Writeln (’Введите номер’)   { п. 4.1. };

ReadLn (k);

Seek(ftel,k);               { п. 4.2. }

If Not Eof(ftel) then       { п. 4.3. }

Begin

With Rtl do

Begin

Writeln(’Фамилия’);

ReadLn (fam);

Writeln (’Адрес’);

ReadLn(Adr);

Nom := k;

end;

Write(ftel,Rtl);

end
Else
Writeln(’Неправильный номер’);

Writeln(’Продолжаем?-Да/Нет’);

Readln(S);

until (S=’H’) Or (S=’н’) Or(S=’n’) Or(S=’N’);

Writeln (’Конец работы. Нажмите клавишу ENTER’); {п. 5}

Readln;

end.

Пример 2. Работа с файлом: поиск и коррекция записей

Задача 1. Выполнить поиск номера телефона и его владельца по справочнику, сформированному в предыдущем примере. При работе программы должен вводиться номер телефона, и если он есть в справочнике, то на экран должны выдаваться фамилия и адрес его владельца. Если набран неправильный номер, которого в справочнике нет, то необходимо вывести сообщение об этом. Программа будет иметь вид

Program PskTel;

Type

Tel=Record

Nom : Integer;

Fam,Adr : String[15];

End;

Var

Ftel : File Of Tel;

Rtel : Tel;

 Num : Integer;

   S : Char;

Begin

Assign (Ftel,’tsprav’);

Reset (Ftel);      { Открытие файла для чтения}

Repeat   {Поиск фамилии и адреса абонента по номеру}

Write(’Введите номер телефона: ’);

ReadLn(Num);

Seek(Ftel,Num);

If Not Eof (Ftel) Then

Begin

Read (ftel,Rtel);

With Rtel do

Writeln(’Фамилия: ’,fam:15, ’  Адрес: ’,
     Adr:15,   ’  Номер телефона: ’,nom);

End

else

Writeln(’Неправильный номер’);

Writeln (’Продолжаем?-Д/Н, Y/N’);

Readln (S);

Until (S=’H’) Or (S=’н’) Or (S=’n’) Or (S=’N’);

Writeln (’Конец работы. Нажмите клавишу ENTER’);

Readln;

End.

Задача 2. Коррекция телефонного справочника ("Tsprav"), например изменение фамилии абонента. Обычно при коррекции сначала просматривают содержимое записи, а потом, если необходимо, изменяют его. Поскольку после выполнения пары операторов:

Seek(Файл, Номер);

Read(Файл, Запись);

или 

Seek (Файл, Номер);

Write(Файл, Запись);) 

номер записи автоматически увеличивается на 1, перед каждой операцией записи (чтения) необходимо задать номер (выполнить Seek).

При работе программы должен вводиться номер телефона, и если он есть в справочнике, то на экран должны выдаваться фамилия и адрес его владельца. После этого задается вопрос о необходимости коррекции фамилии и в случае утвердительного ответа вводится новая фамилия. Если набран номер, которого в справочнике нет, то необходимо вывести сообщение об этом. 

Программа будет иметь вид.

Program Updtel;

Type

Tel=Record

Nom : Integer;

Fam,Adr : String[15];

End;

Var

Ftel : File Of Tel;

Rtel : Tel;

K,Num : Integer;

S : Char;

Begin

Assign(Ftel,’Tsprav’);

Reset(Ftel);              { Открытие файла }

Repeat {Поиск фамилии и адреса абонента по номеру телефона}

Write(’Введите номер телефона: ’);

ReadLn(k);

Seek(Ftel,k);

If Not Eof(Ftel) then

Begin

Read(Ftel,Rtel);

With Rtel do

Write (’Фамилия: ’,fam,’  Адрес: ’,Adr);

Writeln(’Будем менять фамилию?-Д/Н Y/N’);

Readln(S);

If (S=’Д’) (S=’д’) Or (S=’Y’) Or (S=’y’) then

Begin

Writeln(’Введите фамилию’);

Read (Rtel.fam);

Seek (Ftel,k);

Write(Ftel,Rtel);

end;

End

else

Writeln (’Неправильный номер’);

Writeln (’Продолжаем?-Д/Н Y/N’);

Readln(S);

Until (S=’H’) Or (S=’н’) Or (S=’n’) Or (S=’N’);

Writeln (’Конец работы, нажмите клавишу Enter’);

Readln;

end.

21.2.3. Упорядочение записей в файле

Эта операция выполняется довольно часто. Ее можно произвести, используя известный метод "пузырька". Здесь возможны два подхода. 

1. Записи из файла можно переписать в массив, упорядочить их там, а потом переписать этот массив в исходный файл. 

2. Информация упорядочивается непосредственно в наборе данных. 

При значительном размере файла первый метод потребует большого объема памяти. Затраты оперативной памяти у второго подхода минимальны, но по быстродействию он проигрывает первому.

Далее рассматривается пример, использующий второй подход (упорядочение непосредственно в файле). Любые улучшения алгоритма, позволяющие сократить время решения задачи, должны учитываться. Поэтому в программе операция выполняется до тех пор, пока записи в файле меняются местами (есть перестановки).

Рассмотрим алгоритм на примере, в котором используется более общая по сравнению с телефонным справочником структура записей.

Задача. Набор данных, имя которого задается с клавиатуры, содержит информацию о компьютерах. Запись о компьютере имеет вид

Марка, Цена.

Упорядочить записи в наборе по возрастанию цены.

Алгоритм

1.1. Ввести имя набора данных.

1.2. Связать файл с набором данных.

2.1. Признак = "перестановки" (считаем, что перестановки будут).

2.2. Пока "перестановки" выполнить

2.2.1. Признак = " нет перестановок".

2.2.2. Номер записи (k) = 0.

2.2.3. Открыть файл для чтения.

2.2.4. Пока не конец файла выполнить

а) Установить в файле номер записи k;

б) Считать из файла Запись1 ( k-тую );

в) Если не конец файла, то

1)Считать из файла Запись2 ( k+1-вую );

2) Если Запись1.Цена > Запись2.Цена то

2,а) Установить в файле номер записи k;

2,б) Записать в файл Запись2 ( k+1-вую );

2,в) Записать в файл Запись1 ( k-тую );

2,г) Признак = "перестановка";

д) k = k + 1.

2.2.5. Закрыть файл

3. Закончить.

Программа для этого алгоритма будет иметь следующий вид.

Program Sort_File;

Type

Computer=Record

Marka : String[10];

 Cena : Integer;

End;

Var

FilComp : File of Computer;

Comp1,Comp2 : Computer;

K : Integer;

Perest : Boolean;   { Признак перестановок }

FilName : String;   { Имя набора данных}

Begin

Writeln(’Введите имя набора данных’);

Readln(FilName);

Assign(Filan,FilName);

{ Упорядочение }

Perest := True;{ Перестановки будут }

While Perest do

Begin

Perest := False;   {Перестановок еще не было}

K := 0;

Reset(FilComp);    {Открытие файла}

While Not Eof(FilComp) do

Begin

Seek(FilComp,k);

Read(FilComp,Comp1);

If Not Eof(FilComp) then

Begin

Read(FilComp,Comp2);

If Comp1.Cena>Comp2.Cena then

Begin   {Перестановка начинается}

Seek(FilComp,k);

Write(FilComp,Comp2);

Write(FilComp,Comp1);

Perest:=True; {Перестановка}

End;

End;

k:=k+1;

End;

Close(FilComp);

End;

Writeln(’Конец работы, нажмите клавишу Enter’);

Readln;

End.

21.2.4. Удаление записей из файла

Эта операция может быть двух типов:

а) логическое удаление;

б) физическое удаление.

При логическом удалении запись фактически не удаляется. Вместо этого используется специальный признак: активная/пассивная запись. Логическое удаление – установка признака в состояние "пассивная". При этом запись не стирается. Во время обработки файла проверяется признак, и обрабатываются только активные записи.

Пример. В файле Ftel некоторые номера телефонов сняты. Для логического удаления необходимо добавить в запись Tel еще одно поле, например:

Type

Tel = Record

Nom = Integer;

Fam,Adr : String[15];

Act : Boolean; {Признак активная/пассивная (удаленная)}

end;

При создании файла Ftel поле Act всех записей имеет значение True – запись "активна" (не удалена), т.е. программа при начальном заполнении будет содержать такой фрагмент:

{Начальное заполнение}

Assign (Ftel,’tsprav’);

Rewrite(Ftel);

With Rtel do

Begin

Nom := 0;

Fam := ’’;

Adr := ’’;

Act := True;

end;

For k:=0 to n do

Begin

Seek(ftel,k);

Write(Ftel,Rtel);

end;

Удаление записи с номером Num выполняется как коррекция (изменяется содержимое поля Act на противоположное):

Writeln(’Введите удаляемый номер телефона’);

Read(num);

Seek(Ftel,num);

Read(Ftel,Rtel); {Указатель перешел к следующей записи: num+1}

Rtel.Act := False;

Seek(Ftel,num);      { Вернем указатель к записи num }

Write(Ftel,Rtel);

При обработке файла в этом случае необходима дополнительная проверка наличия (активности) записи:

Seek(Ftel,k);

If Not Eof(Ftel) then

Begin

Read(Ftel,Rtel);

if Rtel.Act then     {Запись не удалена}

Begin

.....{ обработка неудаленной записи }

end;

......

end;

При физическом удалении записи на ее место переписывается новая информация, т.е. запись стирается. Эта операция выполняется по схеме, приведенной на рис. 2.12, где показано удаление записи с номером 1.

Составим процедуру удаления записи с заданным номером Num из файла f. Нужно учесть, что удаление сводится к переписи записей после номера Num на 1 место вперед. После переписи (или если для удаления сразу была назначена последняя запись), файл нужно урезать с помощью процедуры Truncate и таким образом убрать из него последнюю запись. Напомним, что после выполнения операций записи/чтения текущая позиция в файле перемещается к следующему элементу.

Считаем, что в программе описан тип Ft = File Of Zap. В процедуре используется функция FileSize(имя_файла), возвращающая размер файла. Описание этой функции приведено в следующем пункте (21.2.5).

Procedure RemoveF(Num : Integer; Var F : Ft);

{Удаление записи с номером num из файла f}

Var

K : Integer;

R : Zap;

Begin

If num<FileSize(f)-1 then {Удаляемая запись не последняя}

  begin
K := num; {номер записи, в которую будем переписывать данные}

    {Перепись записей на 1 вперед – в начало f}

Seek(f,k+1);  {Следующая после удаляемой запись}

While not eof(f) do

Begin

Read(f,R); 

Seek(f,k);     {Возврат назад }

Write (f,R);

k:=k+1;

Seek (f,k+1);

end;

end;

{Урезаем файл f по последней записи}

Seek(f,FileSize(f)-1);

Truncate(f);

Close(f);

end;

21.2.5. Вставка записей в файл

Эта операция предполагает увеличение размеров набора данных и перемещение признака End Of File. В большинстве случаев новую запись допустимо поместить в конец набора данных, а потом новый набор можно, например, отсортировать. 

Для добавления записей в конец текстового файла служит процедура Append, описанная в начале этого раздела.

Для добавления записей в конец типизированного файла, необходимо установить текущую позицию в файле перед местом, куда будет помещена новая запись. Это удобно сделать, вызвав процедуру Seek следующим образом:

Seek(файл, FileSize(файл)),

где FileSize(файл) – функция, возвращающая текущий размер файла в элементах его записей (он не должен быть текстовым). Другими словами, эта функция возвращает  число, на единицу большее номера последней записи в файле.

Если необходимо добавить запись куда-нибудь в середину типизированного файла, то можно воспользоваться следующим алгоритмом.

1. Ввести номер (Num) записи, после которой должна быть вставлена новая.

2. Переписать все записи файла, начиная с номера (Num+1), на одну позицию к концу файла.

3. Записать в файл новую запись по номеру (Num+1).

4. Установить текущую позицию в файле после последней записи в нем и закрыть файл.

22. Динамическое распределение памяти.
Указатели и списки

Все рассмотренные ранее типы данных называют статическими. Компилятор выделяет для них память до выполнения программы, после анализа описательной части соответствующего блока (программы, процедуры или функции). Выделенная память ассоциируется с соответствующим именем переменной. Статические переменные существуют на протяжении всего времени выполнения блока, в котором они локализованы.

Встречаются задачи, в которых нецелесообразно объявлять объекты на их максимальный размер, так как в большинстве случаев они будут слабо заполнены (например, так называемые разреженные таблицы). Во многих задачах объекты после первого использования становятся ненужными, но если эти объекты статические, то они по-прежнему занимают память, которая могла бы использоваться другими объектами. 

В таких задачах целесообразно выделять память данным в процессе выполнения программы, т.е. динамически.

Динамические переменные – это переменные, которым выделяется память и присваиваются значения непосредственно при выполнении программы.

Для динамического размещения необходимо задавать адреса данных особым образом, т.к. в разделе Var нельзя непосредственно описать соответствующие переменные. Это делается с помощью указателей (ссылок, Pointer). Указатель описывается обычно в разделе Type в виде:

Type

Имя_указателя = ^тип_объекта;
Пример. 

Type
p = ^Char;

q = ^Integer;

Здесь p и q – указатели на объекты символьного и целого типа соответственно. Используя эти типы, можно описать, например, следующие переменные:

Var

pt,r : q; {Ссылки на динамические данные типа Integer}.
Данные типа "указатель" (адреса) нельзя вывести на экран или печать с помощью оператора Write. Например, не допустим оператор Write(pt).
Идентификация данного, связанного с указателем, выполняется так:

Имя_указателя^

Пример.
pt^, r^
- данные типа Integer, адреса которых определяются указателями pt и r. 

К ним применимы операторы Read и Write. Например, Write(pt^) соответствует выводу значения переменной целого типа, на которую указывает pt.
Существует особое значение указателя: Nil ("ничего"). Это значение не указывает ни на какую переменную.

При работе с динамическими переменными используют специальные процедуры.

1) New (указатель); – выделяет в памяти место для новой динамической переменной и устанавливает на нее указатель. Размер выделяемой памяти соответствует размеру того типа, на который указывает указатель.
Пример.   New(pt); – выделено место для одной переменной целого типа и адрес этой переменной находится в pt (говорят, что pt указывает на эту переменную).

Теперь этой переменной может быть задано значение одним из известных способов: чтением или присваиванием, например

New(pt);
Read(pt ^);

или
New(r);

r^ := выражение соответствующего указателю типа.

2) Dispose(указатель); – освобождает место в памяти, занятое переменной, на которую указывает указатель (адрес которой хранится в переменной-указателе). Перед вызовом процедуры указатель должен быть определен (не равен Nil), например, с помощью процедуры New или оператора присваивания.

Пример. 

Dispose (r);
Освобождение места в памяти с помощью процедуры Dispose следует выполнять после того, как соответствующие переменные обработаны и место, занимаемое ими, можно использовать для размещения других данных. После выполнения процедуры Dispose переменная–указатель становится неопределенной.

Рассмотрим пример программы, работающей с динамическими переменными целого типа и использующей описанные выше процедуры.

Program Pointers;

Type

point = ^integer;

Var

i,j : Integer;

p,q : point;

Begin

New(p); {Выделение памяти, на которую будет указывать р}

New(q); {Выделение памяти, на которую будет указывать q}

I := 5;  { Запись 5 на место i }

p^ := 7; { Запись 7 на место, на которое указывает p }

q^ := p^ - i; {Запись 7-5=2 на место, на которое указывает q}

q := p; {Теперь q указывает на тоже место, что и p}

{Объект, адрес которого хранился в q, потерян, стал "мусором"}

End.

При работе с указателями необходимо следить, чтобы связи с объектом не терялись, и не создавался мусор. В последнем случае перед выполнением оператора q := p; надо было освободить память, т.е. воспользоваться процедурой Dispose:
Dispose(q);

q := p;

Рассмотрим еще один пример. Пусть для решения задачи необходимо обработать два массива очень большого размера: массив вещественных чисел, содержащий 10000 элементов, и массив целых чисел, содержащий 15000 значений. Особенностью задачи является то, что массивы обрабатываются последовательно – после обработки первого массива он становится не нужным. 

Program PrimPoint;

Const

  Rmax = 10000;

  Imax = 15000;

Type

   tRe = array [1..RMax] of Real; {Тип для массива вещественных }
   tIn = array [1..IMax] of Integer; {Тип для массива целых }
   pR = ^tRe;          { Указатель на тип массива вещественных }
   pI = ^tIn;           { Указатель на тип массива целых }
Var

   k, i, Max : integer;

   f : pR;     {Переменная-Указатель на тип массива вещественных }
   g : pI;       { Переменная-Указатель на тип массива целых }
Begin
   New(f); {Выделяется блок памяти для хранения всего массива вещественных}

{Переменная-Указатель f указывает на начало этого блока }

{f^ – означает весь массив, f^[i] – означает i-ый элемент массива}

   For i := 1 to Rmax do

     begin
       Read(f^[i]);       {Читаем i-ый элемент массива }
         { Здесь выполняется обработка массива вещественных чисел }

     end;

   Dispose(f);{ Освобождаем память, занимаемую массивом вещественных }

   New(g); {Выделяется блок памяти для хранения всего массива целых}

{ Переменная-Указатель g указывает на начало этого блока }

{g^ – означает весь массив, g^[i] – означает i-ый элемент массива}

   Read(g^[1]);         {Читаем i-ый элемент массива }
   Max := g^[1];

   For i := 2 to Imax do

     Begin

       Read(g^[i]);

       If g^[i] > Max then

         Max := g^[i];

     End;

   WriteLn(Max);

End.

22.1. Использование указателей. Списки

Список представляет собой совокупность динамических объектов одного типа, упорядоченных с помощью ссылок. 

Объекты называются элементами списка. Элементом списка обычно является запись, которая содержит, по крайней мере, два поля (рис. 2.13):

- информационное;

- указатель.

Сам список представляется в виде рис. 2.14.

Каждый элемент списка имеет указатель на соседа. У последнего элемента указатель – Nil. Элементы списка и указатели описываются так:

Type

Указатель = ^Эл_Списка; {Опережающее описание }

{допускается только в этом случае }

Эл_Списка = Record

Инф_поле : Тип_Инф_поля;   

Поле_указателя : Указатель; {Этот тип уже определен выше}
End;

Тип информационного элемента - любой (скалярный, строка, массив и т. д.).

Пример.      Тype

Ptr = ^El;

El = Record

      Buk : Char; 

     Ukaz : Ptr;

     End;

Var

tUk,PrUk,NUk,q : Ptr; {Указатели на элемент списка}
     ElSp : El;   { Элемент списка (запись типа El) }
С помощью списков достаточно просто представляются очереди.

Очередь – это понятие, которое широко используется в вычислительной технике. Список позволяет моделировать операции добавления элементов в очередь и выборки из нее.

Основные операции над списками:

1) переход от одного элемента к другому (следующему);

2) включение нового элемента в список;

3) удаление элемента из списка. 

1) Первая операция выполняется просто: значению переменной-указателю, которая сейчас указывает на некоторый элемент, необходимо присвоить значение, находящееся в поле указателя этого элемента (см. рис. 2.15):

tUk := tUk^.Ukaz
В результате tUk станет ссылкой на следующий элемент или Nil, если элемент последний. 

2) Удаление элемента из списка выполняется наиболее просто, если имеется ссылка (указатель) на элемент, предшествующий удаляемому. При этом сначала переносятся ссылки, а потом удаляется элемент (освобождается занимаемая им динамическая память). Ссылка на удаляемый элемент не должна теряться, пока он не будет уничтожен или включен в другой список, иначе из него получится "мусор". 

Процесс удаления элемента, расположенного в списке после элемента, на который указывает ссылка PrUk, иллюстрируется рис. 2.16, а фрагмент соответствующей программы может иметь вид, приведенный далее.

{Запомним указатель на удаляемый элемент}

tUk := PrUk^.Ukaz;

{Полю указателя элемента, стоящего перед удаляемым, присвоим}

{ значение поля указателя удаляемого элемента }
PrUk^.Ukaz := PrUk^.Ukaz^.Ukaz; 

На этом удаление элемента из списка закончено: связь этого элемента с предыдущим разорвана. Но элемент продолжает занимать память и для его физического удаления нужно выполнить процедуру:

Dispose (tUk);


{Физически удалили элемент }
Если нет необходимости освобождать память, то иногда полезно записать в поле указателя удаленного элемента Nil:
tUk^.Ukaz := Nil.
3) Включение в список элемента, на который имеется указатель NUk, после элемента, на который имеется указатель PrUk, иллюстрирует рис. 2.17. Для выполнения этой операции нужно задать ссылку из вставляемого элемента на следующий (она равна ссылке из предыдущего элемента на следующий – см. рис. 2.17, а), а затем – изменить ссылку из предыдущего элемента на вставляемый (см. рис. 2.17, б). Существующая связь разрывается последней, чтобы не потерять элемент.

Фрагмент программы будет иметь вид

New(NUk);        {Создали вставляемый элемент}

Readln(NUk^.Buk);

Nuk^.Ukaz:=PrUk^.Ukaz; {Ссылка из вставляемого на следующий}

PrUk^.Ukaz:= Nuk; {Ссылка из предыдущего на вставляемый}

22.2. Очереди

Наиболее простые способы формирования и обслуживания очередей (дисциплины обслуживания):

1) LIFO — Last Input – First Output («Последним пришел - Первым вышел», т.е. обслуживание в порядке, обратном поступлению);

2) FIFO — First Input – First Output («Первым пришел - Первым вышел», т.е. обслуживание в порядке поступления).

Рассмотрим механизм формирования и выборки очередей на примере списка, в которых информационный элемент имеет тип Char.

22.2.1. Очередь типа LIFO

Это – наиболее простая очередь. Очередной элемент добавляется (или убирается) только в начало очереди.

Для создания и управления очередью типа LIFO достаточно иметь указатель на  начало очереди и указатель на текущий элемент очереди. На бытовом уровне – это странная очередь, а в вычислительной технике она отображает работу стека. Новый объект (на него указывает текущий указатель) настраивает свой указатель на первый элемент существующей очереди, а указатель на первый элемент обновляемой очереди - настраивает на себя. Таким образом, пришедший последним – в очереди оказывается и выбирается из нее первым.

Рассмотрим алгоритм формирования очереди LIFO, считая, что элементы очереди (символы) вводятся с клавиатуры. В конце набирается символ ’*’. Он может быть таким.

1. Задать начальный элемент очереди.

2. Пока Информационный_элемент не равен ’*’ выполнить

Ввести очередной Информационный_элемент и

Поместить его в очередь.

3. Вывести Элементы очереди.

4. Закончить.

Уточним алгоритм, приняв во внимание, что в очереди типа LIFO добавление и выборка элементов осуществляется только в одном месте (в начале очереди), и получим.
1.1. Указатель_из_элемента_очереди = Nil (последний элемент).

1.2. Ввести Информационный_элемент.

2. Пока Информационный_элемент не равен ’*’ выполнить

2.1. Выделить память для Текущего_элемента.

2.2. Задать Информационный_элемент.

2.3. Связать Текущий_элемент со Следующим («хвостом» очереди).

2.4. Ввести Информационный_элемент.

3.1. Указатель_Текущего_элемента = Указатель_Следующего (начального элемента).

3.2. Пока Указатель_Текущего_элемента не Nil выполнить

3.2.1. Вывести Текущий_информационный_элемент;

3.2.2. Перейти к Следующему_элементу_очереди.

4. Закончить.

Программа для этого алгоритма будет иметь вид

Program Lifo;

Type

Ptr = ^El;

 El = Record

Buk : Char;  { Информационное поле }

 Sl : Ptr;   { Поле указателя }

End;

Var

Fp,Tp : Ptr;
    { Указатель на начало и текущий }

    C : Char;

Begin

{ Начало формирования очереди }

Fp := Nil;
{Указатель на начало очереди. Очередь пустая}

Writeln(’Введите элемент очереди’);

Readln(C);

{ Формирование очереди }

While C <> ’*’do

Begin

New(Tp); {Выделение памяти для нового элемента}

Tp^.Buk := C; {Запись информационной компоненты}

Tp^.Sl := Fp; {Запись в поле указателя нового элемента}

{ссылки на начало очереди – связь с остальной очередью}

Fp := Tp;     { Теперь Fp, как и Tp, указывает }

              { на начало очереди }

Writeln(’ Введите элемент очереди’);

Readln(C);

End;

      {В этом месте Fp и Tp - ссылки на начало очереди}

      { Вывод очереди с ее начала (головы) }

WriteLn(’Вот такая очередь:’);

While Tp <> Nil do

Begin

Write (Tp^.Buk:2); {Вывод информационного поля элемента}
Tp := Tp^.Sl;      { Переход к следующему элементу }
End;

WriteLn;

Writeln(’Конец работы. Нажмите ENTER’);

Readln;

End.

На рис. 2.18 показана последовательность формирования очереди LIFO на примере двух элементов в соответствии с текстом программы. 

22.2.2. Очередь FIFO

Для очередей типа FIFO вводят понятия начального ("головы") и последнего ("хвоста") элемента. Добавляемые элементы помещаются в "хвост" очереди, а выборка осуществляется с «головы». Таким образом, для создания этой очереди и выборки из нее нужно иметь указатели на начало, конец и текущий элемент.

Общий алгоритм формирования работы с такой очередью может быть таким же, как для LIFO.

1. Задать начальный элемент очереди.

2. Пока Информационный_элемент не равен ’*’ выполнить

Ввести очередной Информационный_элемент и

Поместить его в очередь.

3. Вывести Элементы очереди.

4. Закончить.

Уточненный алгоритм можно получить, приняв во внимание способ формирования и выборки из очереди типа FIFO, описанный выше.

1.1. Указатель_из_первого_элемента (головы) = Nil.

1.2. Ввести Информационный_элемент.

2. Пока Информационный_элемент не равен ’*’ выполнить

2.1. Выделить память для Текущего_элемента (Tp).

2.2. Задать Информационный_элемент.

2.3. Записать в «хвост» очереди Nil.

2.4. Если Элемент_очереди – первый (головной), то

Запомнить его адрес (Fp)

Иначе

Связать Текущий_элемент с «хвостом» очереди.

2.5. Передвинуть «хвост» очереди (Lp) на один элемент.

2.6. Ввести Информационный_элемент_очереди.

3.1. Указатель_Текущего_элемента = Указатель_Первого_элемента очереди.

3.2. Пока Указатель_Текущего_элемента не Nil выполнить

3.2.1. Вывести Текущий_информационный_элемент;

3.2.2. Перейти к следующему Элементу_очереди.

4. Закончить.

Соответствующая программа будет иметь вид

Program Fifo;

Type

Ptr = ^El;

 El = Record

Buk : Char;

Sl : Ptr;

End;

Var

Fp,Lp,Tp : Ptr; {Указатели на первый, последний и текущий элемент}
      C : Char;

Begin

{Формирование очереди}

Fp := Nil;
          { Очередь пустая}

Writeln (’Введите элемент очереди’);

Readln(C);

While C <> ’*’ do

Begin

New(Tp);                  {Выделение памяти для очередного элемента}
Tp^.Buk := C;     {Заполнение информационного поля}
Tp^.Sl := Nil;    {Очередной элемент – последний в очереди}

If Fp = Nil then   {Если в очереди только один элемент}
Fp := Tp
        {Указатель на "голову" очереди}
Else

Lp^.Sl := Tp; {Ссылка из предпоследнего элемента на последний}
Lp := Tp;
    {Указатель на "хвост" очереди}
Writeln (’Введите элемент очереди’);

Readln(C);

End;

{ Просмотр очереди с "головы" }

Tp := Fp; {Tекущий указатель теперь указывает на первый элемент}

WriteLn(’Вот такая очередь:’);

While Tp <> Nil do

Begin

Write(Tp^.Buk:2);   { Вывод информационного поля элемента }

Tp := Tp^.Sl;      { Переход к следующему элементу }
End;

WriteLn;

Writeln(’Конец работы. Нажмите ENTER’);

Readln;

End.

На рис. 2.19 показана соответствующая тексту этой программы последовательность создания очереди FIFO на примере двух элементов.

23. Стандартные модули Турбо Паскаля

Библиотека Турбо Паскаля содержит большое количество стандартных процедур, функций, констант, типов и переменных, которые разделены на связанные группы, называемые стандартными модулями. Они хранятся в файле Turbo.tpl. Каждый из модулей содержит описания констант, типов, переменных, процедур и функций, необходимых для реализации функционального назначения модуля. Наиболее распространенными являются следующие модули.

1. System – содержит все стандартные функции и процедуры Турбо Паскаля (математические, тригонометрические, обработки строк, процедуры Read и Write и т.д.). Присоединяется ко всем программам автоматически при их загрузке и выполнении.

2. Printer – связывает программу с устройством печати, используется совместно с модулем Crt.

3. Crt – позволяет максимально использовать возможности клавиатуры, экрана и звукового генератора.

4. Dos – обеспечивает связь программы с MS DOS.

5. Graph – позволяет использовать графические возможности компьютера.

Для обеспечения возможности работы программы с любым модулем необходимо после заголовка указать его имя в предложении Uses. В общем виде это предложение записывается так:

Uses

Список_имен_модулей;

Пример. 
Uses

Crt, Dos, Graph;

23.1. Модуль Crt

Этот модуль облегчает работу с клавиатурой и экраном, позволяет создавать окна, задавать цвет текста и фона для цветного дисплея. Содержит около 20 процедур и функций. В зависимости от типа дисплея и драйвера (программы, обеспечивающей связь компьютера с дисплеем) поддерживает до 16 цветовых текстовых режимов.

Замечание. При исполнении на компьютерах с процессорами последних разработок модуль Crt работает некорректно, выдавая ошибку: 200 Division by zero. В Интернете можно найти комплект программ, исправляющих этот недостаток. Такой комплект CrtDevice можно переписать в лаборатории кафедры вычислительной техники.

Рассмотрим наиболее часто употребляемые процедуры и функции модуля.

23.1.1. Работа с клавиатурой и звуком

1) ReadKey - функция, возвращающая символ нажатой клавиши без вывода его на экран. Тип результата – Char. Аргумента у функции нет. Специальные клавиши на клавиатуре (функциональные F1 – F12, управления курсором, Alt, Ctrl и т.д.) генерируют расширенные коды. При нажатии такой клавиши функция ReadKey возвращает сначала нулевой, а затем - расширенный код клавиши.

Пример.


S := ReadKey;

Приводимый ниже фрагмент программы показывает, как можно применить функцию для организации итерационных циклов.

Repeat

 .  .  .

Writeln(’Продолжаем?- Да/Нет (Д/Н)’);

S := ReadKey;

Until (S=’H’) OR (S=’н’) ;

2) KeyPressed -функция без аргумента, возвращает значение True (истина), если клавиша нажата, и False (ложь) в противном случае. Символ, соответствующий нажатой клавише, остается в буфере клавиатуры. Функция не распознает клавиш перевода регистра (Shift, Alt, NumLock и т.д.). Может использоваться для организации итерационных циклов следующего вида

Repeat

- - - - -

Until KeyPressed;

Цикл заканчивается, если нажата любая клавиша. Например, текст программы может завершаться следующим фрагментом:

WriteLn(’Работа окончена. Нажмите любую клавишу ...’);
Repeat Until KeyPressed;

End.

3) Delay(задержка : word); – процедура, которая выполняет задержку на число миллисекунд, задаваемых параметром задержка. 

4) Sound(частота : word); – процедура, которая включает внутренний излучатель звука, который будет издавать звуковой сигнал с частотой, заданной параметром частота. 

5) NoSound выключает внутренний излучатель звука.

Внимание! Звуковой сигнал, включенный процедурой Sound, будет звучать до тех пор, пока его не выключить обращением к процедуре NoSound.

Если в программе часто используются звуковые эффекты, то целесообразно написать процедуру, которая обеспечит излучение звука заданной частоты в течение заданного промежутка времени. Эта процедура может быть такой.

Procedure Zvuk(freq, long : word);

Begin

  Sound(freq);

  Delay(long);

  NoSound;

End;  { Zvuk }

23.1.2. Управление цветом

Если используется цветной графический дисплей с адаптером EGA (Enhansed Graphic Adapter), VGA (Video Graphic Array) и др., то можно задавать и изменять цвет текста и фона с помощью следующих процедур:

1) TextBackGround (Color:Byte); 
– задает цвет фона.

2) TextColor(Color:Byte); 

– задает цвет текста.

Параметр Color (цвет) может задаваться выражением целого типа, которое должно принимать значения в диапазоне 0..15, или одной из определенных в модуле Crt констант цветов (название цвета на английском языке). Константы для фонового и основного цвета и их значения приведены ниже:




Black      = 0 (черный)
DarkGray      = 8 (темно-серый)

Blue       = 1 (синий)
LightBlue     = 9 (светло-синий)

Green      = 2 (зеленый)
LightGreen    = 10 (светло-зеленый)

Cyan       = 3 (бирюзовый)
LightCyan     = 11 (светло-голубой)

Red        = 4 (красный)
LightRed      = 12 (розовый)

Magenta    = 5 (малиновый)
LightMagenta = 13 (светло-малиновый)

Brown      = 6 (коричневый)
Yellow        = 14 (желтый)

LightGray  = 7 (светло-серый)
White         = 15 (белый)

Цвет текста может быть любой, а цвет фона – один из восьми первых. 

Вместо того чтобы устанавливать раздельно цвет текста и цвет фона указанными выше процедурами, можно воспользоваться определенной в модуле Crt переменной TextAttr, длиной в байт. Ей можно присвоить значение, которое сразу определит текстовые атрибуты. Младшие 4 бита у этой переменной кодируют цвет текста (0..15), следующие 3 бита – цвет фона (0..7), а самый старший (левый) бит позволяет сделать текст мерцающим. Константа мерцания имеет имя Blink, а значение – 128.Для размещения цвета фона в правильных позициях битов, его значение нужно умножить на 16. Например, следующее присваивание:

TextAttr := Yellow + Blue*16 + Blink

Обеспечит вывод желтых мерцающих символов на голубом фоне.

3) TextMode (Mode:Integer); 
– процедура; задает указанный текстовый режим (mode). Приведем некоторые определенные в модуле Crt константы (слева) для выбора текстовых режимов и их значения (справа):

BW80 = 2;– черно-белый режим с разрешением экрана 80х25 символов;
 C80 = 3; – цветной режим с разрешением экрана 80х25 символов;

Font8*8 = 256; – цветной режим с разрешением экрана 80х50 символов;

Более подробное описание этих и других режимов для соответствующих адаптеров можно найти в справочниках. 

Предыдущий цветовой режим сохраняется в памяти в переменной LastМode.

23.1.3. Создание окон и позиционирование курсора

1) Window(X1,Y1,X2,Y2:byte); – процедура, которая определяет на экране текстовое окно. Параметры (X1,Y1) представляют собой координаты верхнего левого угла окна, а параметры (X2,Y2) – нижнего правого. 

Это окно не имеет визуально наблюдаемых границ. Минимальный размер текстового окна – один столбец на одну строку. 

Если передаваемые координаты окажутся недопустимыми, то обращение к процедуре игнорируется. По умолчанию при запуске программы определяется окно (1,1,80,25), что соответствует всему экрану.

Пример. 
Window (1,1,80,25);
– окно во весь экран

Window (1,24,80,24);
– одна строка (24-ая) внизу экрана.

Если окно открыто, то все операторы Write осуществляют вывод в него. Доступ к экрану вне окна закрыт. Доходя до правого края окна, курсор переходит на начало следующей строки. Если курсор достигает конца последней строки окна, информация продвигается вверх на одну строку (происходит так называемый скроллинг), а курсор устанавливается в начале последней строки (очищенной в результате скроллинга).

В один и тот же момент времени программа может работать только с одним (активным) окном. Это то окно, которое было открыто последним вызовом процедуры Window. Если ранее было открыто другое окно, оно автоматически закрывается.

После создания окна процедурой Window все координаты, задаваемые в других процедурах и функциях, рассматриваются как координаты, отсчитываемые в этом окне. Другими словами, все координаты экрана, кроме координат самого окна, являются координатами внутри данного окна. Например, GoToXY(1,1) всегда устанавливает курсор в верхний левый угол текущего окна.

Пример. 
Window (1,1,80,1);

– окном стала верхняя строка экрана.

Координата X теперь может принимать значения от 1 до 80, а Y - только 1.

Window(20,1,50,6);

Теперь X может принимать значения от 1 до 30, а Y – от 1 до 6. Это нужно учитывать, например, при задании координат в процедуре GoToXY.

Окна можно использовать при необходимости вносить изменения только в некоторой замкнутой области экрана или независимо выводить несколько результатов (две, три, четыре таблицы в соответствующих окнах).

2) ClrScr – процедура; очищает экран (точнее, последнее открытое окно), и устанавливает курсор в левый верхний угол экрана (окна). Обычно выполняется перед выводом данных на экран. Очистка заключается в окрашивании экрана фоновым цветом.

Пример.

TextBackGround(Blue); {Назначен синий цвет для фона }

ClrScr;    { Весь экран окрашивается в синий цвет }

Window(1,1,60,20);    { Назначено новое окно }

TextBackGround(LightGray); { Цвет фона }

ClrScr;{ Прямоугольная область экрана с координатами }

{ (1,1,60,20) окрасится в светлосерый цвет    }

3) GoToXY(X,Y : Byte); – процедура, которая устанавливает курсор внутри текущего окна в позиции с координатами (X, Y); (X задает столбец, а Y – строку. Координаты крайних точек экрана приведены на рис. 2.20. Имеются две функции, позволяющие узнать текущее положение курсора на экране.

4) WhereX – возвращает координату X для текущей позиции курсора в текущем окне. Тип результа: byte. Аргументов – нет.

Пример. X := WhereX + 5;

5) WhereY - возвращает координату Y для текущей позиции курсора в текущем окне. Тип результа: byte. Аргументов – нет.

Пример. Если от текущего положения курсора необходимо перейти на 2 позиции вправо и на 4 позиции вверх, то следует записать:

GoToXY(WhereX+2, WhereY-4);

Используя процедуру GoToXY, можно выводить на экран графики, заполнять и перемещаться по ячейкам таблиц исходных данных и т.д.

При построении таблиц используются символы псевдографики: (, (, (, (, (, ( и т.д. Обычно эти символы задаются непосредственно с помощью их кода. Для такого представления сначала записывается знак #, а затем – десятичный код символа. Коды символов на ПЭВМ представляются в стандарте ASCII. Их можно найти в любой литературе по персональным компьютерам или программированию.

Наиболее распространенные символы псевдографики и их коды приведены ниже.

Одинарные линии



Двойные линии

#218   #194   #191   #179

 ┌      ┬       ┐    (
 #195   #197   #180   #196

 ├      ┼       ┤     — 
 #192   #193   #217

 └      ┴       ┘ 
#201   #203   #191   #186

 ╔      ╦       ╗    ║
 #204   #206   #185   #205

 ╠      ╬       ╣    ═ 
 #200   #202   #188

 ╚      ╩       ╝ 

Для набора этих символов используются альтернативные цифровые клавиши, расположенные в правой части клавиатуры. При вводе некоторого символа нажимают клавишу Alt и удерживают ее до окончания набора кода на альтернативной клавиатуре.

23.1.4. Построение графика в текстовом режиме

Графики, выводимые на экран в текстовом режиме, не отличаются высоким качеством из-за невысокой разрешающей способности экрана в этом режиме. Тем не менее во многих случаях их применение вполне оправдано.

Задача. Написать программу построения графика заданной функции Y=f(X) для аргумента X, изменяющегося от Xmin до Xmax. Для определенности будем считать, что Y=sin(x), Xmin =–2π, Xmax = 2π.

Если известны минимальное и максимальное значение функции, то областью, в которой располагается реальный (натурный) график, является прямоугольник с координатами (при общепринятом направлении осей):

левый верхний угол – (Xmin, Ymax);

правый нижний угол – (Xmax, Ymin).

На экране для размещения графика тоже должна быть выделена прямоугольная область, координатами которой (с учетом выбранного направления оси Y вниз) будут:

левый верхний угол – (xGmin, yGmin);

правый нижний угол – (xGmax, yGmax),

здесь обозначено:

xGmin – x графическое минимальное,

xGmax – x графическое максимальное,

yGmin – y графическое минимальное,

yGmax – y графическое максимальное.

В общем случае "область реального существования" функции должна быть отражена на "область ее экранного существования". Для этого должны быть вычислены масштабы – коэффициенты, с помощью которых натурные величины (X, Y) будут переводиться в графические координаты (xG, yG).

Масштабы можно определить по следующим формулам:

     Mx := (xGmax-xGmin)/(Xmax-Xmin);  { Масштаб по X }

     My := (yGmax-yGmin)/(Ymax-Ymin);  { Масштаб по Y }

которые получены из условия, что стороны "экранного" прямоугольника точно будут соответствовать сторонам "реального" прямоугольника.

Максимальное и минимальное значения функции (хотя бы приблизительные, лучше с запасом в большую по абсолютной величине сторону) должны быть определены пользователем.

Для перехода к графическим координатам получим следующие соотношения:

          xG:=xGmin+Round((x-Xmin)*mx);

          yG:=yGmin+Round((Ymax-y)*my);

Здесь функция Round(аргумент) используется для получения целого значения с правильным округлением, а "несимметричность" ее аргумента вызвана тем, что на экране ось Y направлена вниз.

Используя эти формулы, можно провести оси X и Y на графике: оси X соответствует y=0 во второй формуле, а оси Y – x=0 в первой формуле. Аналогично, можно легко нанести на график сетку линий: для горизонтальных линий задают необходимые значения Y во второй формуле, а для вертикальных – значения X в первой.

Приводимая ниже программа не является универсальной, но она демонстрирует приемы, которые могут быть использованы при написании более качественных продуктов.

Для проведения горизонтальных и вертикальных линий в программе используются процедуры, для вычисления заданной зависимости используется функция. Все необходимые действия выполняются последовательно и автономно. При необходимости любое количество используемых констант может быть заменено переменными, значения которых будут запрашиваться у пользователя в режиме диалога. В операторах While условие окончания записано так, чтобы гарантированно выводилась последняя линия или точка. Обратите внимание, что если назначены пределы изменения по оси X на экране (xGmin, xGmax), то количество точек на этой оси будет (xGmax–xGmin+1), и количество интервалов (шагов) нет смысла назначать более (xGmax–xGmin). Константы, определяющие размеры графика на экране (xGmin, xGmax, yGmin, yGmax), назначены с учетом возможности сделать необходимые надписи вне поля графика.

Program PlotGraf;

uses 

crt;

const
  Xmin=-pi; { Начальное и }

  Xmax= pi; { конечное значения аргумента}

  xGmin=5;  { Начальное и }

  xGmax=75; {конечное значение координат экрана по оси X}

  Ymin=-1;  { Минимальное и }

  Ymax=1; {максимальное значения функции (предполагаемое)}

  yGmin=5;  { Начальное и }

  yGmax=45; {конечное значение координат экрана по оси Y}

  kgl=11;   { Количество проводимых горизонтальных и }

  kvl=11;   { вертикальных линий }

  kix=xGmax-xGmin; { Количество интервалов по оси X }

var

   x,  y,           { Текущие значения X и Y }

  mx, my,           { Масштабы по X и Y      }

  hx, hy : real;    { Шаги по X и Y }

  xg, yg : integer; {Значения графических координат X и Y}

  Function myFunc(x:real):real;

  begin

    myFunc:=sin(x); { Функция для построения графика }

  end; { myFunc }

  Procedure ver_Line(x,yn,yk:integer; sim:char);

     { Строит вертикальную линию, проходящую через X }

     { от yn до yk, используя символ sim       }

  var

    y : integer;

  Begin

      For y:= yn to yk do

        begin

          GoToXY(x,y);

          Write(sim);

        end;

  End;  { ver_line }

  Procedure Gor_Line(y,xn,xk:integer; sim:char);

     { Строит горизонтальную линию, проходящую через Y }

     { от xn до xk, используя символ sim       }

  var

    x: integer;

  Begin

      GoToXY(xn,y);

      For x:= xn to xk do

          Write(sim);

  End;  { Gor_line }

begin

  TextMode(c80+256);   { Цветной режим с 50 строками }

  TextBackGround(White);

  ClrScr;              { Фон - белый }

  mx:=(xgmax-xgmin)/(xmax-xmin);  { Масштаб по X }

  my:=(ygmax-ygmin)/(ymax-ymin);  { Масштаб по Y }

               { Проведение горизонтальных линий }

  y:=ymin;

  hy:=(ymax-ymin)/(kgl-1); { Шаг между линиями }

  TextColor(Yellow);  {Желтый цвет горизонтальных линий }

  while y<=ymax+0.1*hy do

    begin

      yg:=ygmin+Round((ymax-y)*my);{Графическая координата по Y}

      Gor_Line(yg,xgmin,xgmax,’-’);

      GoToXY(xgmin-4,yg); { Курсор в нужное место и    }

      write(y:4:2);       { вывод значения около линии }

      y:=y+hy;            { Очередное значение Y }

    end;

         { Проведение оси X (на ней y=0) }

  yg:=ygmin+Round(ymax*my); {Графическая координата по Y}

  TextColor(Blue);

  Gor_Line(yg,xgmin,xgmax,'-');

  Write(’->X’);              { Вывод обозначения оси X  }

      { Проведение вертикальных линий }

  x:=xmin;

  hx:=(xmax-xmin)/(kvl-1); { Шаг между линиями }

  TextColor(Red);   { Зеленый цвет горизонтальных линий }

  while x<=xmax+0.1*hx do

    begin

      xg:=xgmin+Round((x-xmin)*mx);

      Ver_Line(xg,ygmin,ygmax,’:’);

      GoToXY(xg-2,ygmin-2); {Курсор в нужное место и }

      write(x:4:2);         {вывод значения около линии }

      x:=x+hx;              { Очередное значение Y }

    end;

      { Проведение оси Y (на ней x=0) }

  xg:=xgmin-Round(xmin*mx); {Графическая координата по X}

  TextColor(Blue);

  Ver_Line(xg,ygmin,ygmax,’!’);

  GoToXY(xg,ygmin-1); { Курсор в нужное место и  }

  Write(’^Y’);        { вывод обозначения оси Y  }

       { Построение графика }

  x:=xmin;

  hx:=(xmax-xmin)/kix;              { Шаг по X }

  while x<=xmax+0.1*hy do

    begin
      xg:=xgmin+Round((x-xmin)*mx);{Графическая координата по X}

      y:=myFunc(x);                 { Вычисление y }

      yg:=ygmin+Round((ymax-y)*my);{Графическая координата по Y}

      GoToXY(xg,yg);            { Курсор в нужное место }

      TextColor(Black);

      Write(’*’);                { Вывод точки графика }

      x:=x+hx;

    end;

    GoToXy(10,48);

    Write(’Нажмите любую клавишу...’);

    Repeat Until KeyPressed;

  end.

Процедуры задания цвета фона и текста широко используются для представления различных меню. Рассмотрим пример программы, которая реализует меню с использованием средств модуля Crt. 

23.1.5. Алгоритм и программа представления меню
средствами Турбо Паскаля

Реализация меню предполагает выполнение следующих операций:

1) вывод пунктов меню основной парой цветов (фона и текста);

2) вывод выбранного пункта меню другой парой цветов (более яркой);

3) передвижение по меню с помощью традиционных управляющих клавиш;

4) выбор пункта меню для выполнения соответствующей операции;

5) выход из меню.

Три последние операции выполняются с использованием управляющих клавиш, которые имеют следующие коды:

#13 – Enter;

#27 – Esc;

#77 – перемещение курсора влево;

#75 – перемещение курсора вправо;

#72 – перемещение курсора вверх;

#80 – перемещение курсора вниз.

Первым двум кодам соответствует один байт (символ), а остальным – два, причем первый байт у них имеет нулевой код. 

Рассмотрим алгоритм и программу на примере горизонтального меню, содержащего не более 5 пунктов. Выбор пункта осуществляется нажатием клавиши Enter, а выход из него и из программы – нажатием клавиши Esc. После выбора пункта для простоты выдается сообщение о том, что он выполняется. Названия пунктов соответствуют названиям операций.

Алгоритм

1.1. Задать пункты меню.

1.2. Очистить экран.

1.3. Начальное положение курсора (пункт меню) pm = 1.

2. Повторять              {движение по меню}

2.1. Повторять

2.1.1. Вывести все пункты меню основной парой цветов.

2.1.2. Вывести текущий пункт меню pm другой парой цветов.

2.1.3. Считать символ с клавиатуры             {действия пользователя}

2.1.4. Если символ = #0, то                      {нажата управляющая клавиша}

а) Считать символ                {читаем второй код нажатой клавиши}

б) Выбрать символ из значений

#77 : Движение вправо

#75 : Движение влево

Пока Символ не будет равен Esc или Enter.

2.2. Если Символ равен Enter то

Выбрать pm из значений

1 : Вывести: ’Выполняется пункт 1’
2 : Вывести: ’Выполняется пункт 2’
3 : Вывести: ’Выполняется пункт 3’
4 : Вывести: ’Выполняется пункт 4’
5 : Закончить (Символ = Esc)

Пока Символ не будет равен Esc.

3. Закончить

Движение вправо и влево должно быть круговым, т.е. при движении вправо номер пункта увеличивается на 1, пока он не станет равен последнему номеру. После этого номер должен принять значение 1. При движении влево после первого пункта номер должен стать последним.

В программе необходимо задать координаты первого пункта. Если принять названия всех пунктов меню одинаковыми, то начало пункта с номером pm на экране будет иметь координаты:

(xn + (pm - 1)*dl, yn),

где   xn и yn – координаты начала первого пункта меню;

dl - длина названия пункта меню (одинаковая для всех пунктов).

Программа для описанного выше алгоритма имеет вид.

Program Gor_Menu;

Uses
Crt;

Const

Dl = 12;  {длина названия пункта меню - шаг перемещения}

 M = 5;      { количество пунктов }

Enter = #13;

Esc = #27;

Var

Punkt : Array[1..m] of String[dl]; { Массив названий }

                                   { пунктов меню} 

pm,i,xn,yn : Integer;

Sim : Char;

Begin

{п. 1.1 }

Punkt[1] := ’ Операция 1 ’;

Punkt[2] := ’ Операция 2 ’;

Punkt[3] := ’ Операция 3 ’;

Punkt[4] := ’ Операция 4 ’;

Punkt[5] := ’ Выход      ’;

     {  Очистка экрана }

TextBackGround(Blue);

ClrScr;

{ Начальное положение курсора и меню }

pm := 1;

xn := 10;

yn := 3;

Repeat                { п. 2 }

Repeat              {  Цикл движения по меню }

GotoXY(xn,yn);

                  { Основные цвета меню }

TextBackGround(Blue);

TextColor(White);

For i := 1 to m do

Write(Punkt[i]);

                 { Текущий пункт перекрашиваем }

TextBackGround(Green);

TextColor(Red);

GotoXY(xn+dl*(pm-1),yn);

Write(Punkt[pm]);

      { Ожидаем действий пользователя на клавиатуре}

Sim := Readkey;

If Sim = #0 then

Begin { Управляющая клавиша }

Sim := Readkey;

Case Sim  of

#77 :              { Движение вправо }

If pm=m then
pm:=1

Else

pm:=pm+1;

#75 :              { Движение влево }

If pm=1 then
pm:=m

Else

pm:=pm-1;

End;  { Case }

End; { If }

Until (Sim = Enter) Or ( Sim = Esc); { Цикл движения }

If Sim = Enter then

{  Пункт меню выбран  }

If pm = m then

Sim := Esc {Выбран последний пункт меню - выход}

Else

Begin

GotoXY(30,10);

Write(’Выполняется ’,Punkt[pm]);

End;

Until Sim = Esc;  { Внешний цикл, окончание работы }

GotoXY(30,10);

Writeln(’Конец работы, нажмите любую клавишу ’);

Repeat Until KeyPressed;

End.

23.2. Модуль DOS
Модуль обеспечивает связь программы с MS DOS и содержит более 20 процедур и функций. Наиболее часто используются следующие из них.

1) GetTime(H,M,S,S100); – процедура; возвращает установленное в операционной системе текущее время в виде: H – часы (0..23), M – минуты (0..59), S – секунды (0..59), S100 – сотые доли секунды (0..99). Все переменные процедуры имеют тип Word.

2) GetDate (Y,M,D,DW); – процедура; возвращает установленную в операционной системе текущую дату в виде: Y – год (1980..2099), M – месяц (1..12), D – день (1..31), DW - день недели (0..6, где значению "0" соответствует воскресенье). Все данные выходные и имеют тип Word.

Пример.

GetTime(H,M,S,S100);

Writeln (’Текущее время: ’, H,’ часов ’, M,’ минут ’,

S,’ секунд’);

GetDate(Y,M,D,DW);

Writeln(’Текущая дата: ’,D,’.’,M:2,’.’,Y:4);

Модуль DOS позволяет запустить из Паскаль-программы другую программу по командной строке DOS, посмотреть атрибуты файла (время создания, имя, тип и т.д.), объем дисковой памяти (общей и свободной) и выполнять другие команды MS DOS.

23.3. Модуль Graph

Модуль представляет собой библиотеку более чем из 50 графических процедур, которые позволяют рисовать простейшие графические изображения (примитивы): линии, окружности, прямоугольники и т.д. Он обеспечивает использование в полной мере графических возможностей компьютера. Работает с графическими драйверами вида

тип_видеокарты.bgi.

Пример. 
Egavga.bgi – для видеокарт VGA и EGA или

Svga256m.bgi - для видеокарты SVGA.

Драйвер VGA поддерживает три цветовых режима с 16-ю цветами каждый:

VGALo

 число точек на экране – 640х200;  
4 страницы видеопамяти.

VGAMed

 число точек на экране 640х350; 
2 страницы видеопамяти.

VGAHi 

 число точек на экране 640х480;
1 страница видеопамяти.

Изображение формируется в видеопамяти. Для этого используют от 1 до 4 страниц. В каждый момент времени только одна страница является видимой (передающей изображение на экран). Еще одна страница может быть активной и использоваться для формирования нового изображения. Эти страницы могут быть разными или совпадать. Такой подход ускоряет формирование и смену изображений на экране.

Модуль Graph использует также драйверы шрифтов вида

название_шрифта.chr.

для вывода надписей.

23.3.1. Инициирование графического режима

Графический режим – это особый режим вывода на экран. Он является альтернативой текстовому режиму и отменяет его. Вначале графический режим обязательно должен быть запущен (инициирован), а в конце работы с ним – закрыт. Для этого используют следующие процедуры и функции.

1) Процедура InitGraph(Var grDr,grMd:integer;path:string);

Здесь 
grDr – графический драйвер,

grMd – графический режим (mode), допустимый для этого драйвера;

Path – полное имя каталога, в котором находятся графические драйверы. Если этот параметр задается пустой строкой (т.е. ’’), то подразумевается, что файлы драйверов находятся в текущем каталоге. Первые два параметра являются параметрами-переменными и должны быть объявлены в описании переменных.

Перед инициированием графического режима необходимо задать тип используемого драйвера. Это можно сделать самим или определить автоматически.

Пример.

grDr : = Detect; - драйвер определяется автоматически или

grDr : = Vga; -  задан тип драйвера Vga.

Система координат в графическом режиме определяется как это показано на рис. 30. Верхний левый угол экрана имеет координату (0,0). Координаты других крайних точек зависят от установленного драйвера (GrDr) и выбранного режима его работы (GrMd).

2) Функция GraphResult; аргументов нет. Возвращает код ошибки для последней выполненной графической операции (в том числе и InitGraph, и DetectGraph). Тип результата – целый.

Определены более десяти кодов ошибок. Если ошибки нет, то функция GraphResult возвращает код, равный 0 (соответствующая константа – GrOk). В противном случае возвращаемый код не равен 0. 

Причину ошибки можно узнать, обратившись к функции GraphErrorMsg. Входным параметром этой функции должен быть код ошибки, а возвращает функция строку, содержащую причину возникновения ошибки.

Пример. WriteLn(’Ошибка графики, код: ’,ErCode,’. ’,GraphErrorMsg(ErCode));
3) Процедура CloseGraph; – закрывает графический режим и восстанавливает исходный текстовый режим экрана, освобождает память, занимаемую графическим драйвером.

Таким образом, в программе, работающей с модулем Graph должны быть фрагменты вида

Program...

Uses

graph;

.....

Var

grDr ,grMd,Er:integer;

......

Begin

grDr := Detect;{автоматическое определение драйвера}

InitGraph(grDr,grMd,’’);{драйвер в текущем каталоге}

Er := GraphResult;  { Код ошибки }

If Er=grOk then

Begin  { Графический режим }

.....

CloseGraph;

End

Else

WriteLn(’Ошибка графики: ’,Er,’. ’,GraphErrorMsg(Er));

.....

end.

Графический режим может быть переключен на текстовый и обратно следующим образом.

Begin

{Графический режим}

.......

RestoreCrtMode; { переход в  текстовый режим Crt }

.......

SetGraphMode(grMd); { Снова графический режим }

.......

end;

23.3.2. Вывод точек на экран

В графическом режиме имеется возможность задавать цвет изображения и фона (из палитры). Для этой цели используются следующие процедуры.

1) Установка цвета фона
SetBkColor(Цвет:Word); { Цвет из палитры }

2) Цвет рисунка устанавливается процедурой

SetColor (Цвет:Word);

Цвет – номер цвета из палитры (от 0 до 15). Палитра устанавливается при инициировании графического режима. Ее можно изменять с помощью процедуры SetAllPalette. Можно заменить один цвет в палитре на другой с помощью процедуры SetPalette.

3) Очистка экрана перед выводом изображения осуществляется с помощью процедуры ClearDevice.
Вызов пары процедур:              SetBkColor(Цвет);

   ClearDevice;

окрасит весь экран заданным цветом.

4) Количество цветов в палитре позволяет определить функция   GetMaxColor.

5) Вывод на экран отдельных точек обеспечивает процедура

PutPixel(x,y,цвет:Word).

Здесь (x,y) – координаты точки, а цвет – ее цвет (из палитры).

В графическом режиме используется своя система координат. Координаты задаются не в символах, а в точках экрана (пикселях, Pixel). 

6) Для перемещения курсора (чаще называемого текущим указателем – CP, Current Pointer) по экрану используется процедура: 

MoveTo(X,Y : integer);
Курсор помещается в точку с координатами (X,Y). Эта процедура аналогична процедуре GotoXY для текстового режима, но в графическом режиме курсор на экране невидим. Максимальные значения координат экрана X и Y (см. рис. 2.21) можно определить с помощью не имеющих аргументов функций:

GetMaxX и  GetMaxY,

Пример.


Xmax := GetMaxX;

Рассмотрим пример программы с использованием описанных выше процедур.

Задача. Составить программу "Звездное небо", которая выводит на экран разноцветные точки в случайных местах. В программе для задания координат и цвета будем использовать генератор случайных чисел Random. Диапазон случайных чисел определяется функциями GetMaxX, GetMaxY и GetMaxColor соответственно. Программа будет иметь вид, приводимый ниже.

Program Plot_Stars;

Uses

Crt, Graph;

Var

Er, GDr, Gmode : Integer;

X, Y : Integer;

Color :Integer;

Begin

{ Параметры графического режима }

Gdr := Vga;

Gmode := VgaLo;

InitGraph(Gdr,GMode,’’);

Er := GraphResult;

If Er = GrOk then

Begin

Randomize; {Инициализация генератора случайных чисел}

SetBkColor(Black);

ClearDevice;

Repeat

Color := Random(GetMaxColor)+1;

X := Random(GetMaxX);

Y := Random(GetMaxY);

PutPixel(x,y, Color);

Delay(10);

Until Keypressed;

CloseGraph;

End

Else

  begin

    Writeln(’Ошибка графики: ’,Er,’. ’,GraphErrorMsg(Er));

    Writeln(’ Нажмите любую клавишу...’);

    Repeat Until KeyPressed;

  end
End.

23.3.3. Вывод текста (надписей) в графическом режиме

В графическом режиме предоставляется значительно больше (по сравнению с текстовым режимом) возможностей для вывода текста. Можно выбрать шрифт, его размер, ориентацию символов, способ выравнивания текста и т.п. К сожалению, кириллица допустима не во всех стандартно поставляемых шрифтах.

Для вывода текста в графическом режиме используются следующие процедуры.

1) SetTextStyle(Шрифт, Направление:Word; Размер:1..10); 

- задает шрифт и размер букв.

Шрифт задается либо предопределенной константой, либо ее значением:

Имя константы
Значение
Характеристика

DefaultFont
0
растровый, матрица 8*8

TriplexFont
1
векторный, с засечками

SmallFont
2
векторный, мелкий

SansSerifFont
3
векторный, без засечек

GothicFont
4
векторный, готический

Направление задается следующим образом:

HorizDir = 0 – вывод текста слева направо (горизонтальная надпись),

VertDir = 1 – вывод текста снизу вверх (вертикальная надпись).

Размер определяет коэффициент увеличения символа. Для шрифта DefaultFont размер, равный 1, определяет исходную матрицу для представления символов (8х8 точек). Размер 2 приведет к выводу на экран образа символа, созданного на матрице 8х8, в область размером 16х16 точек и т.д. При больших размерах символы шрифта DefaultFont сильно искажаются, а качество векторных шрифтов остается приемлемым.

Пример.
SetTextStyle (1,0,1);
Установлен шрифт TriplexFont с размером символов 8х8 точек и горизонтальным направлением вывода.

2) OutТext(строка); – процедура; осуществляет вывод строки от текущего положения курсора выбранным процедурой SetTextStyle шрифтом.

3) OutTextXY(X,Y,строка); – процедура; осуществляет вывод строки от точки с координатами (X,Y) шрифтом, заданным процедурой SetTextStyle.

Пример. 

SetTextStyle (1,0,1);

OutTextXY(GetmaxX Div 2, GetmaxY div 2, ’Graphic’);

В результате выполнения этих процедур будет выведено слово Graphic в середине экрана.

Для вывода чисел в графическом режиме (например, максимального значения координаты х – Xmax) необходимо преобразовать их в строку с помощью процедуры

Str(Число, Строка);

в которой входной параметр Число является выражением целого или вещественного типа, а выходной параметр Строка – его строковым представлением. Например, следующий фрагмент программы обеспечивает вывод на экран, начиная от точки с координатами (80,50), значения Xmax.

Xmax := GetmaxX;

Str(Xmax,StrX);

OutTextXY(80,50,’Xmax=’+StrX);

Текущие параметры, влияющие на вид выводимого текста: шрифт, его размер, направление и выравнивание, можно узнать с помощью процедуры GetTextSettings.

23.3.4. Построение графических изображений

Модуль Graph широко применяется для построения графиков и вычерчивания линий, окружностей, эллипсов, прямоугольников и т.д. Используемые линии могут быть сплошные, штриховые и т.д. Замкнутые области можно закрашивать сплошной заливкой, штриховкой и др. Большинство процедур модуля Graph используется для построения графических изображений. При этом в начале необходимо задать стиль (тип и толщину) линии контура с помощью процедуры

1) SetLineStyle(Тип_линии, Образец, Толщина : Word);

Определены следующие константы и их значения для выбора Типа_линии:

Имя константы
Значение
Характеристика

SolidLn 
0
непрерывная

DottedLn 
1
линия из точек

CenterLn 
2
штрих–пунктирная

DashedLn 
3
пунктирная

UserBitLn 
4
определяемая пользователем

Образец должен быть равен 0, если используется стандартная линия (Типа 0..4) и отличен от 0, если пользователь задает свой образец.

Для параметра Толщина определены константы:

NormWidth = 1 – нормальная толщина,

ThicklWidth = 3 – жирная (тройная) линия.

2) Построение линий выполняется процедурами:

Line(X1,Y1,X2,Y2); – проводит линию между точками с координатами (X1,Y1) и (X2,Y2);

LineTo(X,Y); – проводит линию из текущего положения курсора до точки с  координатами (X,Y).

3) Процедура Circle(X,Y:integer; R:Word) – строит окружность радиусом R с центром в точке (X,Y).

4) Процедура Rectangle(X1,Y1,X2,Y2) – строит прямоугольник с координа​тами: (X1,Y1) – верхний левый, (X2,Y2) – нижний правый угол.

5) Процедура DrawPoly(колич_координат:word; var координаты_вершин); рисует многоугольник, используя текущие цвет и тип линии. Параметр координаты_вершин является нетипизированным и содержит координаты (X, Y) вершин многоугольника. Тип координат – Word. Параметр колич_координат задает их число.

Замечание. Для вычерчивания n-угольника нужно задать n+1 координату, причем вершина (n+1) должна иметь координаты, совпадающие с вершиной 1. Обычно координаты_вершин задают как типизированную константу. При этом используют определенный в модуле Graph тип PointType, представляющий собой запись, содержащую два значения: X и Y.

Ниже приведен текст программы, использующей процедуру DrawPoly для рисования шестиугольника. Обратите внимание на способ задания координат вершин в виде типизированной константы–массива.

Program DemoDrawPoly;

 uses 

    Graph,Crt;

 const

  N=7; { Количество вершин многоугольника +1}

  MultiRect: array[1..N] of PointType = {Координаты вершин}

   ((X: 200; Y: 100), {Координаты первой вершины и }

    (X: 285; Y: 155),

    (X: 285; Y: 255),

    (X: 200; Y: 300),

    (X: 115; Y: 255),

    (X: 115; Y: 155),

    (X: 200; Y: 100)); {последней (замыкающей) должны совпадать}

 var 

     Gd, Gm, GrEr: Integer;

 begin

  Gd := Detect;

  InitGraph(Gd, Gm, ’d:\bp\bgi’);

  GrEr := GraphResult;

  if GrEr = grOk then

    begin

      DrawPoly(N, MultiRect);

      OutTextXY(50,20,’Нажмите любую клавишу...’);

      Repeat Until KeyPressed;

      CloseGraph;

    end

  else

    begin

      WriteLn(’Ошибка графики: ’,GrEr,’. Нажмите Enter...’);

      Readln;

    end;

end.

Сведения о других примитивах можно найти в книгах по Турбо Паскалю или в справочной системе Турбо Паскаля.

Все фигуры, как отмечалось выше, рисуются текущим цветом и типом линии. Цвет линии задается из имеющегося активного набора цветов – палитры.

Пример. Написать программу построения графика заданной функции Y=f(X) для аргумента X, изменяющегося от xn до xk. Для определенности будем считать, что Y=sin(x), xn = –2π, xk = 2π. 

Общие соображения по написанию программы для построения графиков приводились в п. 5.1.4. При построении графика должны быть определены масштабы – коэффициенты, с помощью которых натурные величины (X, Y) будут переводиться в графические координаты (xG, yG). В общем случае пределы изменения натурных величин: аргумента (xn, xk) и функции (Ymin, Ymax) должны быть отображены на выбранные пределы изменения графических координат:

xGn – x графическое начальное,

xGk – x графическое конечное,

yGn – y графическое начальное, 

yGk – y графическое конечное.

С учетом сказанного, масштабы можно определить по формулам:

        mx:=(xGk-xGn)/(xk-xn);      { Масштаб по X }

        my:=(yGk-yGn)/(ymax-ymin);  { Масштаб по Y }

Максимальное и минимальное значения функции (хотя бы приблизительные, лучше с запасом в большую сторону) должны быть заданы пользователем. Для перехода к графическим координатам получим следующие соотношения:

          xG:=Round(xGn+(x-xn)*mx);

          yG:=Round(yGn+(ymax-y)*my);

Для построения графика воспользуемся процедурой LineTo, которая будет проводить линию от точки к точке.

Программа построения графика может быть такой.

Program PlotGrafik;

uses 

   Crt, Graph;

const

  xGn=20;   { Константы, определяющие }

  yGn=50;   { область экрана,         }

  xGk=600;  { в которую будет         }

  yGk=470;  { выводиться график       }

  xn=-2*pi/2;  { Начальное и }

  xk=2*pi/2;   { конечное значение аргумента }

  ymin=-1;   { Предполагаемые минимальное и }

  ymax=1;    { максимальное значения функции }

  kT=150;      { Количество точек графика }

var

  grDr, grMd, grEr : integer;

  x, y, Dx : real;

  xG, yG  : integer;

  mx, my : real;  { Масштабы по X и Y }

Function f(x:real):Real;

    begin

      f:=sin(x); {Функция, для которой строится график}

    end;

begin

  grDr:=Detect;

  InitGraph(GrDr, GrMd, ’D:\bp\BGI’);

  GrEr:=GraphResult;

  if GrEr=GrOk then { Графический режим удачно открыт}

    begin

      SetBkColor(Blue);

      ClearDevice; {Очистим экран, сделаем фон голубым} 

      Dx:=(xk -xn)/(kT-1); { Шаг изменения аргумента }

      mx:=(xGk-xGn)/(xk-xn);      { Масштаб по X }

      my:=(yGk-yGn)/(ymax-ymin);  { Масштаб по Y }

      SetLineStyle(0,0,1);        { Одинарная линия }

      SetColor(Red); { Оси нарисуем красным }

      Line(Round(xgn+(0-xn)*mx),ygn-20,

      Round(xgn+(0-xn)*mx),ygk);    {Ось Y, на ней X=0}

      Line(xgn,Round(ygn+(ymax-0)*my),

      xgk+20,Round(ygn+(ymax-y)*my));{Ось X, на ней Y=0}

       {Изображение стрелки на конце оси X и надписи X}

      MoveTo(xgk+20,Round(ygn+(ymax-y)*my)); 

      LineRel(-8,4);

      MoveTo(xgk+20,Round(ygn+(ymax-y)*my)); 

      LineRel(-8,-4);

      MoveRel(0,-25);

      SetTextStyle(1,0,1); {Шрифт для обозначения осей}

      OutText(’X’);

       {Рисование стрелки на конце оси Y и надписи Y}

      MoveTo(Round(xgn+(0-xn)*mx),ygn-20);

      LineRel(-4,8);

      MoveTo(Round(xgn+(0-xn)*mx),ygn-20);

      LineRel(4,8);

      MoveRel(0,-15);

      OutText(’Y’);

      SetColor(white); { Белый цвет для линии графика }

      x:=xn;           { Текущее значение для X }

      y:=F(x);         { Значение функции для этого X }

      xG:=Round(xgn+(x-xn)*mx); {Графическая коорд. по X}

      yG:=Round(ygn+(ymax-y)*my);{Графическая коорд. по Y}

      MoveTo(xg,yg); { Курсор в эту точку экрана }

      While x<=xk do {Цикл для всех значений аргумента}

        begin

          x:=x+dx;

          y:=f(x);

          xG:=Round(xgn+(x-xn)*mx);

          yG:=Round(ygn+(ymax-y)*my);

          LineTo(xG,yG);

        end;

      SetTextStyle(0,0,1);  { Шрифт для вывода текста }

      OutTextXY(5,5,’График функции Y=Sin(X)’);

      OutTextXY(5,20,’Нажмите любую клавишу...’);

      Repeat Until KeyPressed
      CloseGraph; { Закрываем графический режим }

    end

  else

    begin

      WriteLn(’Ошибка графики ’,GrEr,’ ’,GraphErrorMsg(GrEr));

      WriteLn(’Нажмите Enter...’);

      ReadLn;

    end;

End.

Закраска замкнутой области выполняется процедурами:

5) SetFillStyle(Тип_закраски, Цвет : Word); – устанавливает тип (стиль) закраски и ее цвет. 

В Турбо Паскале определено большое количество типов закраски, которые можно задавать либо с использованием предопределенных имен констант, либо их значений. Приведем некоторые из них:

0 – заполнение области фоновым цветом;

1 – сплошное заполнение области заданным цветом;

2 – заполнение области горизонтальными линиями                  ;

3 – заполнение области наклонными линиями                 ;

4 – заполнение области толстыми наклонными линиями             

. . . . . . . 

11 – заполнение области частыми точками;

12 – заполнение области стилем, определяемым пользователем.

Цвет  -     цвет закраски (из палитры).

Пример. 

SetFillStyle(4,1);

6) FloodFill(x,y,Цвет_границы : Word); – закрашивает ограниченную область, используя текущий образец закраски. 

Точка (x, y) является внутренней точкой закрашиваемой области. Для заполнения области, ограниченной цветом, определяемым параметром Цвет_границы, используется текущий образец закраски. 

Например, программа, которая рисует концентрические окружности с координатами центра (150, 100) всеми 15-ю цветами (MaxColors) и закрашивает получившиеся кольца всеми 11-ю типами заливки и цветом, отличным от цвета контура будет иметь вид.

Program All_Colors;

Uses

Graph;

Var

Er,GraphDriver, GraphMode : Integer;

X, Y : Integer;

Radius, Color : Integer;

Begin

GraphDriver := Vga;

GraphMode := VgaHi;

InitGraph(Graphdriver, GraphMode,’’);

Er := GraphResult;

If Er=Grok then

begin
ClearDevice; {Черный цвет фона – по умолчанию }

For Color := 0 to MaxColors do

Begin

SetColor(Color);

For Radius := 1 to 11 do

Begin

Circle(150, 100, Radius*10);

SetFillStyle(Radius, MaxColors - Color);

Floodfill(150-Radius*10+5, 100, Color);

End;

{ Останов после вывода 11 окружностей одним цветом контура}

readln; 

End;

CloseGraph;

end

Else

Writeln(’Ошибка графики ’,Er);

Readln;

end.

23.3.5. Запоминание и вывод изображений

Паскаль предоставляет возможность запоминать двоичный образ экрана в памяти ЭВМ и выдавать его на экран. Таким образом можно создать мультипликацию. Для этого используются следующие процедуры и функции модуля Graph.

1) ImageSize(X1,Y1,X2,Y2 : Word): Word; – функция, которая определяет размер блока памяти в байтах для сохранения двоичного образа прямоугольной области экрана с координатами левого верхнего (X1,Y1) и правого нижнего (X2,Y2) углов. Максимальный размер блока, в который может быть помещен образ экрана, равен 64 Кбайт.

Пример.
Size := ImageSize(1,1,100,50);
Переменной Size присваивается значение размера области памяти, необходимой для сохранения образа экрана с координатами (1,1,100,50).

2) GetMem(Var Указатель : Pointer; Размер : Word); – процедура, которая создает динамическую переменную указанного размера и возвращает ее адрес в Указатель.
Пример. GetMem (P, Size); {выделена область для хранения образа размером Size из предыдущего примера. Адрес начала области хранится в указателе p.}

3) GetImage(X1,Y1,X2,Y2 : Word; Var битовый_массив); – сохраняет двоичный образ заданной области экрана (область задается координатами (X1,Y1) – левый верхний угол и (X2,Y2) – правый нижний угол). Параметр битовый_массив является выходным и не имеет типа (нетипизированный). Именно в нем сохраняется двоичный образ области экрана, задаваемой первыми четырьмя параметрами процедуры. Размер этого массива необходимо определять, используя функцию ImageSize (см. выше). Так как процедура GetMem (см. выше) создает динамическую переменную именно такого размера и указывает на нее с помощью Указателя, то фактическим параметром, соответствующим формальному битовый_массив, при вызове процедуры GetImage дожнен быть Указатель^.

Пример. 
GetImage (1,1,100,50, P^);
По адресу, находящемуся в указателе P, сохранен образ прямоугольной области экрана с координатами (1,1,100,50).

4) PutImage(X,Y: Word; Var битовый_массив; Двоич_опер : Word); – в прямоугольную область экрана, верхний угол которой задается координатами (X,Y), выводится двоичный образ, сохраняемый в параметре битовый_массив. Параметр Двоич_опер определяет, какая операция будет выполняться между битами выводимого изображения и битами изображения (возможно, фонового), находящегося в прямоугольной области экрана, в которую осуществляется вывод. Определены следующие константы и их значения.

NormalPut  = 0; – вывод изображения безо всякого взаимодействия с имеющимся на экране (как на белый лист).

XorPut     = 1; – выполняется операция XOr между битами выводимого изображения и экраном. Если в области, куда производится вывод, было точно такое же изображение, то оно сотрется. Такой прием стирания изображения часто используют при создании мультипликаций и перемещений картинок по экрану.

OrPut      = 2;выполняется операция Or между битами выводимого изображения и экраном.
AndPut     = 3;выполняется операция And между битами выводимого изображения и экраном.
NotPut     = 4; – сначала выполняется операция Not над битами выводимого изображения (получается негатив), а потом оно выводится как NormalPut. Таким образом, на экране получается негативное изображение образа.

Пример. 
PutImage(20, 10, P^, NotPut); 
Начиная от точки экрана с координатами (20, 10), выведено негативное изображение, сохраненное по адресу, помещенному в указатель P.

Задача. Составить программу, которая передвигает горизонтально по экрану круг радиуса R=20 пикселей.

Алгоритм решения этой задачи может быть таким.

1.1. Задать драйвер и режим.

1.2. Инициировать Graph.

2. Если Ошибка=GrOk то

2.1. Очистить экран.

2.2. Задать тип и цвет линии.

2.3. Нарисовать и закрасить круг.

2.4. Задать границы области, которую нужно сохранить, так, чтобы
в ней уместилось все сохраняемое изображение.

2.5. Сохранить образ.

2.6. Повторять

а) Задержать вывод;

б) Стереть изображение;

в) Изменить координату х1, например: х1=х1+10;

г) Вывести изображение;

Пока не нажата любая клавиша

2.7. Закрыть Graph

3. Закончить.

Программа, реализующая описанный алгоритм, приведена ниже.

Program MoveCircle;

Uses

Crt,Graph;

Var

P : pointer;

Size : Word;

Er,Gd,Gm : Integer;

x1,y1,x2,y2 : Integer;

Begin

Gd := Detect;

Gm := VgaHi;

InitGraph(Gd,Gm,’C:\bp\bgi’);

Er := Graphresult;

If Er = GrOk then

Begin           { графический режим установлен }

SetBkColor(Blue);

Cleardevice;

SetTextStyle(0,0,2);

OutTextXY(10,15,’Нажми любую клавишу ...’);

Setcolor(Red);

SetLineStyle(0,0,3);

Circle(150,120,20);

SetFillStyle(8,Yellow); { штриховка ххх желтым }

         { заштриховать до красного }

FloodFill(150,120,Red);

{ координаты сохраняемой области }

x1 := 129;

y1 := 99;

x2 := 171;

y2 := 141;

{ Сохранение образа }

Size := ImageSize(x1,y1,x2,y2);

GetMem(p,Size);

GetImage(x1,y1,x2,y2,P^);

{Перемещение образа}

Repeat

Delay(50);

{Стираем старое изображение}

PutImage(x1,y1,P^, XorPut);

     {Перемещение круга}

x1:=x1+10;

PutImage(x1,y1,P^, NormalPut);

Until KeyPressed; { Останов изображения }

CloseGraph;

End

Else

begin

 Writeln(’Ошибка графики ’,Er,’. ’+GraphErrorMsg(Er)+

        ’. Нажмите Enter...’);

Readln;

End;

End.

Имитация более сложного движения предполагает перемещение не только самого объекта, но и его частей (рук, ног, глаз и т.д.). Такое движение можно изобразить с помощью нескольких рисунков. Каждый из этих рисунков соответствует определенному положению движущейся части. Для имитации движения необходимо организовать цикл, в котором осуществляется поочередный вывод и стирание указанных рисунков с соответствующим изменением координат выводимых областей. При этом получится маленький мультфильм.

Рассмотрим в качестве примера, как организовать имитацию движения «колобка» за счет вывода двух положений его ножек, приведенных на рис. 2.22, а) и б). Пусть объект движется горизонтально, т.е. при перемещении изменяется только координата х (на величину step).

Можно предложить следующий алгоритм вывода на экран движущегося «колобка».

1. Очистить экран, закрасив его цветом фона.

2. Нарисовать и запомнить в разных местах видеопамяти 2 рисунка.

3. Повторять

Выводить рисунки на экран поочередно, меняя координату х.

Пока не будет нажата клавиша Или координата х не выйдет за экран.

4. Закончить.

Уточняем алгоритм.

1.1.Инициировать графический режим.

1.2. Если Ошибка = GrOk, то

1.2.1.Очистить экран, закрасив его цветом фона.

1.2.2. Нарисовать рис. 31, а).

1.2.3. Определить размер прямоугольника, который охватывает этот рисунок.

1.2.4. Выделить память для сохранения первого рисунка, начиная от адр. Р1.

1.2.5. Сохранить первый рисунок по адресу Р1.

1.2.6. Нарисовать рис. 31, б).

1.2.7. Определить размер прямоугольника, который охватывает этот рисунок.

1.2.8. Выделить память для сохранения второго рисунка, начиная от адр. Р2.

1.2.9. Сохранить второй рисунок по адресу Р2.

1.2.10. Задать значения координат x и y для вывода первого рисунка

1.2.11. Очистить экран.

1.2.12. Повторять

а) Вывести первый рисунок (из адреса Р1), начиная от координат (х,у).

б) Задержать вывод.

в) Стереть изображение (вывести рис. в том же месте с опер. XorPut).

г) Изменить координату х: x:=x+step.

д) Вывести второй рисунок (из адреса Р2), начиная от координат (х,у).

е) Задержать вывод.

ж) Стереть изображение.

з) Изменить координату х: x:=x+step.

Пока не будет нажата клавиша Или координата х не выйдет за экран.

1.2.13. Закрыть графический режим.

Иначе

Вывести ошибку графики.

2. Закончить.

Программа, реализующая этот алгоритм с использованием процедур, которые изображают тело колобка (kolob) и два положения его ног (noga1 и noga2), приведена ниже. Две фазы движения колобка воспроизводятся процедурами kolob1 и kolob2. Они же сохраняют эти изображения в видеопамяти, т.е. выполняют пункты 1.2.2 – 1.2.9 алгоритма.

Program kolobok;

   Uses 

      Crt,Graph;

   Var

       Gd,Gm,Er:integer;

       step:integer;

       X,Y:integer;

       Size1,Size2:Word;

       p1,p2:Pointer;

   Procedure kolob;

         { Рисует колобок в виде закрашенного круга }

    Begin

       SetColor(White);

       Circle(70,269,50);

       SetFillStyle(1,Yellow);

       FloodFill(70,259,White);

       FloodFill(80,249,White);

    End; { Kolob}

   Procedure noga1; { Рисует одну ногу }

    Begin

      SetLineStyle(0,0,3);  { Жирная сплошная }

      SetColor(Yellow);     { Рисовать желтым }

      Line(70,289,70,329);

      Line(70,329,90,329);

      Arc(80,329,0,180,10);

      SetFillStyle(1,Yellow); {Сплошное заполнение желтым}

      FloodFill(80,326,Yellow);

    End; { Noga1 }

   Procedure noga2; { Рисует две ноги }

     Begin

      SetLineStyle(0,0,3);

      SetColor(Yellow);

      SetFillStyle(1,Yellow);

      Line(90,294,100,324);

      Line(98,324,117,324);

      Arc(108,324,15,190,9);

      FloodFill(108,321,Yellow);

      Line(50,294,40,324);

      Line(40,324,56,324);

      Arc(49,324,350,160,9);

      FloodFill(49,321,Yellow);

    End; { noga2 }

    Procedure kolob1;

      { Рисует и запоминает одну фазу движения - }

      { колобок с одной ногой }

    begin

      kolob;

      noga1;

      size1:=ImageSize(15,219,128,331);

      GetMem(p1,size1);

      GetImage(15,219,128,331,p1^);

      ClearDevice;

    end;  { kolob1 }

    Procedure kolob2;

      { Рисует и запоминает другую фазу движения - }

      { колобок с двумя ногами }

    begin

      kolob;

      noga2;

      size2:=ImageSize(15,219,128,331);

      GetMem(p2,size2);

      GetImage(15,219,128,331,p2^);

      ClearDevice;

    end;  { kolob2 }

   Begin

     Gd:=Vga;

     Gm:=VgaMed;

     InitGraph(Gd,Gm,'D:\Bp\Bgi');

     Er:=GraphResult;

     If Er=GrOK then

       Begin

         SetBkColor(blue);

         kolob1;

         kolob2;

        { Начальные координаты и шаг перемещения}

         x := 55;

         y := 100;

         step := 10;

         Repeat

           PutImage(X,Y,p1^,NormalPut);

           Delay(150);

           PutImage(X,Y,p1^,XORPut);

           X:=X+step;

           PutImage(X,Y,p2^,NormalPut);

           Delay(150);

           PutImage(X,Y,p2^,XorPut);

           X:=X+step;

         Until X>469;

         CloseGraph;

       End

     else

       Writeln('Ошибка графики: ',Er,'. ',GraphErrorMsg(Er));

     WriteLn('Нажмите Enter...');

     Readln;

   End.

24. Рекомендации по оформлению текста программ

Одна из фаз разработки программы связана с ее написанием. Следует с самого начала приучиться к хорошему стилю программирования и внешнего оформления программ. Если пишется даже простая учебная программа, лучше настроить себя на то, что это очень важная работа, а программа заинтересует многих пользователей, и найдутся желающие ее приобрести. 

Помимо программиста–разработчика большинство программ читается и используется другими людьми. Поэтому при написании программы в общем случае нужно быть готовым учитывать мнение трех лиц.

Прежде всего, мнение человека, который будет использовать программу (заказчик, в том числе и преподаватель). Естественно желание пользователя иметь дело с текстом, представленным в ясной и понятной форме, с хорошо продуманным и не допускающем двусмысленного толкования вводом и выводом данных.

Другим лицом, мнение которого необходимо учитывать, является программист, который захочет ознакомиться с программой, например, с целью ее адаптации к новым условиям применения. В любом случае автору, написавшему понятную и хорошо составленную программу, будут признательны.

Сам автор является третьим лицом, которое должно приниматься во внимание при разработке программы. Даже если никто не будет использовать или проверять написанную программу, усилия, направленные на достижение легкости в понимании программы, будут потрачены не напрасно. 

Советы, что следует делать и чего избегать, не преследуют цель поставить программиста в жесткие рамки. Эти правила помогут писать более совершенные программы. Если, исходя из здравого смысла, постоянно выполнять приводимые рекомендации, то программы станет легче писать и применять.

1. Используйте содержательные идентификаторы. Лучшим способом передать смысл переменной или константы является присвоение ей имени, которое отражает назначение переменной или константы. Например, если речь идет о количестве положительных чисел, то идентификатор KolPol предпочтительнее идентификаторов A, Y или еще каких-нибудь в этом духе. При выборе идентификаторов, состоящих из нескольких слов (или их частей) рекомендуется начинать их с заглавной буквы и/или разделять символом подчеркивания, например: MasSlov – массив слов; File_Exist – имя функции, определяющей наличие файла и т. д.

2. Используйте ступенчатую форму записи текста программы. Сдвиги (отступы) должны показывать каждый уровень вложенности операторов. Все программы, приведенные в этой работе, написаны именно так. Величина отступа не должна быть ни слишком малой, ни слишком большой. Можно рекомендовать отступ, равный двум символам. Все вводимые отступы запоминаются системой программирования и возврат на предыдущий уровень всегда можно осуществить нажатием клавиши "BackSpase". Не откладывайте использование ступенчатой формы "на потом". Приведение программы в надлежащий вид после ее отладки потребует больше времени или вообще не будет выполнено. 

3. Пишите приглашения ясным языком. Не следует стремиться к литературным шедеврам, вполне достаточно, чтобы они однозначно понимались. Если для завершения работы с программой необходимо нажать какую-либо (или любую) клавишу, выдайте соответствующее сообщение пользователю. Вообще, никогда не оставляйте пользователя в неведении относительно того, что ему делать в каждый момент.
4. Не рекомендуется размещать несколько операторов на одной строке. В этом случае создаются искусственные трудности для чтения программы. Кроме того, такую программу труднее отлаживать.

5. При записи оператора If ... Then ... Else записывайте слово Else против соответствующего ему слова If. Например,

       If X>0 Then
         If Y>0 Then
           Z := Sqrt(X*X+Y*Y)

         Else

           Z := X*X

       Else
         Z := Y*Y;

Разобраться, что делает этот фрагмент программы намного легче по сравнению с таким его представлением:

If  X>0 Then If Y>0 Then Z:=Sqrt(X*X+Y*Y) Else Z:= X*X Else Z:= Y*Y;

6. Не допускайте выхода текста программы за пределы экрана (80 символов в строке). Такую программу неудобно читать, а при выводе на печать длинный оператор может быть "разрезан" принтером так, что это не будет соответствовать правилам языка программирования. Такие операторы пишите на нескольких строках, соблюдая при переносе правила Паскаля. Например, при переносе текстовой константы, запишите ее в виде нескольких констант. Идентификатор оператора рекомендуется не перекрывать строками продолжения, как это показано ниже:

WriteLn(' Введите 3 вещественных числа, ',

        'нажимая после каждого клавишу ENTER');

ReadLn(a, b, c);

или
If ((a>b) and (c>d)) or

   ((KolSym <>10) and (Dlin > nMax)) Then ....

6. Используйте комментарии. Одним из способов достижения легкости понимания программы является включение в программу комментариев. Комментарии можно условно разбить на четыре класса.

К первому классу относится комментарий, располагаемый обычно после заголовка программы и объясняющий ее назначение. Здесь же уместно поместить сведения об авторах. Второй класс состоит из комментариев, в которых приводится назначение каждой константы и переменной. К третьему классу относятся комментарии, которые разбивают программу на ряд частей из взаимосвязанных операторов. В четвертый класс включаются комментарии, объясняющие необычные или неочевидные аспекты используемого алгоритма.

Приводимый ниже пример поясняет сказанное.

  Program Minimum;

    { Нахождение минимального элемента в массиве и его индекса }

    { Программа написана студентом 1-АИТ-9 Ивановым П.И. }

  Const

    Nmax = 100; { Максимальное количество элементов в массиве }

  Var

      A : Array[1..Nmax] Of Real; { Массив }

    Min : Real; { Минимальный элемент }

    Imin,       { Индекс минимального элемента } 

       n,       { Действительное количество элементов в массиве}

       i  : integer;    { Индекс элемента массива }

  Begin

         Ввод исходных данных }

    Writeln('Введите количество чисел');

    Readln(n);

    Writeln('Вводите элементы массива');

    For i := 1 To n Do

      Read(X[i]);

                        { Нахождение минимума }

    Min := X[1]; {За минимальный принимаем 1-ый элемент массива}
    Imin := 1;          { Индекс этого элемента }

    For i := 2 To n Do  {Цикл просмотра элементов массива}
      If Min > X[i] Then

        Begin

           Min := X[i];

          Imin := i;

        End;

                        { Вывод результата и завершение }

    Writeln('Минимум = ', Min:8:2, ' Номер = ', Imin);

    Writeln('Работа завершена. Нажмите клавишу ENTER');

    Readln;             { Ожидание нажатия на ENTER}

  End.

Program Имя;			заголовок программы











Begin				начало обработки

Рис. 2.2. Схема алгоритма работы оператора IF

Рис. 2.3. Схема алгоритма работы �неполного оператора IF





Рис. 2.4. Общая схема цикла

Рис. 2.5. Схема работы оператора For

Рис. 2.6. Схема выполнения оператора цикла Repeat

Рис. 2.7. Схема выполнения оператора цикла While

Рис. 2.9. Механизм вызова процедуры

Рис. 2.11. Структура модуля Турбо Паскаля

Рис. 2.12 Физическое удаление записи из файла

Рис. 2.13. Структура элемента списка

Рис. 2.14. Структура списка

Рис. 2.15. Переход от одного элемента списка к следующему

Рис. 2.16. Удаление элемента из списка 





а





NUk





Nil





NUk^





PrUk





PrUk^





  Nuk^.Ukaz := PrUk^.Ukaz;   {1}


 PrUk^.Ukaz := Nuk;          {2}





NUk





Nil





NUk^





PrUk





PrUk^





2





1





б





Рис. 2.17. Включение элемента в список

Рис. 2.18. Последовательность создания очереди LIFO





Рис. 2.19. Последовательность создания очереди FIFO

Рис.2.20. Координаты углов�экрана в текстовом режиме

Рис.2.21. Координаты углов экрана при работе �в графическом режиме





Рис. 2.22. Представление двух фаз движения

